



***D.Herrmann G.Schnellhardt***

# ***Schneider CPC***

---

## ***Wirtschaft***



***Programm-  
Beispiele für den  
Anwender***

**2**

**iWT**




*D. Herrmann G. Schnellhardt*

# ***Schneider CPC***

---

## ***Wirtschaft***



***Programm-  
Beispiele für den  
Anwender***

**2**

**iWT**

ISBN 3-88 322-153-8

1. Auflage 1985

Alle Rechte, auch die der Übersetzung, vorbehalten. Kein Teil des Werkes darf in irgendeiner Form (Druck, Fotokopie, Mikrofilm oder einem anderen Verfahren) ohne schriftliche Genehmigung des Verlages reproduziert oder unter Verwendung elektronischer Systeme verarbeitet, vervielfältigt oder verbreitet werden.

Der Verlag übernimmt keine Gewähr für die Funktion einzelner Programme oder von Teilen derselben. Insbesondere übernimmt er keinerlei Haftung für eventuelle, aus dem Gebrauch resultierende, Folgeschäden.

Schneider ist ein Warenzeichen der Schneider Rundfunkwerke GmbH. & Co.

Printed in Western Germany

© Copyright 1985 by IWT-Verlag GmbH  
Vaterstetten bei München

Holdenrieds Druck- und Verlags-GmbH, Füssen  
Umschlaggestaltung: Kaselow und Partner, München

# VORWORT


Dieser Band enthält eine Sammlung von 40 BASIC-Programmen aus den Bereichen Finanzmathematik, Unternehmensforschung (Operations Research) und Betriebswirtschaft.

Wirtschaftliche Fragestellungen treffen jeden von uns, sei es als Steuerzahler, Sparer oder Kreditnehmer. Für alle an Wirtschaftsprogrammen Interessierte sind hier zahlreiche nützliche und anwendungsbezogene Programme aus den Bereichen

- Zins- und Rendite-Berechnungen
- Renten- und Tilgungsrechnung
- Optimierungs- und Entscheidungstheorie
- Investitionsrechnung
- Abschreibungen

vorge stellt.

Die Programme sind in BASIC des Schneider CPC geschrieben.



Dem Verlag danke ich für die Herausgabe des Bandes und für die stets freundliche Zusammenarbeit.

Anzing, im Juni 1985



# INHALT

Seite

## KALENDERALGORITHMEN

1. Wochentagsbestimmung .....	9
2. Bewegliche Feiertage .....	13
3. Differenz zwischen zwei Terminen in Tagen .....	17
4. Kalenderdruck .....	21

## ZINSRECHNUNG

5. Zinsrechnung .....	31
6. Zinseszinsrechnung .....	37
7. Mittlerer Zins mehrerer Perioden .....	43
8. Effektivzins bei Kleinkrediten und Ratenzahlungen .....	47
9. Effektivzins bei Annuitätendarlehen .....	51
10. Rendite eines festverzinslichen Wertpapiers .....	55
11. Rendite eines Wertpapiers mit wechselndem Zins .....	57

## RENTENRECHNUNG

12. Endwert regelmäßiger Zahlungen .....	61
13. Barwert regelmäßiger Zahlungen .....	65
14. Umwandlung eines Kapitals in Rente .....	69
15. Ewige Rente .....	73

## TILGUNG

16. Ratentilgung .....	77
17. Annuitätentilgung .....	81

## ABSCHREIBUNG

18. Wertminderung/steigerung eines Objekts .....	85
19. Lineare Abschreibung .....	89
20. Degressive Abschreibung .....	93
21. Gemischt lineare und degressive Abschreibung .....	97
22. Digitale Abschreibung .....	101

	Seite
<b>INVESTITIONSRECHNUNG</b>	
23. Interner Zinsfluß .....	105
24. Kapitalwertmethode .....	109
25. Annuitätenmethode .....	113
<b>ENTSCHEIDUNG BEI UNSICHERHEIT</b>	
26. Entscheidung bei mehreren Zielen .....	117
27. Entscheidung bei gegebenen Wahrscheinlichkeiten .....	123
<b>LAGERHALTUNG</b>	
28. Lagerhaltung ohne Fehlmengen .....	131
29. Lagerhaltung mit Fehlmengen .....	135
30. Optimale Lagerhaltung .....	139
<b>WARTESCHLANGEN</b>	
31. Warteschlange bei Einmannbedienung .....	145
32. Warteschlange bei Mehrfachbedienung .....	149
<b>OPTIMIERUNG</b>	
33. Lineare Optimierung .....	153
34. Rentabilitätsgrenze .....	163
35. Optimale Zuordnung .....	169
<b>ZEITREIHENANALYSE</b>	
36. Lineare Regression .....	175
37. Gleitende Durchschnitte .....	181
38. Monats-Indizes .....	187
39. Exponentielles Glätten von Daten .....	195
40. Prognose durch exponentielles Glätten .....	201
<b>ANHANG</b>	
Statische Lebenserwartung .....	207
Formelsammlung .....	209
Literaturverzeichnis .....	213

# 1. WOCHENTAGSBESTIMMUNG

Bei Zinsrechnungen, Zahlungsfristen usw. ist es wichtig zu wissen, auf welchen Wochentag ein bestimmtes Datum fällt. Zur Wochentagsbestimmung gibt es mehrere Verfahren. Bekannt ist das Verfahren von Zeller (siehe z.B. [6]) oder die Ermittlung über das Julianische Datum, wie es z.B. in der Raumfahrt und der Astronomie geschieht.

## Zum folgenden Programm

Für ein beliebiges Datum zwischen 1901 – 2099 wird der Wochentag durch Zählung der Tage ab dem 1.1.1901 bestimmt. Für das Jahr J gibt

$$(J-1901) \cdot 1461/4$$

die Anzahl der Tage seit dem 1.1.1901 bis zum 31.12. des Vorjahrs. Dazu muß noch die Nummer des Tages im laufenden Jahr gezählt werden.

Die Monattersten eines Nichtschaltjahres haben die Nummern

$$0, 31, 59, 90, 120, 151, 181, 212, 243, 273, 304 \text{ und } 334.$$

Diese Zahlenfolge kann durch den Term

$$\text{INT}((158 \cdot M - 157/5) + (M+1) \cdot (M > 2))$$

beschrieben werden; dabei hat der Boolesche Term  $M > 2$  in CBM-BASIC den Wert

$$(M > 2) = \begin{cases} 1 & \text{für } M > 2 \\ 0 & \text{für } M \leq 2 \end{cases}$$

Bei Schaltjahren erhöht sich die Tageszahl in den Monaten März bis Dezember um eins; dies leistet der Term

$$(\text{INT}(A/4) = A/4) \cdot (M > 2).$$


Da der 1.1.1901 ein Dienstag war, kann durch Berechnung des Siebenerrestes der gesuchte Wochentag bestimmt werden. Es gilt die Codierung:

$$0 = \text{Sonntag}, 1 = \text{Montag}, 2 = \text{Dienstag usw.}$$

## Beispiel

Eingabe des Datums 1.1.2000 in der Form 01.01.2000 liefert den Wochentag Samstag.

# Die Wochentage und ihre namensgebenden Planeten



```

100 REM .....WOCHENTAGSBESTIMMUNG
110 :
120 CLS
130 PRINT STRING$(35,61)
140 PRINT TAB(8)" WOCHENTAGSBESTIMMUNG"
150 PRINT STRING$(35,61)
160 :
170 REM .....EINGABE
175 DIM W$(6)
180 FOR I=0 TO 6
190 READ W$(I)
200 NEXT I
210 PRINT:PRINT" DATUM IN DER FORM TT.MM.JJJJ EINGEBEN !"
220 PRINT:INPUT" DATUM";D$
230 T=VAL(MID$(D$,1,2))
240 M=VAL(MID$(D$,4,2))
250 J=VAL(MID$(D$,7,4))
260 IF T>31 OR M>12 OR J<1901 THEN PRINT CHR$(7);" EINGABEFehler!":GOTO 210
270 :
280 REM .....BERECHNUNG
290 GOSUB 420
300 D=D-INT(D/7)*7
310 :
320 REM .....AUSGABE
330 PRINT:PRINT STRING$(35,45)
340 PRINT" DER ";MID$(STR$(T),2,2)". "MID$(STR$(M),2,2);
350 PRINT". "MID$(STR$(J),2,4);

```

```

360 PRINT" IST EIN ";W$(D);"."
370 PRINT STRING$(35,45)
380 PRINT:INPUT" WEITER J/N";A$
390 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN 210
400 END
410 :
420 REM .....TAGESNUMMER AB 1.1.1901
430 D=INT((J-1901)*1461/4)+1+T+INT((158*M-157)/5)
440 D=D+(M>2)*(M-(INT(J/4)<J/4))
450 RETURN
460 :
470 DATA SONNTAG,MONTAG,DIENSTAG,MITTWOCH
480 DATA DONNERSTAG,FREITAG,SAMSTAG

```

```

=====
          WOCHENTAGSBESTIMMUNG
=====

```

DATUM IN DER FORM TT.MM.JJJJ EINGEBEN !

DATUM? 01.01.2000

```

-----
DER 1.1.2000 IST EIN SAMSTAG.
-----

```

## 2. BEWEGLICHE FEIERTAGE

Zur Terminplanung und Bewertung von Umsatzstatistiken ist die Bestimmung der beweglichen Feiertage von Bedeutung.

Seit dem Konzil von Nicäa (325 nach Chr.) ist das Osterfest auf den ersten Sonntag nach dem Vollmond festgelegt, der dem Frühlingsanfang (Frühlings-Tagundnachtgleiche) folgt. Da das Sonnenjahr kein Vielfaches der Mondperiode von 29,5 Tagen ist, verschiebt sich somit der jährliche Ostertermin.

Verfahren zur Bestimmung des Osterdatums stammen von dem Astronomen Aloysius Lilius und dem Jesuiten Christopher Clavius und später von dem Mathematiker Carl Friedrich Gauß. Die Gaußsche Osterformel ist in [6] dargestellt.

### Zum folgenden Programm

Das Programm berechnet den Termin des Osterfestes für die Jahre 1901 bis 2099 nach der Gaußschen Osterformel.

Bei Kenntnis des Osterdatums lassen sich auch die übrigen beweglichen Feiertage berechnen: Es liegt

Rosenmontag	48 Tage vor	} Ostern
Christi Himmelfahrt	39 Tage nach	
Pfingstsonntag	49 Tage nach	
Fronleichnam	60 Tage nach	

Mit Hilfe der in Programm 1 verwendeten Tageszählung können die Zeitpunkte der genannten Feiertage bestimmt werden.

### Beispiel

Für das Jahr 1984 liefert das Programm  
Rosenmontag 5. März  
Ostersonntag 22. April  
Christi Himmelfahrt 31. Mai  
Pfingstsonntag 10. Juni  
Fronleichnam 21. Juni

```

100 REM .....BEWEGLICHE FEIERTAGE
110 :
115 DIM M$(6)
120 CLS
130 PRINT STRING$(39,61)
140 PRINT"  BERECHNUNG VON BEWEGLICHEN FEIERTAGEN"
150 PRINT STRING$(39,61)
160 :
170 PRINT:INPUT" WELCHES JAHR";J
180 IF J<1901 OR J>2099 THEN PRINT"ANDERES JAHR EINGEBEN":GOTO 170
190 PRINT:PRINT STRING$(39,45)
200 PRINT" FEIERTAGE IM JAHR";J;"SIND"
210 PRINT STRING$(39,45)
220 :
230 FOR I=2 TO 6
240 READ M$(I)
250 NEXT I
260 :
270 REM .....BERECHNUNGEN
280 A=J MOD 19
290 B=J MOD 4
300 C=J MOD 7
310 D=(19*A+24) MOD 30
320 E=(2*B+4*C+6*D+5) MOD 7
330 F=D+E-9
340 N=F+90-(INT(J/4)=J/4)
350 :

```

```

360 REM .....AUSGABE
370 PRINT" ROSENMONTAG           "; M=N-48
380 GOSUB 530
390 PRINT" OSTERSONNTAG         "; M=N
400 GOSUB 530
410 PRINT" CHRISTI HIMMELFAHRT  "; M=N+39
420 GOSUB 530
430 PRINT" PFINGSTSONNTAG      "; M=N+49
440 GOSUB 530
450 PRINT" FRONLEICHNAM        "; M=N+60
460 GOSUB 530
470 PRINT STRING$(40,45)
480 RESTORE
490 PRINT:INPUT"WEITER (J/N)";A$
500 IF A$="J" OR A$="j" THEN GOTO 130
510 END
520 :
530 REM .....DATUMSBERECHNUNG
540 FOR I=6 TO 2 STEP -1
550 T=INT((158*I-157)/5)+(I>2)*(1-(INT(J/4)<J/4))
560 IF M>T THEN PRINT USING "## . ö        ö ";M-T,M$(I):RETURN
570 NEXT I
580 :
590 DATA FEBRUAR,MAERZ,APRIL,MAI,JUNI

```

=====  
BERECHNUNG VON BEWEGLICHEN FEIERTAGEN  
=====

WELCHES JAHR? 1984

-----  
FEIERTAGE IM JAHR 1984 SIND  
-----

ROSENMONTAG	5 . MAERZ
OSTERSONNTAG	22 . APRIL
CHRISTI HIMMELFAHRT	31 . MAI
PFINGSTSONNTAG	10 . JUNI
FRONLEICHNAM	21 . JUNI

-----

=====  
BERECHNUNG VON BEWEGLICHEN FEIERTAGEN  
=====

WELCHES JAHR? 1985

-----  
FEIERTAGE IM JAHR 1985 SIND  
-----

ROSENMONTAG	18 . FEBRUAR
OSTERSONNTAG	7 . APRIL
CHRISTI HIMMELFAHRT	16 . MAI
PFINGSTSONNTAG	26 . MAI
FRONLEICHNAM	6 . JUNI

-----

### **3. DIFFERENZ ZWISCHEN 2 TERMINEN**

Für Zinsberechnungen, Wechseldiskontierung u.ä. benötigt man die Kenntnis der Anzahl von Tagen zwischen zwei Terminen.

Dabei ist zwischen privat- und handelsrechtlichen Vorgängen zu unterscheiden, da bei letzteren das Bankjahr mit 360 Tagen festgesetzt ist. Da hierbei alle Monate einheitlich zu je 30 Tagen gerechnet werden, ist das Bankjahr rechnerisch sehr einfach zu handhaben.

#### **Zum folgenden Programm**

Das folgende Programm liefert die Anzahl von Tagen zwischen zwei Terminen nach dem bürgerlichen Kalender. Es benützt das Verfahren der Tageszählung seit dem 1.1.1901, wie es in den Programmen 1 und 2 verwendet wurde.

#### **Beispiel**

Ein am 5.8.1942 Geborener ist am 31.12.2000 21333 Tage alt. Zu beachten ist, daß Tage und Monate zweistellig eingegeben werden müssen.

```
100 REM .....TAGESDIFERENZ ZWEIER TERMINE
110 :
120 CLS
130 PRINT STRING$(35,61)
140 PRINT" DIFFERENZ ZWISCHEN ZWEI TERMINEN "
150 PRINT STRING$(35,61):PRINT
160 :
170 REM .....EINGABE
180 DIM D(2)
190 PRINT"TERMINE IN DER FORM TT.MM.JJJJ EINGEBEN":PRINT
200 :
210 REM .....BERECHNUNG
220 FOR I=1 TO 2
230 PRINT I;".TERMIN";:INPUT D$
240 T=VAL(MID$(D$,1,2))
250 M=VAL(MID$(D$,4,2))
260 J=VAL(MID$(D$,7,4))
270 IF T>31 OR M >12 THEN PRINT CHR$(7);" EINGABEFehler!":GOTO 230
280 IF J<1901 THEN PRINT CHR$(7);" JAHR >1901 ":GOTO 230
290 GOSUB 390
300 D(I)=D
310 NEXT I
320 D=ABS(D(1)-D(2)):PRINT
330 REM .....AUSGABE
340 PRINT STRING$(35,45)
350 PRINT"Die Differenz betraegt =" ;D;"Tage"
```

```
360 PRINT STRING$(35,45)
370 END
380 :
390 REM .....TAGESNUMMER SEIT DEM 1.1.1901
400 D=INT((J-1901)*1461/4)+1+T+INT((158*M-157)/5)
410 D=D+(M>2)*(M-(INT(J/4)<J/4))
420 RETURN
```

19

```
=====
DIFFERENZ ZWISCHEN ZWEI TERMINEN
=====
```

TERMINE IN DER FORM TT.MM.JJJJ EINGEBEN

```
1 .TERMIN? 31.12.2000
2 .TERMIN? 05.08.1942
```

```
-----
Die Differenz betraegt = 21333 Tage
-----
```



## 4. KALENDERDRUCK

### Zum folgenden Programm

Das Programm liefert entweder einen vollständigen Jahresüberblick nach Wochentagen am Drucker oder eine Monatsübersicht am Bildschirm. Als Muster dienen die Programmausdrücke auf den folgenden Seiten.

Weltkalender mit festen Wochentagen (1937 vom Völkerbund vorgeschlagen, 1954 von der UNO befürwortet)

	Januar April Juli Oktober	Februar Mai August November	März Juni September Dezember
Sonntag	1 8 15 22 29	5 12 19 26	3 10 17 24
Montag	2 9 16 23 30	6 13 20 27	4 11 18 25
Dienstag	3 10 17 24 31	7 14 21 28	5 12 19 26
Mittwoch	4 11 18 25	1 8 15 22 29	6 13 20 27
Donnerstag	5 12 19 26	2 9 16 23 30	7 14 21 28
Freitag	6 13 20 27	3 10 17 24	1 8 15 22 29
Samstag	7 14 21 28	4 11 18 25	2 9 16 23 30
			W

W = Weltfeiertag

```

100 REM .....KALENDER
110 :
120 CLS
130 PRINT STRING$(37,61)
140 PRINT TAB(15)" Kalender "
150 PRINT STRING$(37,61)
160 :
170 REM .....EINGABE
180 DIM T(3),MT(12),B$(2),M$(12)
190 FOR I=1 TO 12
200   READ M$(I)
210 NEXT I
220 FOR I=1 TO 12
230   READ MT(I)
240 NEXT I
250 :
260 PRINT CHR$(150);STRING$(37,154);CHR$(156)
270 PRINT CHR$(149);SPACE$(37);CHR$(149)
280 PRINT CHR$(149);" Monatskalender am Bildschirm ";CHR$(246);" 1  ";CHR$(149)
)
290 PRINT CHR$(149);SPACE$(37);CHR$(149)
300 PRINT CHR$(149);"   Jahreskalender am Drucker ";CHR$(246);" 2  ";CHR$(149)
)
310 PRINT CHR$(149);SPACE$(37);CHR$(149)
320 PRINT CHR$(147);STRING$(37,154);CHR$(153)
330 PRINT:PRINT:INPUT A
340 IF A<>1 AND A<>2 THEN PRINT CHR$(7):GOTO 330
350 INPUT " Welches Jahr ";JHR

```

```

360 IF JHR<1901 OR JHR>2099 THEN PRINT CHR$(7):GOTO 350
370 IF JHR MOD 4 = 0 THEN MT(2)=29
380 ON A GOTO 400,880
390 :
400 REM .....Monatskalender
410 B$(1)=" * * * * * "
420 INPUT" Welcher Monat 1..12";M
430 IF M<>INT(M) OR ABS(M-6.5)>5.5 THEN PRINT CHR$(7):GOTO 420
440 GOSUB 1520
450 CLS
460 PRINT" *****";M$(M);"*****"
470 PRINT B$(1)
480 PRINT " * Mo Di Mi Do Fr Sa So *"
490 PRINT B$(1)
500 PRINT" *****"
510 PRINT B$(1)
520 PRINT" * ";SPC(5*WT-5);
530 T=1
540 IF WT=7 THEN 590
550 FOR I=WT TO 6
560 PRINT USING "##";T;
570 T=T+1:PRINT" ";
580 NEXT I
590 PRINT USING "##";T;
600 PRINT" *"
610 T=T+1

```

```

620 PRINT B$(1)
630 FOR J=1 TO 5
640 PRINT " * ";
650 FOR I=1 TO 6
660     PRINT USING "##";T;
670     PRINT "    ";
680     T=T+1
690     IF T>MT(M) THEN K=I:I=6
700 NEXT I
710 IF T>MT(M) THEN J=5:GOTO 770
720 PRINT USING "##";T;
730 PRINT " *"
740 T=T+1
750 IF T>MT(M) THEN J=5:H=1:GOTO 770
760 PRINT B$(1)
770 NEXT J
780 IF H=1 THEN 800
790 PRINT SPC(33-5*K); "*"
800 PRINT B$(1)
810 PRINT " *****";JHR;" *****"
820 H=0:MT(2)=28
830 PRINT:INPUT "Weiter J/N";A$
840 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN 350
850 END
860 :
870 REM .....Jahreskalender

```

```
880 J#=STR$(JHR)
890 PRINT " Drucker einschalten und Taste druecken"
900 E#=INKEY#:IF E#="" THEN 900
910 PRINT #8,SPC(33);MID$(J#,2,1);SPC(4);MID$(J#,3,1);
920 PRINT #8,SPC(4);MID$(J#,4,1);SPC(4);MID$(J#,5,1)
930 PRINT #8
940 B$(1)="Mo Di Mi Do Fr Sa So"
950 B$(2)="*****"
960 FOR K=0 TO 3
970     PRINT #8:PRINT #8
980     FOR L=1 TO 2
990         PRINT #8,SPC(3);
1000         PRINT #8,"****";M$(L+3*K);"****";SPC(5);
1010     NEXT L
1020     PRINT #8,SPC(3);"****";M$(3+3*K);"****"
1030     FOR I=1 TO 2
1040         FOR L=1 TO 2
1050             PRINT #8,SPC(3);
1060             PRINT #8,B$(I);SPC(5);
1070         NEXT L
1080     PRINT #8,SPC(3);
1090     PRINT #8,B$(I)
1100 NEXT I
1110 PRINT #8,SPC(3);
1120 FOR I=1 TO 3
1130     M=I+3*K
```

```

1140     GOSUB 1520
1150     PRINT #8,SPC(3*WT-3);
1160     T=1
1170     IF WT=7 THEN 1230
1180     FOR J=WT TO 6
1190         PRINT #8,USING "##";T;
1200         PRINT #8,SPC(1);
1210         T=T+1
1220     NEXT J
1230     PRINT #8,USING "##";T;
1240     T(I)=T
1250     IF I=3 THEN 1270
1260     PRINT #8,SPC(8);
1270     NEXT I
1280     PRINT #8
1290     FOR H=1 TO 5
1300         PRINT #8,SPC(3);
1310         FOR I=1 TO 3
1320             T=T(I)+1
1330             IF T>MT(I+K*3) THEN PRINT #8,SPC(20);:GOTO 1430
1340             FOR L=1 TO 6
1350                 PRINT #8,USING "##";T;
1360                 PRINT #8,SPC(1);
1370                 T=T+1
1380                 IF T>MT(I+K*3) THEN PRINT #8,SPC(20-3*L);:L=6
1390             NEXT L

```

```

1400      IF T>MT(I+K*3) THEN 1420
1410      PRINT #8,USING "##";T;
1420      T(I)=T
1430      IF I=3 THEN 1450
1440      PRINT #8,SPC(8);
1450  NEXT I
1460  PRINT #8
1470 NEXT H
1480 NEXT K
1490 FOR K=1 TO 5:PRINT #8:NEXT K
1500 END
1510 :
1520 REM .....Berechnung der Monatsersten
1530 D=INT((JHR-1901)*1461/4)+INT((158*M-157)/5)+(M+1)*(M>2)+1
1540 D=D+(INT(JHR/4)=JHR/4)*(M>2)
1550 WT=1+D-INT(D/7)*7
1560 RETURN
1570 :
1580 DATA ** Januar **,** Februar *,*** Maerz **,*** April **
1590 DATA **** Mai ***,*** Juni ***,*** Juli ***,** August **
1600 DATA * September*,** Oktober *,* November *,* Dezember *
1610 DATA 31,28,31,30,31,30,31,31,30,31,30,31

```

```

***** Mai *****
*
* Mo    Di    Mi    Do    Fr    Sa    So *
*
*****
*
*          1    2    3    4    5 *
*
*  6    7    8    9   10   11   12 *
*
* 13   14   15   16   17   18   19 *
*
* 20   21   22   23   24   25   26 *
*
* 27   28   29   30   31           *
*
***** 1985 *****

```

```

***** Juni *****
*
* Mo    Di    Mi    Do    Fr    Sa    So *
*
*****
*
*                1    2 *
*
*  3    4    5    6    7    8    9 *
*
* 10   11   12   13   14   15   16 *
*
* 17   18   19   20   21   22   23 *
*
* 24   25   26   27   28   29   30 *
*
***** 1985 *****

```

\*\*\*\*\* Januar \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6  
 7 8 9 10 11 12 13  
 14 15 16 17 18 19 20  
 21 22 23 24 25 26 27  
 28 29 30 31

\*\*\*\*\* Februar \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3  
 4 5 6 7 8 9 10  
 11 12 13 14 15 16 17  
 18 19 20 21 22 23 24  
 25 26 27 28

\*\*\*\*\* Maerz \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3  
 4 5 6 7 8 9 10  
 11 12 13 14 15 16 17  
 18 19 20 21 22 23 24  
 25 26 27 28 29 30 31

\*\*\*\*\* April \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6 7  
 8 9 10 11 12 13 14  
 15 16 17 18 19 20 21  
 22 23 24 25 26 27 28  
 29 30

\*\*\*\*\* Mai \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5  
 6 7 8 9 10 11 12  
 13 14 15 16 17 18 19  
 20 21 22 23 24 25 26  
 27 28 29 30 31

\*\*\*\*\* Juni \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2  
 3 4 5 6 7 8 9  
 10 11 12 13 14 15 16  
 17 18 19 20 21 22 23  
 24 25 26 27 28 29 30

\*\*\*\*\* Juli \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6 7  
 8 9 10 11 12 13 14  
 15 16 17 18 19 20 21  
 22 23 24 25 26 27 28  
 29 30 31

\*\*\*\*\* August \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4  
 5 6 7 8 9 10 11  
 12 13 14 15 16 17 18  
 19 20 21 22 23 24 25  
 26 27 28 29 30 31

\*\*\*\*\* September\*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1  
 2 3 4 5 6 7 8  
 9 10 11 12 13 14 15  
 16 17 18 19 20 21 22  
 23 24 25 26 27 28 29  
 30

\*\*\*\*\* Oktober \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6  
 7 8 9 10 11 12 13  
 14 15 16 17 18 19 20  
 21 22 23 24 25 26 27  
 28 29 30 31

\*\*\*\*\* November \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3  
 4 5 6 7 8 9 10  
 11 12 13 14 15 16 17  
 18 19 20 21 22 23 24  
 25 26 27 28 29 30

\*\*\*\*\* Dezember \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1  
 2 3 4 5 6 7 8  
 9 10 11 12 13 14 15  
 16 17 18 19 20 21 22  
 23 24 25 26 27 28 29  
 30 31

\*\*\*\*\* Januar \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2  
 3 4 5 6 7 8 9  
 10 11 12 13 14 15 16  
 17 18 19 20 21 22 23  
 24 25 26 27 28 29 30  
 31

\*\*\*\*\* Februar \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6  
 7 8 9 10 11 12 13  
 14 15 16 17 18 19 20  
 21 22 23 24 25 26 27  
 28 29

\*\*\*\*\* Maerz \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5  
 6 7 8 9 10 11 12  
 13 14 15 16 17 18 19  
 20 21 22 23 24 25 26  
 27 28 29 30 31

\*\*\*\*\* April \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2  
 3 4 5 6 7 8 9  
 10 11 12 13 14 15 16  
 17 18 19 20 21 22 23  
 24 25 26 27 28 29 30

\*\*\*\*\* Mai \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6 7  
 8 9 10 11 12 13 14  
 15 16 17 18 19 20 21  
 22 23 24 25 26 27 28  
 29 30 31

\*\*\*\*\* Juni \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4  
 5 6 7 8 9 10 11  
 12 13 14 15 16 17 18  
 19 20 21 22 23 24 25  
 26 27 28 29 30

\*\*\*\*\* Juli \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2  
 3 4 5 6 7 8 9  
 10 11 12 13 14 15 16  
 17 18 19 20 21 22 23  
 24 25 26 27 28 29 30  
 31

\*\*\*\*\* August \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5 6  
 7 8 9 10 11 12 13  
 14 15 16 17 18 19 20  
 21 22 23 24 25 26 27  
 28 29 30 31

\*\*\*\*\* September\*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3  
 4 5 6 7 8 9 10  
 11 12 13 14 15 16 17  
 18 19 20 21 22 23 24  
 25 26 27 28 29 30

\*\*\*\*\* Oktober \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1  
 2 3 4 5 6 7 8  
 9 10 11 12 13 14 15  
 16 17 18 19 20 21 22  
 23 24 25 26 27 28 29  
 30 31

\*\*\*\*\* November \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3 4 5  
 6 7 8 9 10 11 12  
 13 14 15 16 17 18 19  
 20 21 22 23 24 25 26  
 27 28 29 30

\*\*\*\*\* Dezember \*\*\*\*\*  
 Mo Di Mi Do Fr Sa So  
 \*\*\*\*\*  
 1 2 3  
 4 5 6 7 8 9 10  
 11 12 13 14 15 16 17  
 18 19 20 21 22 23 24  
 25 26 27 28 29 30 31

# 5. ZINSRECHNUNG

Der Zinsrechnung liegt die bekannte Zinsformel

$$z = \frac{K \cdot p \cdot t}{360 \cdot 100}$$

zugrunde, dabei ist

- z der Zinsertrag in DM
- K das Kapital in DM
- p der jährliche Zinssatz in %
- t die Zeit in Tagen.

Jede der 4 Größen kann mit der Zinsformel aus den übrigen berechnet werden.

## Die Frühgeschichte des Geldes



Silbertetradrachme von Athen (um 460 v. Chr.)  
Vorderseite:  
Kopf der Göttin Athene



Rückseite: Eule



Griechische Münze mit dem Kopf des Poliarketes von Makedonien (306 - 283 v. Chr.)



Solidus aus Gold mit dem Kopf Konstantins des Großen (um 280 - 337 n. Chr.)



Chinesische Köschmünze (500 v. Chr. - 1912)



Münzer Pfennig (um 780) - Die Buchstaben CAROLVS sind das Signet Karls des Großen (742 - 814)



Hohlpfennig Brokreat, Heinrichs des Löwen (1129 - 1195) 2. Hälfte des 12. Jahrhunderts



Maria Theresianer Zahlungsmittel bis in die Gegenwart (z. B. in Afrika)



### **Zum folgenden Programm**

Nach Eingabe von 1 bis 4 werden folgende Größen berechnet:

- 1 Zinsertrag
- 2 Zeit in Tagen
- 3 Kapital
- 4 Zinssatz

Die jeweils noch fehlenden Größen werden abgefragt.

### **Beispiel**

Ein Kapital von 4000 DM wird 144 Tage auf einem Sparkonto zu 4,5% verzinst. Das Programm liefert den Zinsertrag: 72 DM.

```
100 REM .....ZINSRECHNUNG
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT"      Zinsrechnung"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT CHR$(154);STRING$(25,196);CHR$(156)
190 PRINT CHR$(149);SPACE$(25);CHR$(149)
200 PRINT CHR$(149);" Zinsertrag gesucht ";CHR$(246);" 1  ";CHR$(149)
210 PRINT CHR$(149);SPACE$(25);CHR$(149)
220 PRINT CHR$(149);" Laufzeit gesucht ";CHR$(246);" 2  ";CHR$(149)
230 PRINT CHR$(149);SPACE$(25);CHR$(149)
240 PRINT CHR$(149);" Kapital gesucht ";CHR$(246);" 3  ";CHR$(149)
250 PRINT CHR$(149);SPACE$(25);CHR$(149)
260 PRINT CHR$(149);" Zinssatz gesucht ";CHR$(246);" 4  ";CHR$(149)
270 PRINT CHR$(149);SPACE$(25);CHR$(149)
280 PRINT CHR$(147);STRING$(25,154);CHR$(153)
290 INPUT A
300 IF A=1 OR A=2 OR A=3 OR A=4 THEN 320 ELSE 310
310 PRINT CHR$(7);" 1 bis 4 eingeben !":GOTO 290
320 ON A GOTO 340,440,540,640
330 :
340 REM .....Zinsertrag
350 PRINT:INPUT "Kapital in DM";K
```

```
360 PRINT:INPUT "Jahreszinssatz in %";P
370 PRINT:INPUT "Laufzeit in Tagen";T
380 Z=K*P*T/36000
390 PRINT:PRINT STRING$(30,45)
400 PRINT USING "Zinsertrag = #####.## DM ";Z
410 PRINT STRING$(30,45)
420 END
430 :
440 REM .....Laufzeit
450 PRINT:INPUT "Kapital in DM";K
460 PRINT:INPUT "Jahreszinssatz in %";P
470 PRINT:INPUT "Zinsertrag in DM";Z
480 T=Z*36000 /K/P
490 PRINT:PRINT STRING$(30,45)
500 PRINT USING "Laufzeit = ##### Tage ";T
510 PRINT STRING$(30,45)
520 END
530 :
540 REM .....Kapital
550 PRINT:INPUT "Laufzeit in Tagen";T
560 PRINT:INPUT "Jahreszinssatz in %";P
570 PRINT:INPUT "Zinsertrag in DM";Z
580 K=Z*36000 /P/T
590 PRINT:PRINT STRING$(30,45)
600 PRINT USING "Kapital = #####.## DM";K
610 PRINT STRING$(30,45)
```

```
620 END
630 :
640 REM .....Zinssatz
650 PRINT:INPUT "Kapital in DM";K
660 PRINT:INPUT "Laufzeit in Tagen";T
670 PRINT:INPUT "Zinsertrag in DM";Z
680 P=Z*36000 /K/T
690 PRINT:PRINT STRING$(30,45)
700 PRINT USING "Jaehr1.Zinssatz = ###.## %";P
710 PRINT STRING$(30,45)
720 END
```

=====

Zinsrechnung

=====

Zinsertrag gesucht	1
Laufzeit-gesucht	2
Kapital gesucht	3
Zinssatz gesucht	4

? 1

Kapital in DM? 4000

Jahreszinssatz in %? 4.5

Laufzeit in Tagen? 144

-----

Zinsertrag =        72.00 DM

-----

## 6. ZINSESZINSRECHNUNG

Werden die erhaltenen Zinsen wieder dem Kapital zugeschlagen und erneut verzinst, so spricht man von Zinseszinsrechnung.

Ist  $K_0$  das Anfangskapital, so erhält man nach  $n$  Jahren bei  $p$  % Zins das Endkapital

$$K_n = K_0 \left(1 + \frac{p}{100}\right)^n.$$

Diese Zinseszinsformel wird auch zur Berechnung von Zuwachs- und Wertsteigerungsraten benützt.

Eine Einmalprämie von 50.000 DM für eine Lebensversicherung wächst in 20 Jahren und 4% Jahreszins auf

$$50.000 \text{ DM} \cdot 1,04^{20} = 109.556,15 \text{ DM}$$

an.


Bei unterjähriger Verzinsung, wie sie z.B. bei Festgeldkonten gegeben ist, gilt

$$K_n = K_0 \left(1 + \frac{p}{100i}\right)^{ni},$$

dabei ist  $i$  die Anzahl der jährlichen Verzinsungen. So liefern z.B. 10.000 DM in 3 Jahren bei 8% Jahreszins und vierteljährlicher Verzinsung

$$10.000 \text{ DM} \left(1 + \frac{8}{400}\right)^{12} = 12.682,42 \text{ DM}.$$

## Anwachsen einer DM bei 6% Jahreszins



### Zum folgenden Programm

Nach Eingabe von 1 bzw. 2 wird das Endkapital bzw. das Anfangskapital berechnet. Ist der Zinssatz gesucht, so kann er mittels Programm 18 berechnet werden.

### Beispiel

Was kostet ein abgezinster Sparbrief, der bei jährlich 8,5% Zins nach 5 Jahren 10.000 DM erbringt?

Das Programm liefert  $K_0 = 6650,45$  DM.

```
100 REM .....ZINSESZINSRECHNUNG
110 :
120 CLS
130 PRINT STRING$(25,61)
140 PRINT" ZINSESZINSRECHNUNG"
150 PRINT STRING$(25,61)
160 :
170 REM .....EINGABE
180 PRINT CHR$(150);STRING$(29,154);CHR$(156)
190 PRINT CHR$(149);SPACE$(29);CHR$(149)
200 PRINT CHR$(149);"      Endkapital gesucht ";CHR$(246);"  1 ";CHR$(149)
210 PRINT CHR$(149);SPACE$(29);CHR$(149)
220 PRINT CHR$(149);" Anfangskapital gesucht ";CHR$(246);"  2 ";CHR$(149)
230 PRINT CHR$(149);SPACE$(29);CHR$(149)
240 PRINT CHR$(147);STRING$(29,154);CHR$(153):PRINT
250 INPUT A
260 IF A<>1 AND A<>2 THEN PRINT CHR$(7);" 1 ODER 2 EINGEBEN!":GOTO 250
270 ON A GOTO 290,400
280 :
290 REM .....ENDKAPITAL GESUCHT
300 PRINT:INPUT" ANFANGSKAPITAL in DM";K
310 GOSUB 540
320 K2=K*P:K1=K
330 Z=K2-K1
340 PRINT:PRINT STRING$(30,45)
350 PRINT USING " ZINSERTRAG = #####.## DM";Z
```

```

360 PRINT USING " ENDKAPITAL = #####.## DM";K2
370 PRINT STRING$(30,45)
380 GOTO 500
390 :
400 REM .....ANFANGSKAPITAL GESUCHT
410 PRINT:INPUT " ENDKAPITAL";K
420 GOSUB 540
430 K1=K/P:K2=K
440 Z=K2-K1
450 PRINT:PRINT STRING$(30,45)
460 PRINT USING " ZINSERTRAG = #####.## DM";Z
470 PRINT USING " ANFANGSKAPITAL= #####.## DM";K1
480 PRINT STRING$(30,45)
490 :
500 PRINT:INPUT " Weitere Berechnungen (J/N)";A$
510 IF A$="J" OR A$="j" THEN 120
520 END
530 :
540 REM .....AUFZINSFAKTOR
550 PRINT:INPUT" LAUFZEIT IN ZINSPERIODEN";N
560 PRINT:INPUT" JAHRESZINSFUSS IN %";P
570 PRINT:INPUT" ANZAHL DER ZINSPERIODEN PRO JAHR";I
580 P=P/I
590 P=(1+P/100)^N
600 RETURN

```

=====  
ZINSESZINSRECHNUNG  
=====

Endkapital gesucht            1  
Anfangskapital gesucht       2

? 2

ENDKAPITAL? 10000

LAUFZEIT IN ZINSERIODEN? 5

JAHRESZINSFUSS IN %? 8.5

ANZAHL DER ZINSERIODEN PRO JAHR? :

-----  
ZINSERTRAG =    3349.55 DM  
ANFANGSKAPITAL=       6650.45 DM  
-----



## 7. MITTLERER ZINSSATZ

Bei vielen Wertpapieren, Renditeobjekten und Wertsteigerungen ist der Zinssatz bzw. die Zuwachsrate während der Laufzeit nicht konstant. Man benützt daher eine durchschnittliche oder mittlere Zins- bzw. Zuwachsrate.

Sind  $p_1, p_2, \dots, p_n$  die jeweiligen Zinssätze, so wächst der Anfangswert  $K_0$  nach  $n$  Zinsperioden auf den Wert

$$K_n = K_0 \left(1 + \frac{p_1}{100}\right) \left(1 + \frac{p_2}{100}\right) \dots \left(1 + \frac{p_n}{100}\right)$$

Der mittlere Zinssatz bzw. die mittlere Zuwachsrate ist dann


$$p = \left(\sqrt[n]{\left(1 + \frac{p_1}{100}\right) \left(1 + \frac{p_2}{100}\right) \dots \left(1 + \frac{p_n}{100}\right)} - 1\right) \cdot 100\%$$

Herrscht 5 Jahre lang eine Inflationsrate von

5,2%, 5,4%, 6,0%, 6,2% und 5,9%

so ist die mittlere Inflationsrate

$$\left(\sqrt[5]{1,052 \cdot 1,054 \cdot 1,060 \cdot 1,062 \cdot 1,059} - 1\right) \cdot 100\% = 0,057 \cdot 100\% = 5,7\%$$



Wird eine Lohnsteigerung von 4,6% erst ab Juni ausbezahlt, so entspricht dies einer jährlichen Lohnsteigerung von

$$\left( \sqrt[12]{1,0467} - 1 \right) \cdot 100\% = (1,027 - 1) \cdot 100\% = 2,7\%.$$

Werden bei einem Wertpapier mit wechselndem Zins die Erträge jährlich ausbezahlt, so darf oben genannte Formel nicht angewandt werden, da sie auf der Zinseszinsformel beruht. In diesem Fall kann Programm 11 angewendet werden.

Sucht man bei unterjähriger Verzinsung denjenigen Zinssatz  $p_0$ , der denselben Zinsertrag wie der Jahreszinssatz  $p$  erbringt, so erhält man den sog. konformen Zinssatz

$$p_0 = \left( \sqrt[i]{1 + \frac{p}{100}} - 1 \right) \cdot 100\%,$$

dabei ist  $i$  die Anzahl der jährlichen Verzinsungen. Der konforme Zinssatz für monatliche Verzinsung ( $i=12$ ) und den Jahreszinsfuß 10% ist somit

$$p_0 = \left( \sqrt[12]{1,10} - 1 \right) \cdot 100\% = (1,00797414 - 1) \cdot 100\% = 0,80\%$$

### Zum folgenden Programm

Nach Eingabe der Anzahl der Zinsperioden und des jeweiligen Zinssatzes, berechnet das Programm den mittleren Zinssatz nach der oben angegebenen Formel.

### Beispiel

Ein Bundesschatzbrief vom Typ B wird bei 7 Jahren Laufzeit wie folgt verzinst (Stand Herbst 1982):

1. Jahr	5,75 %
2. Jahr	7,25 %
3. Jahr	7,50 %
4. Jahr	8,00 %
5. Jahr	8,25 %
6. Jahr	9,00 %
7. Jahr	9,00 %

Das Programm liefert die Rendite von 7,82 %.

```
100 REM .....MITTLERER ZINSSATZ
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT" MITTLERER ZINSSATZ"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 INPUT" Wieviele Zinsperioden";N
190 S=1:DIM P(N)
200 PRINT:PRINT" Eingabe in % fuer"
210 FOR I=1 TO N
220     PRINT" Zinssatz fuer Periode";I,:INPUT P(I)
230     S=S*(1+P(I)/100)
240 NEXT I
250 :
260 REM .....AUSGABE
270 S=S^(1/N)-1
280 PRINT:PRINT STRING$(30,45)
290 PRINT USING "Mittlerer Zinssatz=###.## %";S*100
300 PRINT STRING$(30,45)
310 END
```

=====

MITTLERER ZINSSATZ

=====

Wieviele Zinsperioden? 7

Eingabe in % fuer

Zinssatz fuer Periode 1	? 5.75
Zinssatz fuer Periode 2	? 7.25
Zinssatz fuer Periode 3	? 7.5
Zinssatz fuer Periode 4	? 8
Zinssatz fuer Periode 5	? 8.25
Zinssatz fuer Periode 6	? 9
Zinssatz fuer Periode 7	? 9

-----

Mittlerer Zinssatz= 7.82 %

-----

## 8. EFFEKTIVZINS VON KLEINKREDITEN UND RATENZAHLUNGEN

Eigentlich sollte der Begriff "Effektivzins" Kreditkonditionen von Banken und Warenhäusern vergleichbar machen. Da nun aber Banken wieder spezielle Kreditarten anbieten, sind die Angaben jedoch nicht immer vergleichbar.

Der Effektivzins ist nach Definition der Zinsfuß, zu dem der ausbezahlte Anteil des Kredits verzinst werden müßte, damit sein Endwert gleich dem Endwert der geleisteten Zins- und Tilgungszahlungen ist (vgl. Programm 12).

Seit Beginn des Jahres 1981 wird der Effektivzins über die Gleichung

$$\frac{1 + \frac{B}{100}}{N} + \frac{p}{100} = \frac{q^J}{\left(5,5 + \frac{1200}{e}\right) (q^J - 1) + \left(1 + \frac{M-1}{24} \cdot \frac{e}{100}\right) \cdot \frac{M}{1+Me/1200}}$$

ermittelt, dabei ist

- e = gesuchter Effektivzins in %
- q = 1 + e/100
- N = Laufzeit in Monaten
- p = nominaler Monatszinssatz in %
- J = Laufzeit in ganzen Jahren
- M = Anzahl der Restmonate
- B = Bearbeitungsgebühr in %.

Da diese Gleichung nicht nach e aufgelöst werden kann, muß die Berechnung indirekt erfolgen. Mit Hilfe einer Intervallschachtelung wird e solange systematisch variiert, bis beide Seiten der Gleichung übereinstimmen.

Bis zum Jahr 1980 wurde eine sehr viel einfachere Formel zur Ermittlung des Effektivzins benützt, die finanzmathematisch nicht sinnvoll war, da sie nicht auf Zinseszinsrechnung beruht:

$$e = \frac{24 (pN + B)}{N+1}$$

Für das im Programm berechnete Beispiel ergibt sich nach der alten Formel 17,48 %, gemäß der neueren jedoch 16,44 %.

## Zum folgenden Programm

Nach Eingabe von Kredithöhe, Laufzeit, Monatszinssatz und Bearbeitungsgebühr berechnet das Programm mittels einer Intervallschachtelung den Effektivzins nach der neuen Formel. Die Intervallschachtelung selbst wird in einem Unterprogramm ausgeführt. Die im Nenner der neuen Formel auftretenden Funktionen werden mittels DEF FN berechnet.

## Beispiel

Bei einem Großversandhaus wurden Möbel im Wert von 5000 DM auf Ratenzahlung gekauft. Die Laufzeit der Zahlungen beträgt 72 Monate, der monatliche Zins 0,69 % und die Bearbeitungsgebühr 3,5 %.

Das Programm liefert dafür eine Monatsrate von 106,37 DM.

Der Effektivzins beläuft sich auf 16,44 %.



“Bittsteller bei einem jüdischen Geldverleiher:  
Ich bitt eüch jud leicht (leiht) mir zu hand / was eüch  
gebürt (an Zinsen), gebt mir verstand (kund) / Bargelt  
auff bürgen oder pfand”  
Holzschnitt von Jörg Breu, Augsburg 1531

```
100 REM .....EFFEKTIVZINS BEI KLEINKREDITEN
110 :
120 CLS
130 PRINT STRING$(34,61)
140 PRINT" Effektivzins eines Kleinkredits"
150 PRINT STRING$(34,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Kredithoehe in DM";K
190 PRINT:INPUT" Laufzeit in Monaten";N
200 PRINT:INPUT" Monatszinssatz in %";P
210 PRINT:INPUT" Bearbeitungsgebuehr in %";B
220 :
230 REM .....Berechnungen
240 J=INT(N/12):M=N MOD 12
250 DEF FNA(X,M,J)=(5.5+12/X)*((1+X)^J-1)
260 DEF FNB(X,M,J)=(1+(M-1)/24*X)*M/(1+M*X/12)
270 DEF FNR(X,M,J)=(1+X)^J/(FNA(X,M,J)+FNB(X,M,J))
280 L=(1+B/100)/N+P/100
290 GOSUB 440
300 :
310 REM .....Monatsrate
320 Z=P*N*K/100:REM Zinsanteil
330 G=B*K/100:REM Gebuehrenanteil
340 RT=(K+Z+G)/N
350 :
```

```

360 REM .....Ausgabe
370 PRINT:PRINT STRING$(34,45)
380 PRINT USING " Monatsrate   = ####.## DM";RT
390 PRINT USING " Effektivzins = ##.## % ";100*
400 PRINT STRING$(34,45)
410 END
420 :
430 REM .....Intervallschachtelung
440 E=.1:D=1
450 R=FNR(E,M,J)
460   S=R-L
470   E=E-SGN(S)*D/2
480   D=D/2
490   IF D<.0000001 THEN 510
500 GOTO 450
510 RETURN

```

```

=====
Effektivzins eines Kleinkredits
=====

```

Kredithoehc in DM? 5000

Laufzeit in Monaten? 72

Monatszinzsatz in %? .69

Bearbeitungsgebuehr in %? 3.5

```

-----
Monatsrate   = 106.38 DM
Effektivzins = 16.44 %
-----

```

## 9. EFFEKTIVZINS BEI ANNUITÄTENDARLEHEN

Ebenso wie bei Ratenzahlung und Kleinkrediten werden Bankdarlehen und Hypotheken durch den Effektivzins beschrieben. Seine Berechnung erfolgt prinzipiell wie bei Kleinkrediten, nur werden die Zahlungen meist jährlich verzinst und die Bearbeitungsgebühr durch einen Auszahlungsverlust (= Disagio) ersetzt.

### **Zum folgenden Programm**

Nach Eingabe des nominellen Zinssatzes, des Tilgungssatzes, der prozentualen Auszahlung berechnet das Programm die Laufzeit und den Effektivzins. Die Zinsermittlung erfolgt analog zum vorhergehenden Programm über eine Intervallschachtelung.

### **Beispiel**

Ein Käufer einer Eigentumswohnung benötigt zur Finanzierung noch 96.000 DM. Er nimmt dazu bei einer Bank ein Hypothekendarlehen über 100.000 DM bei 96% Auszahlung, 9% Zins und 1% Tilgung auf. Das Programm berechnet die Laufzeit zu 26,7 Jahre. Der Effektivzins ist 9,49 %.

Die Tilgungsrate eines Annuitätendarlehens kann mit Hilfe von Programm 17 berechnet werden.

Literaturhinweis: [13]

```

100 REM .....EFFEKTIVZINS BEI ANNUITAETENDARLEHEN
110 :
120 CLS
130 PRINT STRING$(32,61)
140 PRINT" Effektivzins eines Darlehens"
150 PRINT STRING$(32,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Zinssatz in %";P
190 PRINT:INPUT" Tilgungssatz in %";I
200 PRINT:INPUT" Auszahlung in %";A
210 :
220 REM .....BERECHNUNGEN
230 N=LOG(1+P/I)/LOG(1+P/100)
240 J=N
250 M=N-INT(N):REM Restzeit
260 N=INT(N):REM Ganze Jahre
270 T=(P+I)/100:REM Tilgungsrate
280 A=A/100
290 :
300 REM .....INTERVALLSCHACHTELUNG
310 X0=0:X1=1
320 WHILE ABS(X0-X1)>.00001
330     X=(X0+X1)/2
340     S=T:REM Endwert der Tilgung
350     FOR I=1 TO N-1

```

```
360         S=S*(1+X)+T
370     NEXT I
380     Z=(1+X)^M:REM KONFORMER ZINS FUER RESTZEIT
390     S=S*Z
400     Y=T*((1+X)^M-1)/X:REM Resttilgung
410     S=S+Y
420     GOSUB 540
430     IF ABS(K-S)<.00001 THEN 500
440     IF S>K THEN X0=X ELSE X1=X
450 WEND
460 :
470 REM .....AUSGABE
480 PRINT:PRINT STRING$(32,45)
490 PRINT USING " Laufzeit = ###.## Jahre";J
500 PRINT USING " Effektivzins = ##.## % ";100*X
510 PRINT STRING$(32,45)
520 END
530 :
540 REM .....ENDWERT DES AUSGEZAHLTEN KAPITALS
550 K=A
560 K=K*(1+X)^N
570 Z=(1+X)^M-1
580 K=K*(1+Z)
590 RETURN
```

=====  
Effektivzins eines Darlehens  
=====

Zinssatz in %? 9

Tilgungssatz in %? 1

Auszahlung in %? 96

-----  
Laufzeit = 26.72 Jahre

Effektivzins = 9.49 %  
-----

## 10. RENDITE VON FESTVERZINSLICHEN WERTPAPIEREN

Festverzinsliche Wertpapiere wie Pfandbriefe, Kommunalobligationen, Anleihen usw. werden auch Rentenpapiere genannt.

Ist  $N$  der Nominalzins,  $R$  der Rückzahlungskurs,  $A$  der Ausgabe- bzw. der Anschaffungskurs und  $L$  die Laufzeit in Jahren, so berechnet sich die Rendite  $p$  aus der Formel

$$p = \frac{\left(N + \frac{R - A}{L}\right) \cdot 100\%}{A}$$

Da sich diese Rendite noch um die jeweiligen Bank- und Maklergebühren verringert, werden von Banken und Tageszeitungen manchmal etwas differierende Beträge genannt.

### Zum folgenden Programm

Nach Eingabe von Nominalzins, Rückzahlungs- und Anschaffungskurs und Laufzeit berechnet das Programm die Rendite nach obengenannter Formel.

### Beispiel

Ein Pfandbrief vom Nominalzins 7% und der Restlaufzeit von 6 Jahren wird bei Kurswert 96% gekauft. Die Rendite ist

$$\frac{\left(7 + \frac{100 - 96}{6}\right) \cdot 100\%}{96} = \frac{(7 + 0,67) \cdot 100\%}{96} = 7,99\%$$

```

100 REM ..RENDITE VON FESTVERZINS.WERTPAPIEREN
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT" RENDITE VON FESTVERZ. WERTPAPIEREN"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Nominalzins in %";N
190 PRINT:INPUT" Rueckzahlungskurs in %";R
200 PRINT:INPUT" Ausgabekurs in %";A
210 PRINT:INPUT" Laufzeit in Jahren";L
220 :
230 REM .....BERECHNUNG
240 P=(N+(R-A)/L)
250 P=P*100/A
260 PRINT STRING$(39,45)
270 PRINT TAB(8) USING " Rendite = ###.## %";P
280 PRINT STRING$(39,45)
290 END

```

```

=====
RENDITE VON FESTVERZ. WERTPAPIEREN
=====

```

Nominalzins in %? 7

Rueckzahlungskurs in %? 100

Ausgabekurs in %? 96

Laufzeit in Jahren? 6

```

-----
Rendite = 7.99 %
-----

```

## 11. RENDITE EINES WERTPAPIERS MIT WECHSELNDEM ZINS

Bei einigen Wertpapieren, wie z.B. Bundesschatzbriefen vom Typ A, werden die Zinserträge jährlich ausgezahlt. Für solche Wertpapiere berechnet das folgende Programm die Rendite; werden die Zinsen nicht ausbezahlt, so muß das Programm 7 zur Renditebestimmung gewählt werden.

Die Rendite  $r$  eines solchen Wertpapiers ist dadurch definiert, daß der Endwert des mit  $r$  verzinsten Kapitals gleich dem Endwert der nachschüssig verzinsten Zinserträge und dem Wert des Papiers selbst ist (vgl. Programm 12).

### Zum folgenden Programm

Ähnlich wie bei den Programmen 8 und 9 wird die gesuchte Rendite über eine Intervallschachtelung bestimmt. Die Verzinsung der ausgeschütteten Zinserträge wird bei jedem Schritt in einem Unterprogramm ermittelt.

### Beispiel

Ein Bundesschatzbrief vom Typ A hat bei 6 Jahren Laufzeit folgende Verzinsung (Stand Herbst 1982):

1. Jahr	5,75 %
2. Jahr	7,25 %
3. Jahr	7,50 %
4. Jahr	8,00 %
5. Jahr	8,25 %
6. Jahr	9,00 %

Die Zinsen werden jährlich ausgeschüttet.  
Das Programm liefert die Rendite 7,51 %.

Literaturhinweis: [13]

```

100 REM RENDITE E. WERTPAPIERS BEI WECHS. ZINS
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT" Rendite eines Wertpapiers"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Laufzeit in Jahren";N
190 DIM P(N)
200 PRINT:PRINT" Zins in %"
210 FOR I=1 TO N
220     PRINT" im Jahr";I;:INPUT P(I)
230 NEXT I
240 :
250 REM INTERVALLSCHACHTELUNG FUER EFFEKTIVZINS
260 X0=0:X1=100
270 X=(X0+X1)/2
280 GOSUB 400
290 R=(1+X/100)^N-1
300 IF ABS(Q-R)<.00001 THEN 350
310 IF Q>R THEN X0=X :GOTO 270
320 X1=X:GOTO 270
330 :
340 REM .....AUSGABE
350 PRINT:PRINT STRING$(30,45)

```

```
360 PRINT TAB(8) USING " Rendite = #####.## %";X
370 PRINT STRING$(30,45)
380 END
390 :
400 REM .....UNTERPROGRAMM FUER EFFEKTIVZINS
410 Q=0
420 FOR I=1 TO N
430   Q=Q+P(I)/100
440   Z=Q*X/100
450   IF I=N THEN 470
460   Q=Q+Z
470 NEXT I
480 RETURN
```

=====  
Rendite eines Wertpapiers  
=====

Laufzeit in Jahren? 6

Zins in %

im Jahr 1 ? 5.75

im Jahr 2 ? 7.25

im Jahr 3 ? 7.5

im Jahr 4 ? 8

im Jahr 5 ? 8.25

im Jahr 6 ? 9

-----  
Rendite =           7.51 %  
-----

## 12. ENDWERT REGELMÄSSIGER ZAHLUNGEN

Viele Zahlungen für Sparverträge, Versicherungen, Unterhaltszahlungen usw. werden in Form von regelmäßigen Raten geleistet.

Werden jeweils am Anfang von  $n$  Jahren jährlich  $R$  DM auf ein Konto eingezahlt und jährlich mit  $p$  % verzinst, so ist am Ende des letzten Jahres der Endwert der Zahlungen

$$E_n = Rq \frac{q^n - 1}{q - 1}$$

(Endwert einer vorschüssigen Rente). Dabei gilt

$$q = 1 + \frac{p}{100} \text{ (Aufzinsfaktor).}$$

Erfolgen die Zahlungen nachschüssig, d.h. am Ende des Jahres, so gilt

$$E_n = R \frac{q^n - 1}{q - 1}$$

(Endwert einer nachschüssigen Rente).

Werden die Zahlungen unterjährig verzinst, so ist theoretisch der Zinssatz  $p$  durch  $\frac{p}{i}$  und die Anzahl  $n$  der Zinsperioden durch  $ni$  zu ersetzen.

Jedoch berechnen Banken Rentenendwerte nach folgender Formel

$$E_n = R \left( \frac{i}{q-1} + \frac{i-1}{2} \right) (q^n - 1) \text{ nachschüssig}$$

bzw.

$$E_n = R \left( \frac{i}{q-1} + \frac{i+1}{2} \right) (q^n - 1) \text{ vorschüssig.}$$

Diese Bankformeln gelten bei unterjähriger Verzinsung; für  $i=1$  gehen sie näherungsweise in die vorher angegebenen Formeln über.

Für einen Sparvertrag werden beispielsweise 20 Jahre lang am Monatsersten 200 DM eingezahlt. Welchen Endwert haben die Zahlungen bei 4,5% Jahreszins?

Die Bankenformel liefert

$$\begin{aligned} E_{20} &= 200 \text{ DM} \left( \frac{12}{0,045} + 6,5 \right) (1,045^{20} - 1) \\ &= 77.126,64 \text{ DM} \end{aligned}$$

### Zum folgenden Programm

Nach Eingabe der Anzahl der jährlichen Zahlungen, des Zinssatzes, der Laufzeit und der Zahlungshöhe, berechnet das Programm den Endwert der Rente nach der Bankenformel.

### Beispiel

Für einen 10.000 DM-Kredit müssen 6 Jahre lang am Monatsende 213,— DM bezahlt werden. Welchen Endwert haben die Zahlungen, wenn sie mit 8% verzinst werden?

Das Programm liefert den Endwert 19.438,16 DM.



“Geldwechsler hinter seiner Wechselbank”  
Holzschnitt von Hans Weiditz, Augsburg 16. Jahrhundert

```
100 REM .....ENDWERT REGELMAESSIGER ZAHLUNGEN
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT"   ENDWERT REGELMAESSIGER ZAHLUNGEN"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Wieviele Zahlungen pro Jahr";I
190 PRINT:INPUT" Hoehe der Zahlungen";R
200 PRINT:INPUT" Zinssatz in %";P
210 PRINT:INPUT" Laufzeit in Jahren";N
220 PRINT:INPUT" Vorschuessig(J/N)";A#
230 :
240 REM .....BERECHNUNGEN
250 P=P/100
260 Q=(1+P)^N
270 E=R*(Q-1)
280 IF LEFT$(A#,1)="J" OR LEFT$(A#,1)="j" THEN 290 ELSE 300
290 E=E*(I/P+(I+1)/2):GOTO 330
300 E=E*(I/P+(I-1)/2)
310 :
320 REM .....AUSGABE
330 PRINT:PRINT STRING$(39,45)
340 PRINT USING " Endwert der Zahlungen ist #####.## DM";E
350 PRINT STRING$(39,45)
360 END
```

=====  
ENDWERT REGELMAESSIGER ZAHLUNGEN  
=====

Wieviele Zahlungen pro Jahr? 12

Hoehe der Zahlungen? 213

Zinssatz in %? 8

Laufzeit in Jahren? 6

Vorschuessig (J/N)? n

-----  
Endwert der Zahlungen ist 19438.16 DM  
-----

### 13. BARWERT REGELMÄSSIGER ZAHLUNGEN

Bei Versicherungs- und Unterhaltszahlungen werden regelmäßige Zahlungen oft durch eine einmalige Zahlung oder Prämie abgelöst.

Der Barwert solcher Zahlungen ist somit der Anfangswert eines Kapitals, dessen Umwandlung in Rente die entsprechenden Zahlungen liefert. Die Formel für den Barwert von Zahlungen erhält man, indem man die Rentenendformel abzinst; d.h. durch  $q^n$  dividiert:

$$B = Rq \frac{q^n - 1}{q^n(q-1)} \text{ vorschüssig}$$

bzw.

$$B = R \frac{q^n - 1}{q^n(q-1)} \text{ nachschüssig}$$

Die entsprechenden Bankformeln für unterjährige Verzinsung lauten

$$B = R \frac{\left(\frac{m}{q-1} + \frac{m+1}{2}\right)}{q^n} (q^n - 1) \text{ vorschüssig}$$

$$B = R \frac{\left(\frac{m}{q-1} + \frac{m-1}{2}\right)}{q^n} (q^n - 1) \text{ nachschüssig}$$

Durch welche Einmalprämie kann eine Versicherungsprämie über 200,— DM, die 20 Jahre lang am Monatsersten bezahlt und mit 4% verzinst wurde, abgegolten werden?

Der Barwert der Zahlungen ist 33.323,48 DM.

#### Zum folgenden Programm

Nach Eingabe der Anzahl der Zahlungen, des Zinssatzes, der Laufzeit, der Zahlungshöhe wird der Barwert der Zahlungen nach der Bankformel berechnet.

### **Beispiel**

Durch ein Testament wird ein Mann verpflichtet, seiner 50jährigen Schwester lebenslänglich eine Wohnung in dem testamentarisch vermachten Haus zu überlassen. Durch welche einmalige Zahlung kann das Wohnrecht abgelöst werden, wenn die Monatsmiete der Wohnung 400 DM beträgt?

Da die Schwester eine mittlere Lebenserwartung von 78 Jahren hat, entspricht dem Wohnrecht eine monatliche Rente von 400 DM mit einer Laufzeit von 28 Jahren. Der Barwert der Rente beträgt bei 4% Zins vorschüssig 81.715,66 DM.

Siehe Anhang B: Statistische Lebenserwartung.

Literaturhinweis: [1], [9]

```
100 REM .....BARWERT REGELMAESSIGER ZAHLUNGEN
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT "   Barwert regelmaessiger Zahlungen"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Wieviele Zahlungen pro Jahr";I
190 PRINT:INPUT" Hoehe der Zahlungen";R
200 PRINT:INPUT" Zinssatz in %";P
210 PRINT:INPUT" Laufzeit in Jahren";N
220 PRINT:INPUT" Vorschuessig(J/N)";A$
230 :
240 REM .....BERECHNUNGEN
250 P=P/100
260 Q=(1+P)^N
270 B=R*(Q-1)/Q
280 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN 290 ELSE 300
290 B=B*(I/P+(I+1)/2):GOTO 330
300 B=B*(I/P+(I-1)/2)
310 :
320 REM .....AUSGABE
330 PRINT:PRINT STRING$(39,45)
340 PRINT USING " Barwert der Zahlungen ist #####.## DM";B
350 PRINT STRING$(39,45)
360 END
```

=====  
Barwert regelmaessiger Zahlungen  
=====

Wieviele Zahlungen pro Jahr? 12

Hoehe der Zahlungen? 400

Zinssatz in %? 4

Laufzeit in Jahren? 28

Vorschuessig(J/N)? j

-----  
Barwert der Zahlungen ist 81715.66 DM  
-----

## 14. UMWANGLUNG EINES KAPITALS IN RENTE

Sollen aus einem Kapital  $K$  regelmäßige Zahlungen  $R$  geleistet werden, so wird das Kapital aufgezehrt, wenn die entnommenen Zahlungen die Zins-erträge übersteigen.

Die Höhe der Rente kann der Rentenbarwertformel entnommen werden (vgl. Programm 13):

$$R = K \frac{q^n (q-1)}{q^n - 1} \text{ nachschüssig}$$

bzw.

$$R = K \frac{q^{n-1} (q-1)}{q^n - 1} \text{ vorschüssig.}$$

Für unterjährige Verzinsung wird die Bankenformel angewandt:

$$R = K \frac{q^n}{\left(\frac{i}{q-1} + \frac{i-1}{2}\right) (q^n - 1)} \text{ nachschüssig}$$

$$R = K \frac{q^n}{\left(\frac{i}{q-1} + \frac{i+1}{2}\right) (q^n - 1)} \text{ vorschüssig}$$

Eine Abiturientin möchte ihre Aussteuerversicherung in Höhe von 25.000 DM zur Finanzierung ihres Studiums verwenden. Bei 4% Verzinsung und fünfjährigem Studium erhält sie monatlich vorschüssig

$$R = 25.000 \text{ DM} \cdot \frac{1,04^5}{\left(\frac{12}{0,04} + 6,5\right) (1,04^5 - 1)} = 458,05 \text{ DM}$$

### Zum folgenden Programm

Nach Eingabe des Kapitals, der Anzahl der jährlichen Auszahlungen, der Laufzeit und des Zinssatzes berechnet das Programm die sich ergebende Rente nach der Bankenformel.

### Beispiel

Eine 50jährige Witwe soll für den Verkauf eines Grundstücks im Wert von 240.000 DM eine Leibrente erhalten. Da sie eine Lebenserwartung von 78 Jahren hat, kann sie bei 4% eine monatliche Rente von 1174,81 DM erwarten.

Die hier angewandte Methode zur Berechnung von Leibrenten gilt nur näherungsweise. Die genaue Berechnung von Leibrenten stützt sich auf Sterbetafeln (siehe dazu z.B. [1]).



“Kaufmann inmitten seiner Familie”  
Holzschnitt von H. Schauer, Augsburg 1477

```

100 REM .....UMWANDLUNG EINES KAPITALS IN RENTE
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT"   Umwandlung eines Kapitals in Rente"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Wieviele Zahlungen pro Jahr";I
190 PRINT:INPUT" Hoehe des Kapitals";K
200 PRINT:INPUT" Zinssatz in %";P
210 PRINT:INPUT" Laufzeit in Jahren";N
220 PRINT:INPUT" Vorschuessig(J/N)";A$
230 :
240 REM .....BERECHNUNGEN
250 P=P/100
260 Q=(1+P)^N
270 R=K*Q/(Q-1)
280 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN 290 ELSE 300
290 R=R/(I/P+(I+1)/2):GOTO 330
300 R=R/(I/P+(I-1)/2)
310 :
320 REM .....AUSGABE
330 PRINT:PRINT STRING$(39,45)
340 PRINT USING " Hoehe der Rente #####.## DM";R
350 PRINT STRING$(39,45)
360 END

```

=====  
Umwandlung eines Kapitals in Rente  
=====

Wieviele Zahlungen pro Jahr? 12

Hoehe des Kapitals? 240000

Zinssatz in %? 4

Laufzeit in Jahren? 28

Vorschuessig(J/N)? j

-----  
Hoehe der Rente      1174.81 DM  
-----

## 15. EWIGE RENTE

Ist die Laufzeit einer Rente nicht beschränkt, so spricht man von einer ewigen Rente.

Bringt das Kapital  $B$ , jährlich zu  $p\%$  verzinst,  $R$  DM Zins, so kann dieser Zinsertrag als ewige Rente ausgezahlt werden, ohne das Kapital zu mindern.

Der Barwert einer ewigen Rente ist somit

$$B = \frac{R}{p/100} \text{ nachschüssig}$$

bzw.

$$B = \frac{Rq}{p/100} \text{ vorschüssig.}$$

Eine Stadt möchte jährlich einen Kulturpreis in Höhe von 10.000 DM aussetzen. Bei 5% Zins muß bei nachschüssiger Zahlung ein Stiftungsfond über

$$\frac{10000 \text{ DM}}{0,05} = 200.000 \text{ DM}$$

eingerrichtet werden.

### Zum folgenden Programm

Je nachdem, ob der Barwert oder die ewige Rente berechnet werden soll, ist 1 bzw. 2 einzugeben. Das Programm berechnet die gesuchte Größe nach den oben angegebenen Formeln.

### Beispiel

Eine Gemeinde möchte die Wasserrechte eines Landwirts, die einen jährlichen Wert von 500 DM entsprechen, durch eine einmalige Zahlung ablösen. Bei 4% Zins und vorschüssiger Zahlung haben die Wasserrechte den Barwert 13.000 DM.

Literaturhinweis: [1], [9]

```

100 REM .....EWIGE RENTE
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT"      Ewige Rente"
150 PRINT STRING$(30,61):PRINT
160 :
170 REM .....EINGABE
180 PRINT
190 PRINT
200 PRINT"  Barwert gesucht ";"      1  "
210 PRINT
220 PRINT"  Rente gesucht  ";"      2  "
230 PRINT
240 PRINT
250 INPUT A
260 IF A<>1 AND A<>2 THEN PRINT CHR$(7); "1 oder 2 eingeben":GOTO 250
270 ON A GOTO 290,460
280 :
290 REM .....BARWERT GESUCHT
300 PRINT:INPUT" Jahresrente in DM";R
310 PRINT:INPUT" Zinsfuss in % ";P
320 PRINT:INPUT" Vorschuessig(J/N)";A$
330 :
340 REM .....BERECHNUNG
350 P=P/100

```

```
360 Q=1+P
370 B=R/P
380 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN B=B*Q
390 :
400 REM .....AUSGABE
410 PRINT:PRINT STRING$(39,45)
420 PRINT USING " Der Barwert betraegt #####.## DM";B
430 PRINT STRING$(39,45)
440 END
450 :
460 REM .....RENTE GESUCHT
470 PRINT:INPUT" Vorhandenes Kapital";B
480 PRINT:INPUT" Zinsfuss in % ";P
490 PRINT:INPUT" Vorschuessig(J/N)";A$
500 :
510 REM .....BERECHNUNG
520 P=P/100
530 Q=1+P
540 R=B*P
550 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN R=R*Q
560 :
570 REM .....AUSGABE
580 PRINT:PRINT STRING$(39,45)
590 PRINT USING "Die ewig.Rente betr. jhr1. #####.## DM ";R
600 PRINT STRING$(39,45)
610 END
```

=====  
Ewige Rente  
=====

Barwert gesucht            1  
Rente gesucht                2

? 1

Jahresrente in DM? 500

Zinsfuss in % ? 4

Vorschuessig(J/N)? j

-----  
Der Barwert betraegt    13000.00 DM  
-----

## 16. RATENTILGUNG

Bei der Ratentilgung wird – im Gegensatz zur Annuitätentilgung – der Kreditbetrag in gleichbleibenden Raten getilgt.

Wird der Kreditbetrag  $K$  gleichmäßig in  $n$  Jahren getilgt, so beträgt eine Tilgungsrate

$$T = \frac{K}{n}.$$

Neben den Tilgungen  $T$  sind noch Zinszahlungen für die jeweils vorhandene Restschuld fällig. Da diese Restschuld gleichmäßig kleiner wird, vermindert sich ebenfalls der Zinsanteil bei den Rückzahlungen. Beträgt der Zins  $p$  %, so ist am Ende des  $k$ -ten Jahres die Rate

$$R = \frac{Kp}{100n} (n-k+1)$$

zu zahlen.

Die Belastung sinkt jährlich um einen festen Betrag (abnehmende arithmetische Folge). Werden jährlich z.B. 8000 DM getilgt bei 12 % Zins, so verringert sich die Belastung jährlich um

$$8000 \text{ DM} \cdot 0,12 = 960 \text{ DM}.$$

### Zum folgenden Programm

Nach Eingabe der Kredithöhe, des jährlichen Zinssatzes und der Laufzeit in Jahren, druckt das Programm den zugehörigen Tilgungsplan aus.

### Beispiel

Ein Darlehen in Höhe von 60.000 DM mit 12% Zins soll innerhalb von 10 Jahren durch Ratentilgung zurückbezahlt werden. Im ersten Jahr sind 6.000 DM Tilgung und

$$60000 \text{ DM} \cdot 0,12 = 7200 \text{ DM Zins}$$

fällig. Die anfängliche jährliche Belastung von

$$6000 \text{ DM} + 7200 \text{ DM} = 13200 \text{ DM}$$

nimmt somit jährlich um 720 DM ab.

Der genaue Tilgungsplan kann dem Programmausdruck entnommen werden.

Literaturhinweis: [9]



“Ein Kaufmann liest einem Schuldner die fälligen Zahlungen aus dem Schuldbuch vor; der Geselle berechnet am Rechenbrett die fällige Zahlung”  
Holzschnitt aus Luthers Deutschem Katechismus von 1530

```
100 REM .....RATENTILGUNG
110 :
120 CLS
130 PRINT STRING$(25,61)
140 PRINT" Ratentilgung"
150 PRINT STRING$(25,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Kredithoehe";K
190 PRINT:INPUT" Jahreszinssatz in %";P
200 PRINT:INPUT" Laufzeit in Jahren";N
210 PRINT:INPUT" Wieviele Zinsperioden pro Jahr";I
220 T=K/(N*I)
230 P=P/I
240 S=0
250 :
260 REM .....AUSGABE
270 PRINT:PRINT STRING$(39,45)
280 PRINT"Per. Rate Tilg.ant. Zinsant. Rest"
290 PRINT STRING$(39,45)
300 FOR J=1 TO N*I
310 Z=K*P/100
320 S=S+Z
330 R=Z+T
340 K=K-T
350 PRINT USING "## #####.## #####.## #####.## #####.##";J,R,T,Z,K
360 NEXT J
```

```

370 :
380 PRINT STRING$(39,45)
390 PRINT TAB(8) USING " Zinssumme= #####.## DM";S
400 PRINT STRING$(39,45)
410 END

```

```

=====
      Ratentilgung
=====

```

Kredithoehel? 60000

Jahreszinssatz in %? 12

Laufzeit in Jahren? 10

Wieviele Zinsperioden pro Jahr? 1

```

-----
Per.  Rate  Tilg.ant.  Zinsant.  Rest
-----
  1 13200.00  6000.00  7200.00  54000.00
  2 12480.00  6000.00  6480.00  48000.00
  3 11760.00  6000.00  5760.00  42000.00
  4 11040.00  6000.00  5040.00  36000.00
  5 10320.00  6000.00  4320.00  30000.00
  6  9600.00  6000.00  3600.00  24000.00
  7  8880.00  6000.00  2880.00  18000.00
  8  8160.00  6000.00  2160.00  12000.00
  9  7440.00  6000.00  1440.00   6000.00
 10  6720.00  6000.00   720.00   0.00
-----
                Zinssumme=   39600.00 DM
-----

```

## 17. ANNUITÄTENTILGUNG

Während bei der Ratentilgung der Tilgungsanteil konstant bleibt, ist bei der Annuitätentilgung die jährliche Belastung (= Annuität) gleichbleibend.

Wie beim Effektivzins geht man davon aus, daß der Endwert des ausgeliehenen Kapitals gleich dem Endwert der geleisteten nachschüssigen Zahlungen ist:

$$K \cdot q^n = R \frac{q^n - 1}{q - 1}$$

Auflösen dieser Gleichung nach  $R$  liefert die Annuitätenformel

$$R = K \frac{q^n(q-1)}{q^n - 1},$$

die identisch ist mit der Rentenformel, die man erhält, wenn ein Kapital in eine Rente verwandelt wird (vgl. Programm 14).

Für eine Hypothek von 100.000 DM ist bei 8% Zins und 25 Jahren Laufzeit eine Annuität

$$R = 100000 \text{ DM} \frac{1,08^{25} \cdot 0,08}{1,08^{25} - 1} = 9367,88 \text{ DM}$$

fällig. Oft wird jedoch nicht die Laufzeit, sondern die Tilgungsrate  $i$  vorgegeben. Die Laufzeit errechnet sich dann aus der Formel

$$n = \frac{\log\left(1 + \frac{p}{i}\right)}{\log\left(1 + \frac{p}{100}\right)},$$

dabei kann für  $\log$  der natürliche oder der Zehner-Logarithmus gewählt werden.

Bei Hypotheken wird häufig eine Tilgungsrate  $i = 1\%$  genommen; dies ergibt bei 7,5% Zins die Laufzeit

$$n = \frac{\log(1+7,5)}{\log 1,075} = 29,5913 \text{ Jahre.}$$

Bei Bauspardalehen sind  $i = 7\%$  und  $p = 5\%$  üblich; hier ergibt sich eine Laufzeit von

$$n = \frac{\log(1+5/7)}{\log 1,05} = 11,0472 \text{ Jahre.}$$

### **Zum folgenden Programm**

Nach Eingabe der Kredithöhe, des jährlichen Zinssatzes, der Laufzeit in Jahren, wird der Tilgungsplan der Annuitätentilgung ausgedruckt.

### **Beispiel**

Eine Hypothek von 100.000 DM soll in 10 Jahren bei 12% Zins durch Annuitäten getilgt werden. Die Annuität beträgt

17.698,42 DM.

Dies entspricht einer Monatsrate von

1474,87 DM,

falls die Tilgungszahlungen nicht unterjährig verzinst werden.

Der genaue Tilgungsplan kann dem Programmausdruck entnommen werden.

Literaturhinweis: [9]

```

100 REM .....ANNUITAETENTILGUNG
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT" Annuitaetentilgung"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Kredithoeh";K
190 PRINT:INPUT" Jahreszinssatz in %";P
200 PRINT:INPUT" Laufzeit in Jahren";N
210 PRINT:INPUT" Wieviele Zinsperioden pro Jahr";I
220 :
230 REM .....BERECHNUNG
240 P=P/(100*I)
250 Q=(1+P)^(N*I)
260 A=K*Q*P/(Q-1)
270 T=A/Q
280 S=0
290 :
300 REM .....AUSGABE
310 PRINT:PRINT STRING$(39,45)
320 PRINT"Per. Annuitt. Tilg Zinsant. Rest      "
330 PRINT STRING$(39,45)
340 FOR J=1 TO N*I
350   Z=A-T
360   K=K-T
370   S=S+Z
380   PRINT USING "## #####.## #####.## #####.## #####.##";J,A,T,Z,K
390   T=T*(1+P)
400 NEXT J
410 :
420 PRINT STRING$(39,45)
430 PRINT TAB(8) USING " Zinssumme= #####.## DM";S
440 PRINT STRING$(39,45)
450 END

```

=====  
 Annuitaetentilgung  
 =====

Kredithoehel? 100000

Jahreszinssatz in %? 12

Laufzeit in Jahren? 10

Wieviele Zinsperioden pro Jahr? 1

Per.	Annuit.	Tilg	Zinsant.	Rest
1	17698.42	5698.42	12000.00	94301.58
2	17698.42	6382.23	11316.19	87919.35
3	17698.42	7148.09	10550.32	80771.26
4	17698.42	8005.87	9692.55	72765.40
5	17698.42	8966.57	8731.85	63798.82
6	17698.42	10042.56	7655.86	53756.26
7	17698.42	11247.66	6450.75	42508.60
8	17698.42	12597.38	5101.03	29911.22
9	17698.42	14109.07	3589.35	15802.15
10	17698.42	15802.16	1896.26	-0.01
Zinssumme=			76984.16	DM

## 18. WERTMINDERUNG bzw. STEIGERUNG EINES OBJEKTS

Die Wertänderung eines Objekts kann wie die Verzinsung eines Kapitals nach der Zinseszinsformel

$$K_n = K_0 q^n$$

berechnet werden (vgl. Programm 6). Der Aufzinsungsfaktor  $q = 1 + \frac{p}{100}$  bestimmt sich daraus zu

$$q = \sqrt[n]{\frac{K_n}{K_0}}$$

entsprechend der Zinssatz- bzw. die Änderungsrate

$$p = (q-1) \cdot 100\% = \left( \sqrt[n]{\frac{K_n}{K_0}} - 1 \right) \cdot 100\%.$$

Ein Reihenhaus vom Kaufpreis 342.000 DM konnte nach 4 Jahren zu 475.000 DM verkauft werden. Es ergibt sich eine jährliche Rendite von

$$\left( \sqrt[4]{\frac{475000}{342000}} - 1 \right) \cdot 100\% = (1,085596 - 1) \cdot 100\% = 8,56 \%$$

Das Vermögenseinkommen der privaten Haushalte in der Bundesrepublik stieg von 23,03 Mrd. DM 1970 auf 72,73 Mrd. DM im Jahre 1980 an. Die jährliche Wachstumsrate beträgt

$$\left( \sqrt[10]{\frac{72,73 \cdot 10^9}{23,03 \cdot 10^9}} - 1 \right) \cdot 100\% = (1,121869 - 1) \cdot 100\% = 12,2 \%$$

Die Kaufkraft der DM fiel von 100% 1970 auf 62% im Jahr 1979. Dies entspricht einer mittleren Geldentwertungsrage von

$$\left( 1 - \sqrt[9]{\frac{62}{100}} \right) \cdot 100\% = (1 - 0,948271) \cdot 100\% = 5,2 \%$$

### Zum folgenden Programm

Nach Eingabe des Anschaffungs-, Wiederverkaufswerts und der Nutzungsdauer berechnet das Programm die Wertsteigerungs- bzw. minderungsrate.

### Beispiel

Ein PKW zum Listenpreis von 32.768 DM wird nach 4 Jahren zu 16.000 DM weiterverkauft. Dies entspricht einer jährlichen Wertminderungsrate von 16,4 %.

```

100 REM .....WERTMINDERUNG/STEIGERUNG EINES OBJEKTS
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT" WERTMINDERUNG/-STEIGERUNG"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" ANSCHAFUNGSWERT";A
190 PRINT:INPUT" WIEDERVERKAUFSWERT";W
200 PRINT:INPUT" NUTZUNGSDAUER IN JAHREN";N
210 PRINT:PRINT STRING$(40,45)
220 IF W>A THEN 290
230 :
240 REM .....WERTMINDERUNG
250 R=1-(W/A)^(1/N)
260 PRINT"MITTL. WERTMINDERUNG =";
270 GOTO 320
280 :
290 REM .....WERTSTEIGERUNG
300 R=(W/A)^(1/N)-1
310 PRINT"MITTL. WERTSTEIGERUNG =";
320 PRINT USING "####.## %";100*R
330 PRINT STRING$(40,45)
340 END

```

=====

WERTMINDERUNG/-STEIGERUNG

=====

ANSCHAFFUNGSWERT? 32768

WIEDERVERKAUFSWERT? 16000

NUTZUNGSDAUER IN JAHREN? 4

-----

MITTL. WERTMINDERUNG = 16.41 %

-----



## 19. LINEARE ABSCHREIBUNG

Zum Ausgleich der Wertminderung von Gütern erlaubt der Gesetzgeber eine steuerliche Abschreibung, im Amtsdeutsch auch Absetzung für Abnutzung (AfA) genannt (vgl. § 7,1 EStG).

Eine häufige Form der Abschreibung ist die lineare Methode, bei der der gesamte Wertverlust in gleichbleibenden Raten abgeschrieben wird.

Kann z.B. eine Büro-Einrichtung vom Wert 5600 DM innerhalb von 10 Jahren linear abgeschrieben werden, so werden jährlich 560 DM abgesetzt.

Gebäude können i.a. nur mit 2% linear abgeschrieben werden. In bestimmten Fällen sind nach dem bekannten Paragraphen § 7b EStG erhöhte Abschreibungen möglich.

### **Zum folgenden Programm**

Nach Eingabe von Anschaffungswert, Schrottwert und Nutzungsdauer, wird für jedes Jahr die Höhe der Abschreibung und der verbleibende Restwert (= Buchwert) berechnet.

### **Beispiel**

Eine Maschine vom Anschaffungswert 40.000 DM soll in 10 Jahren linear auf den Schrottwert 500 DM abgeschrieben werden. Die jährliche AfA ist somit 3950 DM (vgl. Programmausdruck).

Literaturhinweis: [5]

```
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT" LINEARE ABSCHREIBUNG"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Anschaffungswert";W
190 INPUT" Schrottwert";S
200 INPUT" Nutzungsdauer in Jahren";N
210 :
220 REM .....BERECHNUNGEN
230 A=(W-S)/N
240 :
250 REM .....AUSGABE
260 PRINT STRING$(30,45)
270 PRINT" Jahr Abschreibung Buchwert"
280 PRINT STRING$(30,45)
290 PRINT USING " 0 0 DM ##### DM";W
300 W=W-A
310 FOR I=1 TO N
320 PRINT USING " ### ##### DM ##### DM";I,A,W
330 W=W-A
340 NEXT I
350 PRINT STRING$(30,45)
360 :
370 INPUT"Weitere Berechnungen (J/N)";A#
380 IF LEFT$(A#,1)="j" OR LEFT$(A#,1)="J" THEN 180
390 END
```

=====

LINEARE ABSCHREIBUNG

=====

Anschaffungswert? 40000  
 Schrottwert? 500  
 Nutzungsdauer in Jahren? 10

Jahr	Abschreibung	Buchwert
0	0 DM	40000 DM
1	3950 DM	36050 DM
2	3950 DM	32100 DM
3	3950 DM	28150 DM
4	3950 DM	24200 DM
5	3950 DM	20250 DM
6	3950 DM	16300 DM
7	3950 DM	12350 DM
8	3950 DM	8400 DM
9	3950 DM	4450 DM
10	3950 DM	500 DM

Weitere Berechnungen (J/N)?



## 20. DEGRESSIVE ABSCHREIBUNG

Da in den ersten Nutzungsjahren die Wertminderung besonders groß ist, verwendet man statt der linearen die degressive Abschreibung (auch geometrisch-degressiv oder Buchwert-AfA genannt).

Da vom jeweiligen Buchwert ein konstanter Prozentsatz  $p$  abgeschrieben wird, stellen die Restwerte eine fallende geometrische Folge dar

$$R_n = A \left(1 - \frac{p}{100}\right)^n$$

dabei ist  $A$  der Anschaffungswert.

Nach § 7 Abs. 2 EStG ist die degressive Abschreibung für bewegliche Güter und zwar mit dem 2,5fachen des linearen Satzes, höchstens jedoch 25%, zulässig.

Eine Büromaschine im Wert von 20.000 DM soll in 10 Jahren weitestgehend abgeschrieben werden. Dem linearen Satz von 10% entspricht dann der degressive von 25%.

Dabei ergeben sich folgende Buchwerte

$$R_1 = 20000 \text{ DM} \cdot 0,25 = 5000 \text{ DM}$$

$$R_2 = (20000 - 5000) \text{ DM} \cdot 0,25 = 3750 \text{ DM}$$

$$R_3 = (20000 - 5000 - 3750) \text{ DM} \cdot 0,25 = 2812 \text{ DM}$$


$$R_4 = (20000 - 5000 - 3750 - 2812) \text{ DM} \cdot 0,25 = 2110 \text{ DM}$$

.....

$$R_{10} = 1501 \text{ DM} \cdot 0,25 = 375 \text{ DM.}$$

Wie man sieht, werden die AfA-Beträge stets kleiner, so daß die Restwerte niemals Null werden.

Restwert einer Maschine vom Anschaffungswert 40.000 DM  
bei 20% degressiver Abschreibung



**Zum folgenden Programm**

Nach Eingabe des Anschaffungswertes, des Abschreibungssatzes und der Nutzungsdauer wird für jedes Jahr der Abschreibungsbetrag und der jeweilige Restwert berechnet.

**Beispiel**

Eine Maschine vom Anschaffungswert 40.000 DM soll degressiv 10 Jahre lang um 20% abgeschrieben werden.

Wie man dem Programmausdruck entnimmt, beträgt der Buchwert nach 10 Jahren noch immer 4295 DM.

Literaturhinweis: [5]

```
100 REM .....DEGRESSIVE ABSCHREIBUNG
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT" Degressive Abschreibung"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Anschaffungswert";W
190 INPUT" jaehr1.Abschreibungssatz in %";P
200 INPUT" Nutzungsdauer in Jahren";N
210 P=P/100
220 :
230 REM .....AUSGABE
240 PRINT STRING$(30,45)
250 PRINT" Jahr Abschreibung Buchwert"
260 PRINT STRING$(30,45)
270 PRINT USING " 0 0 DM ##### DM";W
280 W=W-A
290 FOR I=1 TO N
300 A=W*P
310 W=W-A
320 PRINT USING " ### ##### DM ##### DM";I,A,W
330 NEXT I
340 PRINT STRING$(30,45)
350 :
360 PRINT:INPUT"Weitere Berechnungen (J/N)";A#
370 IF LEFT$(A#,1)="j" OR LEFT$(A#,1)="J" THEN 180
380 END
```

=====

Degressive Abschreibung

=====

Anschaffungswert? 40000

jaehrl.Abschreibungssatz in %? 20

Nutzungsdauer in Jahren? 10

---

Jahr	Abschreibung	Buchwert
0	0 DM	40000 DM
1	8000 DM	32000 DM
2	6400 DM	25600 DM
3	5120 DM	20480 DM
4	4096 DM	16384 DM
5	3277 DM	13107 DM
6	2621 DM	10486 DM
7	2097 DM	8389 DM
8	1678 DM	6711 DM
9	1342 DM	5369 DM
10	1074 DM	4295 DM

---

## 21. GEMISCHT LINEARE UND DEGRESSIVE ABSCHREIBUNG

Da bei der degressiven Abschreibung der Schrottwert Null nicht erreicht werden kann, ist nach § 7,3 EStG der Übergang von der degressiven zur linearen Abschreibung erlaubt, jedoch nicht umgekehrt. Der günstigste Zeitpunkt dafür ist, wenn der lineare Abschreibungsbetrag dem degressiven gleichkommt. Bei einer Nutzungsdauer von 10 Jahren ist dies im 7. Jahr der Fall, der Restwert nach dem 6. Nutzungsjahr kann dann linear auf Null abgeschrieben werden.

### **Zum folgenden Programm**

Nach Eingabe des Anschaffungswerts, des degressiven Abschreibungssatzes und der Nutzungsdauer wird für jedes Jahr der jeweilige Abschreibungsbetrag und der Buchwert berechnet. Das Programm geht zum optimalen Zeitpunkt von der degressiven zur linearen Abschreibung über; die Art der Abschreibung wird jeweils angegeben.

### **Beispiel**

Eine Maschine vom Anschaffungswert 50.000 DM soll in 10 Jahren optimal auf den Wert Null abgeschrieben werden.

Wie man dem Programmausdruck entnehmen kann, erfolgt die AfA 7 Jahre lang degressiv. Der Buchwert von 8899 DM nach dem 6. Jahr wird dann linear 4 Jahre lang mit je 2225 DM abgeschrieben. Dabei stimmt im 7. Jahr der lineare und degressive Abschreibungsbetrag überein.

Literaturhinweis: [5]

```

100 REM ...LINEARE UND DEGRESSIVE ABSCHREIBUNG
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT TAB(10)"Gemischte Abschreibung"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Anschaffungswert";A
190 INPUT" degress.Abschreibungssatz in %";F
200 INPUT" Nutzungsdauer in Jahren";N
210 :
220 REM .....AUSGABE
230 PRINT STRING$(39,45)
240 PRINT" Jahr   Abschreibung   Buchwert   Art"
250 PRINT STRING$(39,45)
260 PRINT USING "  O           O DM   ##### DM   \ \ ";A,"Deg"
270 :
280 REM .....Degress.Abschreibung
290 J=1
300 B=A:F=P/100
310 D=B*F
320 L=B/(N-J+1)
330 IF L>D THEN 390
340 B=B-D
350 PRINT USING "###   ##### DM   ##### DM   \ \ ";J,D,B,"Deg"

```

```
360 J=J+1:GOTO 310
370 :
380 REM .....Lin.Abschreibung
390 B=B-L
400 IF B<=0 THEN B=0
410 PRINT USING "###     ##### DM     ##### DM           ";J,L,B,"Lin"
420 J=J+1
430 IF J<=N THEN 390
440 PRINT STRING$(39,45)
450 :
460 PRINT:INPUT"Weitere Berechnungen (J/N)";A$
470 IF LEFT$(A$,1)="j" OR LEFT$(A$,1)="J" THEN 180
480 END
```

=====

Gemischte Abschreibung

=====

Anschaffungswert? 50000  
 degress.Abschreibungssatz in %? 25  
 Nutzungsdauer in Jahren? 10

Jahr	Abschreibung	Buchwert	Art
0	0 DM	50000 DM	Deg
1	12500 DM	37500 DM	Deg
2	9375 DM	28125 DM	Deg
3	7031 DM	21094 DM	Deg
4	5273 DM	15820 DM	Deg
5	3955 DM	11865 DM	Deg
6	2966 DM	8899 DM	Deg
7	2225 DM	6674 DM	Deg
8	2225 DM	4449 DM	Lin
9	2225 DM	2225 DM	Lin
10	2225 DM	0 DM	Lin

## 22. DIGITALE ABSCHREIBUNG

Auch bei der digitalen Abschreibung erhält man monoton fallende Abschreibungsbeträge, so daß in den ersten Nutzungsjahren die Abschreibungsbeträge besonders hoch sind.

Werden dabei in den ersten drei Jahren die Abschreibungsbeträge der degressiven Methode überschritten, so ist die digitale AfA nach § 7, Abs. 2 EStG nicht gestattet. Sie kann jedoch zur innerbetrieblichen Buchführung verwendet werden.

Die digitale Abschreibung berechnet sich wie folgt:

Ist die Nutzungsdauer  $n$  Jahre, so bildet man die Summe

$$s = 1 + 2 + 3 + \dots + n$$

und erhält damit folgende Abschreibungsbeträge

$$\begin{array}{ll} 1. \text{ Jahr} & \frac{n}{s} A \\ 2. \text{ Jahr} & \frac{n-1}{s} A \\ 3. \text{ Jahr} & \frac{n-2}{s} A \\ & \vdots \\ & \vdots \\ n\text{-tes Jahr} & \frac{1}{s} A \end{array}$$

Dabei ist  $A$  der Anschaffungswert.

### Beispiel

Eine Anlage im Wert von 30.000 DM soll digital in 5 Jahren vollständig abgeschrieben werden. Mit der Summe


$$s = 1 + 2 + 3 + 4 + 5 = 15$$

berechnen sich die Abschreibungsbeträge zu

$$\begin{array}{ll} 1. \text{ Jahr} & \frac{5}{15} \cdot 30.000 \text{ DM} = 10.000 \text{ DM} \\ 2. \text{ Jahr} & \frac{4}{15} \cdot 30.000 \text{ DM} = 8.000 \text{ DM} \\ 3. \text{ Jahr} & \frac{3}{15} \cdot 30.000 \text{ DM} = 6.000 \text{ DM} \\ 4. \text{ Jahr} & \frac{2}{15} \cdot 30.000 \text{ DM} = 4.000 \text{ DM} \\ 5. \text{ Jahr} & \frac{1}{15} \cdot 30.000 \text{ DM} = 2.000 \text{ DM} \end{array}$$

Es ergibt sich hier eine fallende arithmetische Folge mit dem Anfangswert 10.000 DM und der Differenz 2.000 DM.

Restwert einer Maschine vom Anschaffungswert 50.000 DM  
bei digitaler Abschreibung in 10 Jahren



**Zum folgenden Programm**

Nach Eingabe des Anschaffungswerts, Schrottwerts und der Nutzungsdauer wird für alle Jahre der jeweilige Abschreibungsbetrag und der Restwert ermittelt.

**Beispiel**

Eine Maschine vom Anschaffungswert 50.000 DM soll in 10 Jahren digital auf den Schrottwert 200 DM abgeschrieben werden.

Wie man dem Programmausdruck entnimmt, nimmt der Abschreibungsbetrag vom Anfangswert 9.055 DM jährlich jeweils um 905,45 DM ab.

```
100 REM .....DIGITALE ABSCHREIBUNG
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT"    Digitale Abschreibung"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Anschaffungswert";W
190 INPUT" Restwert in DM";S
200 INPUT" Nutzungsdauer in Jahren";N
210 :
220 REM .....BERECHNUNG
230 D=2*(W-S)/N/(N+1)
240 A=2*(W-S)/(N+1)
250 :
260 REM .....AUSGABE
270 PRINT STRING$(30,45)
280 PRINT" Jahr  Abschreibung  Buchwert"
290 PRINT STRING$(30,45)
300 PRINT USING "    O          O  DM    ##### DM";W
310 FOR I=1 TO N
320     W=W-A
330     PRINT USING "  ###    ##### DM    ##### DM"; I,A,W
340     A=A-D
350 NEXT I
```

```

360 PRINT STRING$(30,45)
370 :
380 PRINT:INPUT"Weitere Berechnungen (J/N)";A$
390 IF LEFT$(A$,1)="j" OR LEFT$(A$,1)="J" THEN
400 END

```

```

=====
          Digitale Abschreibung
=====

```

```

Anschaffungswert? 50000
Restwert in DM? 200
Nutzungsdauer in Jahren? 10

```

Jahr	Abschreibung	Buchwert
0	0 DM	50000 DM
1	9055 DM	40945 DM
2	8149 DM	32796 DM
3	7244 DM	25553 DM
4	6338 DM	19215 DM
5	5433 DM	13782 DM
6	4527 DM	9255 DM
7	3622 DM	5633 DM
8	2716 DM	2916 DM
9	1811 DM	1105 DM
10	905 DM	200 DM

## 23. INTERNER ZINSFUSS BEI INVESTITIONEN

Eine Methode zur Berechnung der Rentabilität einer Investition ist die Methode des internen Zinsfusses.

Sind  $E_1, E_2, \dots, E_n$  die Investitionserträge, so wird der interne Zinsfuß dadurch bestimmt, daß der Barwert der Erträge gleich der Investitionshöhe  $K$  ist.

$$\frac{E_1}{1+r} + \frac{E_2}{(1+r)^2} + \dots + \frac{E_n}{(1+r)^n} = K$$

Da diese Beziehung eine nichtlineare Gleichung für den Zinsfuß  $r$  darstellt, muß eine Intervallschachtelung zur Berechnung durchgeführt werden. Ist der interne Zinsfuß kleiner als die zu erzielende Rendite, so ist die Investition unrentabel. Mit dem internen Zinsfuß können auch verschiedene zur Diskussion stehende Investitionen miteinander verglichen werden.

### Zum folgenden Programm

Nach Eingabe der Investitionshöhe, der Nutzungsdauer und der jeweiligen Erträge, berechnet das Programm mit Hilfe einer Intervallschachtelung den internen Zinsfuß.

### Beispiel

Durch den Kauf einer Maschine im Wert von 10.000 DM kann ein Fabrikant während der nächsten 5 Jahre die Erträge

1000 DM, 2000 DM, 2500 DM, 3000 DM und 3500 DM

erzielen. Der Schrottwert der Maschine ist 500 DM. Das Programm liefert den internen Zinsfuß 6,57 %.

Literaturhinweis: [10]

```
100 REM ..INTERNER ZINSFUSS BEI INVESTITIONEN
110 :
120 CLS
130 PRINT STRING$(25,61)
140 PRINT" Interner Zinsfuss"
150 PRINT STRING$(25,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Investitionshoehe";K
190 PRINT:INPUT" Zeitraum in Jahren";N
200 PRINT:INPUT" Restwert";R
210 DIM E(N):PRINT
220 FOR I=1 TO N
230     PRINT" Ertrag von Jahr";I;:INPUT E(I)
240 NEXT I
250 E(N)=E(N)+R
260 :
270 REM .....INTERVALLSCHACHTELUNG
280 P1=0:P2=1:P3=0
290 P=(P1+P2)/2
300 IF ABS(P-P3)<.00001 THEN 410
310 P3=P
320 S=0
330 FOR I=1 TO N
340 S=S+E(I)/(P+1)^I
350 NEXT I
```

```
360 IF S=K THEN 410
370 IF K>S THEN P2=P ELSE P1=P
380 GOTO 290
390 :
400 REM .....AUSGABE
410 PRINT:PRINT STRING$(29,45)
420 PRINT USING "Interner Zinsfuss = ####.## %";100*F
430 PRINT STRING$(29,45)
440 END
```

=====  
Interner Zinsfuss  
=====

Investitionshoehe? 10000

Zeitraum in Jahren? 5

Restwert? 500

Ertrag von Jahr 1 ? 1000

Ertrag von Jahr 2 ? 2000

Ertrag von Jahr 3 ? 2500

Ertrag von Jahr 4 ? 3000

Ertrag von Jahr 5 ? 3500

-----  
Interner Zinsfuss =      6.57 %  
-----

## 24. KAPITALWERTMETHODE BEI INVESTITIONEN

Da die Methode des internen Zinsfußes ohne Rechner nicht durchführbar ist, verwendete man früher die Kapitalwertmethode, bei der mit geeigneten Tabellen gearbeitet wird.

Dabei berechnet man den Barwert der Investitionserträge für einen vorgegebenen Zinssatz (Kalkulationszinssatz) und vermindert ihn um die Investitionshöhe. Ist der so berechnete Kapitalwert positiv, so ist die Investition rentabel. Ein negativer Kapitalwert zeigt, daß der kalkulierte Zinsfuß nicht erreicht worden ist.



### Zum folgenden Programm

Nach Eingabe der Investitionshöhe, der Nutzungsdauer, der angestrebten Rendite und des Restwerts, berechnet das Programm den Kapitalwert der Investition nach dem oben angegebenen Verfahren.

### Beispiel

Durch eine Investition von 10.000 DM könnte ein Unternehmer in den folgenden 5 Jahren folgende Erträge erzielen:

1000 DM, 2000 DM, 2500 DM, 3000 DM und 3500 DM.

Bei einer angestrebten Verzinsung von 10% ist der Kapitalwert der Investition –1027 DM; d.h. die Investition erbringt nicht die gewünschte Rendite.

Literaturhinweis: [10]

```

100 REM ..KAPITALWERTMETHODE BEI INVESTITIONEN
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT" Kapitalwertmethode"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Investitionshoehe";K
190 PRINT:INPUT" Zeitraum in Jahren";N
200 PRINT:INPUT" Gewuenschte Rendite in %";P
210 PRINT:INPUT" Restwert";R
220 PRINT
230 :
240 REM .....BERECHNUNGEN
250 Q=1+P/100
260 FOR I=1 TO N
270     PRINT" Ertrag im Jahr";I;:INPUT E
280     S=S+E/Q^I
290 NEXT I
300 S=S+R/Q^N-K
310 :
320 REM .....AUSGABE
330 PRINT:PRINT STRING$(39,45)
340 PRINT USING "Kapitalwert d.Investition #####.## DM";S
350 PRINT STRING$(39,45)
360 END

```

=====  
Kapitalwertmethode  
=====

Investitionshoehe? 10000

Zeitraum in Jahren? 5

Gewuenschte Rendite in %? 10

Restwert? 500

Ertrag im Jahr 1 ? 1000

Ertrag im Jahr 2 ? 2000

Ertrag im Jahr 3 ? 2500

Ertrag im Jahr 4 ? 3000

Ertrag im Jahr 5 ? 3500

-----  
Kapitalwert d. Investition      -1027.00 DM  
-----



## 25. ANNUITÄTENMETHODE BEI INVESTITIONEN

Ebenfalls aus der Zeit des Handrechnens stammt die Annuitätenmethode. Aus Einfachheitsgründen geht sie von den durchschnittlichen Kosten und Erträgen aus, die durch die Investition anfallen.

Die mittleren Kosten werden als Annuitätenzahlung (vgl. Programm 17) aufgefaßt und können daher mit der Annuitätenformel berechnet werden. Übersteigen die so ermittelten durchschnittlichen Kosten die durchschnittlichen Einnahmen, so ist die Investition als unrentabel zu betrachten. Die Differenz der mittleren Einnahmen und Kosten stellt den durchschnittlichen Ertrag dar. Diese Differenz erlaubt auch wieder einen Vergleich zwischen mehreren zur Diskussion stehenden Investitionen.

### **Zum folgenden Programm**

Nach Eingabe der Investitionshöhe, der Nutzungsdauer, der gewünschten Rendite, des Restwerts und der mittleren Kosten und Einnahmen pro Jahr, berechnet das Programm den durchschnittlichen Ertrag der Investition nach der Annuitätenformel.

### **Beispiel**

Durch eine Investition in Höhe von 10.000 DM könnte ein Fabrikant 5 Jahre lang, bei durchschnittlich 1000 DM Kosten, durchschnittlich 3400 DM einnehmen. Die Rendite soll 10% betragen.

Es ergeben sich mittlere Kosten in Höhe von 3506,08 DM. Der durchschnittliche Ertrag beträgt somit -106,08 DM. Die angestrebte Verzinsung wird also nicht erreicht.

Literaturhinweis: [10]

```

100 REM ..ANNUITAETENMETHODE BEI INVESTITIONEN
110 :
120 CLS
130 PRINT STRING$(35,61)
140 PRINT " ANNUITAETENMETHODE"
150 PRINT STRING$(35,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Investitionshoehe";K
190 PRINT:INPUT" Zeitraum in Jahren";N
200 PRINT:INPUT" Gewuenschte Rendite in %";P
210 PRINT:INPUT" Restwert";R
220 PRINT:INPUT" Mittlere Kosten pro Jahr";C
230 PRINT:INPUT" Mittlere Einnahmen pro Jahr";E
240 :
250 REM .....BERECHNUNG
260 Q=1+P/100:QN=Q^N
270 C=(K-R)*P*QN/(QN-1)/100+C
280 :
290 REM .....AUSGABE
300 PRINT:PRINT STRING$(39,45)
310 PRINT USING "Durchschnittl. Kosten      #####.##";C
320 PRINT USING "Durchschnittl. Einnahmen  #####.##";E
330 PRINT USING "Durchschnittl. Ertrag    #####.##";E-C
340 PRINT STRING$(39,45)
350 END

```

=====  
ANNUITAETENMETHODE  
=====

Investitionshoehe? 10000

Zeitraum in Jahren? 5

Gewuenschte Rendite in %? 10

Restwert? 500

Mittlere Kosten pro Jahr? 1000

Mittlere Einnahmen pro Jahr? 3400

-----  
Durchschnittl. Kosten           3506.08  
Durchschnittl. Einnahmen       3400.00  
Durchschnittl. Ertrag           -106.08  
-----



## 26. ENTSCHEIDUNG BEI MEHREREN ZIELEN

Das Vorgehen wird an einem Beispiel (entnommen aus [4]) gezeigt. Von einer Unternehmensleitung werden folgende Firmenziele  $Z_j$  und ihre Bewertung  $g_j$  formuliert:

$Z_1$ : Beibehaltung der bestehenden Firmenleitung	0,25
$Z_2$ : Garantie einer 6%-igen Dividende	0,30
$Z_3$ : 15% Rendite bei zusätzlichen Investitionen	0,10
$Z_4$ : Keine Personalentlassungen	0,15
$Z_5$ : Stillhalteabkommen zwischen Firmenleitung und Personal	0,05
$Z_6$ : Beibehaltung der Fertigungsqualität	0,05
$Z_7$ : Förderung der Ortsgemeinde	0,10
Summe:	1,00

Zur Erreichung dieser Ziele werden drei Strategien diskutiert:

1 : Erweiterung der Produktionspalette und Erschließung neuer Märkte
2 : Beibehaltung der Firmengröße und technische Verbesserung der bisher erzeugten Produkte
3 : Beibehaltung der Firmengröße und Ersetzung von weniger gewinnbringenden Produkten durch rentablere

Das Erreichen der Ziele  $Z_j$  durch die Strategien  $j=1,2,3$  wird durch (subjektive) Wahrscheinlichkeiten  $p_{ij}$  mit  $0 < p_{ij} < 1$  gekennzeichnet:

0 heißt, daß die Strategie das Ziel höchst unwahrscheinlich fördert; 0,5 bedeutet: die Strategie spricht weder für noch gegen das Ziel; 1 zeigt an, daß die Strategie das Ziel höchstwahrscheinlich fördert. Entsprechend sind die mittleren Werte zu verstehen.

Folgende Wahrscheinlichkeitstabelle wird gewählt:

Strategie \ Ziel	Ziel						
	Z1	Z2	Z3	Z4	Z5	Z6	Z7
1	0,4	0,2	0,8	0,8	0,3	0,6	0,8
2	0,9	0,9	0,2	0,3	0,8	0,4	0,3
3	0,7	0,7	0,4	0,3	0,7	0,8	0,5

Multipliziert man die Zeilenelemente der Wahrscheinlichkeitstabelle mit den entsprechenden Bewertungseinheiten der Ziele, so erhält man ein Maß für den Nutzen der einzelnen Strategien. Für Strategie 1 ergibt sich die Summe

$$0,4 \cdot 0,25 + 0,2 \cdot 0,3 + 0,8 \cdot 0,1 + 0,8 \cdot 0,15 + 0,3 \cdot 0,05 + 0,6 \cdot 0,05 + 0,8 \cdot 0,1 = 0,485.$$

Entsprechend folgt für Strategie 2 0,650 und für 3 0,595. Strategie 2 hat die größte Nutzensumme und ist daher im Sinne der von der Firma angestrebten Ziele optimal.

### Zum folgenden Programm

Die Anzahl der Ziele und Strategien, die Wahrscheinlichkeitstabelle und die Zielbewertungseinheiten werden in Form von DATA-Werten eingelesen. Das Programm berechnet daraus die Nutzensummen und gibt die optimale Strategie aus.

Im Programm wird das oben behandelte Beispiel berechnet (vgl. Programm-ausdruck).

```
100 REM ....ENTSCHEIDUNG UNTER MEHREREN ZIELEN
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT" Entscheidung unter mehreren Zielen"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINLESEN
180 READ N: REM ANZAHL DER ZIELE
190 READ M: REM ANZAHL DER STRATEGIEN
200 DIM E(M,N),B(N)
210 :
220 PRINT:PRINT " Entscheidungsmatrix:"
230 FOR I=1 TO M
240     FOR K=1 TO N
250         READ E(I,K): PRINT E(I,K);
260     NEXT K: PRINT
270 NEXT I:PRINT
280 PRINT" Bewertung der Ziele:"
290 FOR I=1 TO N
300     READ B(I): PRINT B(I);
310 NEXT I:PRINT
320 :
330 REM .....STRATEGIE-WERTE
340 PRINT:PRINT STRING$(39,45)
350 MX=0
```

```

360 FOR I=1 TO M
370   S=0
380   FOR K=1 TO N
390     S=S+E(I,K)*B(K)
400   NEXT K
410   PRINT TAB(6)" Wert der Strategie";I:"";S
420   IF S>MX THEN MX=S: J=I
430 NEXT I
440 :
450 REM .....OPTIMALE ENTSCHEIDUNG
460 PRINT STRING$(39,45)
470 PRINT " Optimale Entscheidung fuer Strategie";J
480 PRINT STRING$(39,45)
490 END
500 :
510 REM .....DATEN
520 DATA 7
530 DATA 3
540 DATA .4,.2,.8,.8,.3,.6,.8
550 DATA .9,.9,.2,.3,.8,.4,.3
560 DATA .7,.7,.4,.3,.7,.8,.5
570 DATA .25,.3,.1,.15,.05,.05,.1

```

=====  
Entscheidung unter mehreren Zielen  
=====

Entscheidungsmatrix:

.4	.2	.8	.8	.3	.6	.8
.9	.9	.2	.3	.8	.4	.3
.7	.7	.4	.3	.7	.8	.5

Bewertung der Ziele:

.25	.3	.1	.15	.05	.05	.1
-----	----	----	-----	-----	-----	----

-----  
Wert der Strategie 1 : .485

Wert der Strategie 2 : .65

Wert der Strategie 3 : .595

-----  
Optimale Entscheidung fuer Strategie 2  
-----



## 27. ENTSCHEIDUNG BEI GEGEBENEN WAHRSCHEINLICHKEITEN

Während bei Programm 26 Entscheidungen unter frei wählbaren Zielen zu fällen waren, sind nun die äußeren Umstände, z.B. die wirtschaftliche Entwicklung, vorgegeben. In der Entscheidungstheorie spricht man von den Gegebenheiten als dem Zustand der Welt, die Entscheidung ist ein Spiel gegen die Natur.

Auf die Zustände  $Z_j$ , die mit den (subjektiven) Wahrscheinlichkeiten  $p_j$  eintreten, kann durch Wahl einer Strategie  $i$  ( $i=1,2,..$ ) reagiert werden.

Die Ergebnisse  $E_{ij}$ , die man bei der Wahl der Strategie  $i$  beim Zustand der Welt  $Z_j$  erhält, werden in einer Tabelle (Ergebnismatrix genannt) dargestellt:

Strategie \ Zustand	Zustand				
	$Z_1$	$Z_2$	$Z_3$	....	$Z_m$
1	$E_{11}$	$E_{12}$	$E_{13}$	....	$E_{1m}$
2	$E_{21}$	$E_{22}$	$E_{23}$	....	$E_{2m}$
.	...	...	...	....	...
n	$E_{n1}$	$E_{n2}$	$E_{n3}$	....	$E_{nm}$

Die Elemente der Tabelle können z.B. Gewinnerwartungen in DM sein.

Die Gewinnerwartung eines Unternehmens sei, abhängig vom Konjunkturverlauf, durch folgende Ergebnismatrix gegeben:

	Aufschwung	Stagnation	Rezession	Zeilen- minimum	Zeilen- maximum
1	10.000	5.000	3.000	3.000	10.000
2	5.000	5.000	5.000	5.000	5.000
3	20.000	5.000	-1.000	-1.000	20.000
4	10.000	3.000	4.000	3.000	10.000

Es wird geschätzt, daß alle Zustände mit gleicher Wahrscheinlichkeit eintreten:

$$p_1 = \frac{1}{3}, p_2 = \frac{1}{3} \text{ und } p_3 = \frac{1}{3}.$$

In der Entscheidungstheorie (vgl. [11]) werden folgende Kriterien zur Wahl der optimalen Strategie genannt:

a. Das **Erwartungswert**-Kriterium

Es wird diejenige Strategie gewählt, die den größten Erwartungswert

$$E_i = E_{i1}p_1 + E_{i2}p_2 + \dots + E_{im}p_m$$

erzielt. Diese Vorgehensweise entspricht der von Programm 28.

Beim angegebenen Beispiel gilt:

$$E_1 = 10.000 \text{ DM} \cdot \frac{1}{3} + 5000 \text{ DM} \cdot \frac{1}{3} + 3000 \text{ DM} \cdot \frac{1}{3} = 6000 \text{ DM}$$

$$E_2 = 5000 \text{ DM}$$

$$E_3 = 8000 \text{ DM}$$

$$E_4 = 5667 \text{ DM.}$$

Nach dem Erwartungswert-Kriterium ist 3 die optimale Strategie. Eine Entscheidung nach diesem Kriterium ist risikoneutral.

b. Das **Minimax**-Kriterium

Das Minimax-Kriterium ist extrem pessimistisch; es wählt nämlich diejenige Strategie aus, deren ungünstigstes Ergebnis noch möglichst günstig ist. Dazu sucht man in der Ergebnismatrix das Maximum aller Zeilenminima; hier 5000 DM.

Im Sinne des Minimax-Kriteriums ist somit Strategie 2 optimal.

c. Das **Maximax**-Kriterium

Gegenüber dem Minimax-Kriterium ist das Maximax-Kriterium äußerst optimistisch; es wählt nämlich diejenige Strategie aus, die den überhaupt größten Gewinn verspricht. In der Entscheidungsmatrix wird dazu das Maximum aller Zeilenmaxima ausgesucht; hier 20.000 DM.

Im Sinne des Maximax-Kriteriums ist also Strategie 2 optimal.

d. Das **Hurwicz**-Kriterium

Einen Ausgleich zwischen dem pessimistischen Standpunkt des Minimax-Kriteriums und dem optimistischen Standpunkt des Maximax-Kriteriums schafft das Hurwicz-Kriterium. Dabei werden für alle Strategien die gewichteten Summen

$$\alpha \cdot \text{Zeilenmaximum} + (1 - \alpha) \cdot \text{Zeilenminimum}$$

berechnet und diejenige Strategie ausgewählt, deren Summe am größten ist. Die Zahl  $\alpha$  mit  $0 < \alpha < 1$  heißt Optimismusindex.

Für  $\alpha = 0,75$  ergibt sich bei

Strategie 1:  $0,75 \cdot 10.000 \text{ DM} + 0,25 \cdot 3000 \text{ DM} = 8250 \text{ DM}$

2:  $0,75 \cdot 5000 \text{ DM} + 0,25 \cdot 5000 \text{ DM} = 5000 \text{ DM}$

3:  $0,75 \cdot 20.000 \text{ DM} + 0,25 \cdot (-1000) \text{ DM} = 14.750 \text{ DM}$

4:  $0,75 \cdot 10.000 \text{ DM} + 0,25 \cdot 3000 \text{ DM} = 8250 \text{ DM}$ .

Im Sinne des Hurwicz-Kriteriums ist also Strategie 3 optimal für  $\alpha = 75\%$ .

### **Zum folgenden Programm**

Die Anzahl der Zustände, der Strategien, die Ergebnismatrix und die Wahrscheinlichkeiten werden in Form von DATA-Werten eingelesen, ebenso der Optimismusindex.

Das Programm druckt dann für jedes der beschriebenen Kriterien die optimale Strategie aus.

Im Programm wird das angegebene Beispiel behandelt.

```

100 REM ENTSCHEIDUNG BEI GEG.WAHRSCHEINLICHK.
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT " Entscheidung bei geg.Wahrscheinlichk."
150 PRINT STRING$(39,61)
160 :
170 REM .....EINLESEN
180 READ M: REM ANZAHL DER ZUSTAENDE
190 READ N: REM ANZAHL DER STRATEGIEN
200 DIM A(N,M),P(M),MX(N),MI(N),E(N)
210 FOR I=1 TO N
220     FOR J=1 TO M
230         READ A(I,J)
240     NEXT J
250 NEXT I
260 S=0
270 FOR I=1 TO M
280     READ P(I)
290     S=S+P(I)
300 NEXT I
310 IF ABS(S-1)>.000001 THEN PRINT CHR$(7);"Wahrscheinlichkeit <>1": END
320 :
330 REM .....BESTIMMUNG MINIMA UND MAXIMA
340 FOR I=1 TO N
350     MI=A(I,1): MX=A(I,1)

```

```
360   FOR J=2 TO M
370     IF A(I,J)>MX THEN MX=A(I,J)
380     IF A(I,J)<MI THEN MI=A(I,J)
390   NEXT J
400   MI(I)=MI
410   MX(I)=MX
420 NEXT I
430 :
440 REM .....ERWARTUNGSWERTKRITERIUM
450 FOR I=1 TO N
460   S=0
470   FOR J=1 TO M
480     S=S+A(I,J)*P(J)
490   NEXT J
500   E(I)=S
510 NEXT I
520 E=0: K=0
530 PRINT" Erwartungswertkriterium:"
540 PRINT" Strategie Gewinnerwartung"
550 FOR I=1 TO N
560   PRINT USING " ###          ##### ";I,E(I)
570   IF E(I)>E THEN E=E(I): K=I
580 NEXT I
590 PRINT " Optimale Strategie=";K
600 :
610 REM .....MINIMAX-KRITERIUM
```

```

620 PRINT STRING$(39,45)
630 E=MX(1): K=0
640 FOR I=2 TO N
650     IF MX(I)<E THEN E=MX(I): K=I
660 NEXT I
670 PRINT" Minimax=";E
680 PRINT" Optimale Strategie=";K
690 :
700 REM .....MAXIMAX-KRITERIUM
710 PRINT STRING$(39,45)
720 E=MX(1)
730 FOR I=2 TO N
740     IF MX(I)>E THEN E=MX(I): K=I
750 NEXT I
760 PRINT" Maximax=";E
770 PRINT" Optimale Strategie=";K
780 :
790 REM .....HURWITZKRITERIUM
800 READ A
810 PRINT STRING$(39,45)
820 PRINT" Hurwicz-Kriterium Optimismusindex=";A
830 PRINT" Strategie Gewinnerwartung"
840 E=0: K=0
850 FOR I=1 TO N
860     E(I)=A*MX(I)+(1-A)*MI(I)
870     PRINT USING " ###          ##### ";I,E(I)

```

```
880     IF E(I)>E THEN E=E(I): K=I
890 NEXT I
900 PRINT" Optimale Strategie=";K
910 PRINT STRING$(39,45)
920 END
930 :
940 REM .....DATEN
950 DATA 3,4
960 DATA 10000,5000,3000
970 DATA 5000,5000,5000
980 DATA 20000,5000,-1000
990 DATA 10000,3000,4000
1000 DATA .33333333,.33333333,.33333333
1010 DATA .75
```

=====  
Entscheidung bei geg.Wahrscheinlichk.  
=====

Erwartungswertkriterium:

Strategie Gewinnerwartung

1 6000

2 5000

3 8000

4 5667

Optimale Strategie= 3

-----  
Minimax= 5000

Optimale Strategie= 2

-----  
Maximax= 20000

Optimale Strategie= 3

-----  
Hurwicz-Kriterium Optimismusindex= .75

Strategie Gewinnerwartung

1 8250

2 5000

3 14750

4 8250

Optimale Strategie= 3  
-----


## 28. LAGERHALTUNG OHNE FEHLMENGEN

Ist der Bedarf eines Lagers bekannt und zeitlich konstant und kann es jederzeit ergänzt werden, so läßt sich eine optimale Lagerhaltung angeben. Da der Bedarf als bekannt vorausgesetzt wird, spricht man hier von einem deterministischen Lagerhaltungsmodell, im Gegensatz zu einem stochastischen Modell, bei dem der Bedarf zufälligen Schwankungen unterliegt.

Ist  $T$  der optimale Bestellrhythmus und  $B$  der Bedarf je Zeiteinheit, so gilt für die optimale Bestellmenge  $Q$

$$Q = BT.$$

Der Lagerbestand nimmt vom Stand  $Q$  gleichförmig ab und wird beim Stand Null erneut aufgefüllt:



Sind  $H$  die Lagerhaltungskosten je Stück und Zeiteinheit, so stellt

$$HQT/2$$

die variablen Lagerhaltungskosten dar. Zusammen mit den Fixkosten  $K$ , ergeben sich die gesamten Lagerkosten aus

$$K + HQT/2.$$

Bezieht man die Gesamtkosten auf die Zeiteinheit, so erhält man

$$KB/Q + HQ/2.$$

Diese Funktion hat ein Minimum für die optimale Bestellmenge

$$Q = \sqrt{\frac{2KB}{H}}$$

Diese Formel heißt Wilsonsche Losgrößenformel oder Andlersche Lagerwurzel. Daraus berechnen sich der optimale Bestellrhythmus zu

$$T = \sqrt{\frac{2K}{BH}}$$

und die minimalen Lagerhaltungskosten zu

$$M = \sqrt{2KHB}$$

### Zum folgenden Programm


Nach Eingabe der Fixkosten, des Bedarfs und der Lagerkosten pro Stück und Zeiteinheit berechnet das Programm nach den angegebenen Formeln die optimale Lagerhaltung.

### Beispiel

Ein Lager benötigt wöchentlich 300 Stück einer Ware. Die Fixkosten einer Bestellung sind 1000 DM und die Lagerkosten 1,90 DM je Stück und Woche.

Das Programm berechnet hier den optimalen Bestellrhythmus zu 1,87 Wochen und die optimale Bestellmenge zu 562 Stück. Die minimalen Kosten betragen damit 1067,71 DM.

Literaturhinweis: [2], [4], [16]



“Kaufmann mit seinen Waren im Hauptbuch blättern”  
Holzschnitt von Hans Weiditz, Augsburg 1539

```
100 REM .....LAGERHALTUNG OHNE FEHLMENGEN
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT"    Lagerhaltung ohne Fehlmengen"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Fixkosten";K
190 PRINT:INPUT" Bedarf je Zeiteinheit";B
200 PRINT:INPUT" Lagerkosten je Stck.u.Zeiteinh";H
210 :
220 REM .....BERECHNUNGEN
230 M=SQR(2*K*B/H)
240 T=SQR(2*K/B/H)
250 Q=SQR(2*K*B/H)
260 :
270 REM .....AUSGABE
280 PRINT:PRINT STRING$(39,45)
290 PRINT USING" Minimalkosten           = #####.## DM";M
300 PRINT USING" Opt.Bestellrhythmus = ###.## Zeiteinh.";T
310 PRINT USING" Opt.Bestellmenge     = ##### Stueck";Q
320 PRINT"           Servicegrad = 100 %"
330 PRINT STRING$(39,45)
340 END
```

=====  
Lagerhaltung ohne Fehlmengen  
=====

Fixkosten? 1000

Bedarf je Zeiteinheit? 300


Lagerkosten je Stck.u.Zeiteinh? 1.9

-----  
Minimalkosten = 1067.71 DM  
Opt.Bestellrhythmus = 1.87 Zeiteinh.  
Opt.Bestellmenge = 562 Stueck  
Servicegrad = 100 %  
-----

## 29. LAGERHALTUNG MIT FEHLMENGEN

Die Lagerhaltung ohne Fehlmengen (Programm 28) ist aufwendig, da stets die gesamte Nachfrage befriedigt werden soll. Läßt man vorübergehend auch Fehlmengen zu, erreicht man u.U. eine wesentliche Kostenreduzierung.

Ist die maximale Bestellmenge  $S$ , so ändert sich der Warenbestand wie folgt



Sind  $G$  die Fehlmengenkosten je Stück und Zeiteinheit, so sind die gesamten Fehlmengenkosten

$$G(Q-S) (T-t)/2$$

wie man der Skizze entnimmt. Die Gesamtkosten addieren sich damit zu

$$K + HS^2/(2B) + G(BT-S) (T-S/B)/2$$

Auf die Zeiteinheit bezogen, sind die Gesamtkosten

$$\frac{K}{T} + \frac{BGT}{2} - GS + \frac{(G+H)S^2}{2BT}$$

Mit Hilfe der Differentialrechnung läßt sich zeigen, daß diese Funktion für die optimale Bestellmenge

$$Q = R \sqrt{\frac{2KB}{H}}$$

ein Minimum hat. Entsprechend ergeben sich der optimale Bestellrhythmus

$$T = R \sqrt{\frac{2K}{BH}}$$

die minimalen Lagerkosten

$$M = \frac{1}{R} \sqrt{2KHB}$$

und der befriedigte Bedarf

$$S = \frac{1}{R} \sqrt{\frac{2KB}{H}}$$

Dabei wurde zur Abkürzung

$$\sqrt{\frac{H+G}{G}} = R$$

gesetzt. Das Verhältnis

$$\frac{S}{Q} = \frac{1}{R^2}$$

nennt man den Servicegrad.

### **Zum folgenden Programm**

Nach Eingabe der Fixkosten, Lagerkosten und Fehlmengenkosten berechnet das Programm nach den angegebenen Formeln die optimale Lagerhaltung.

### **Beispiel**

Ein Lager benötigt wöchentlich 300 Stück einer Ware. Die Fixkosten einer Bestellung sind 1000 DM, die Lagerkosten 1,90 DM und die Fehlmengenkosten 2 DM je Stück und Woche.

Der optimale Bestellrhythmus ergibt sich dann zu 2,62 Wochen, die optimale Bestellmenge liegt bei 785 Stück. Die Minimalkosten betragen 764,60 DM und sind damit gegenüber dem Beispiel von Programm 30 deutlich reduziert. Dagegen kann nur 51 % des Bedarfs gedeckt werden.

Literaturhinweis: [2], [4], [16]

```

100 REM .....LAGERHALTUNG MIT FEHLMENGEN
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT"      Lagerhaltung mit Fehlmengen"
150 PRINT STRING$(39,61)
160 :
170 REM .....EINGABE
180 PRINT:INPUT" Fixkosten";K
190 PRINT:INPUT" Bedarf je Zeiteinheit";B
200 PRINT:INPUT" Lagerkosten je Stck.u.Zeiteinh";H
210 PRINT:INPUT" Fehlkosten je Stck.u.Zeiteinh";G
220 :
230 REM .....BERECHNUNGEN
240 R=SQR((H+G)/G)
250 M=SQR(2*K*B*H)/R
260 T=SQR(2*K/B/H)*R
270 S=SQR(2*K*B/H)/R
280 Q=SQR(2*K*B/H)*R
290 :
300 REM .....AUSGABE
310 PRINT:PRINT STRING$(39,45)
320 PRINT USING" Minimalkosten          = #####.## DM";M
330 PRINT USING" Opt.Bestellrhythmus    = ###.## Zeiteinh.";T
340 PRINT USING" Opt.Bestellmenge       = ##### Stueck";Q
350 PRINT USING" Befried. Bedarf        = ##### Stueck";S
360 PRINT USING" Servicegrad            = ###.## %";100*S/Q
370 PRINT STRING$(39,45)
380 END

```

=====  
Lagerhaltung mit Fehlmengen  
=====

Fixkosten? 1000

Bedarf je Zeiteinheit? 300

Lagerkosten je Stck.u.Zeiteinh? 1.9

Fehlkosten je Stck.u.Zeiteinh? 2

-----  
Minimalkosten = 764.60 DM  
Opt.Bestellrhythmus = 2.62 Zeiteinh.  
Opt.Bestellmenge = 785 Stueck  
Befried. Bedarf = 402 Stueck  
Servicegrad = 51.28 %  
-----

## 30. OPTIMALE LAGERHALTUNG

Während die vorangegangenen Lagerhaltungs-Programme von einem gleichförmigen Bedarf ausgingen, läßt das folgende Programm wechselnden Bedarf zu. Die Gesamtkosten setzen sich zusammen aus Bestell- und Lagerhaltungskosten. Sind die Bestellkosten niedriger als die Lagerhaltungskosten, so wird man möglichst wenig Ware auf Lager halten und bei Bedarf neu bestellen.

Sind die Bestellkosten höher als die Lagerkosten, so wird man möglichst viel Ware auf Lager halten, damit wenig Ware bestellt werden muß.

Vorausgesetzt wird im weiteren, daß keine Fehlmengen zugelassen werden und Lieferungen momentan erfolgen.

Da jede Entscheidung über Lager- oder Bestellmenge Konsequenzen für die folgenden Monate hat, ist hier ein mehrstufiger Prozeß zu durchlaufen, wie er in der Dynamischen Programmierung häufig vorkommt. Im Gegensatz zum Lösungsverfahren zur optimalen Zuordnung (Programm 35) werden hier nicht alle Lösungsmöglichkeiten erzeugt und dann die optimale ausgewählt. Alternativen die nicht zu einer verbesserten Lösung führen können, werden von vorn herein ausgeschlossen.

### **Zum folgenden Programm:**

Nach einlesen der Gesamt-Bestellkosten, der Lagerkosten je Stück und Zeiteinheit und der Bedarfsmengen in Form von DATA-Werten berechnet das Programm mit Hilfe der dynamischen Programmierung die optimale Bestell- bzw. Lagermenge. Dabei wird angenommen, daß die Lagermenge am Anfang und am Ende Null sind. Das Ergebnis wird in tabellarischer Form ausgedruckt.

### **Beispiel:**

Der Bedarf einer Firma an einem bestimmten Artikel sei während 12 Monaten wie folgt gegeben:

80, 90, 120, 160, 100, 90, 120, 140, 60, 100, 150 und 200 Stück. Die Kosten je Bestellung betragen 50 DM, die Lagerhaltung 0.40 DM je Stück und Monat.

Da hier bei einem Bedarf von mehr als 125 Stück die Lagerkosten die Bestellkosten übertreffen, wird bei 9 Monaten jeweils neu bestellt. Bei 3 Monaten wird der Bedarf aus der Lagerhaltung entnommen (vgl. Programmausdruck). Die minimalen Kosten betragen hier 546 DM. Diese Summe setzt sich zusammen aus 9 Bestellkosten zu je 50 DM und den Lagerhaltungskosten für insgesamt 240 Stück Ware in den Monaten 1, 5 und 8.

```
100 REM Optimale Lagerhaltung
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT"   Optimale Lagerhaltung"
150 PRINT STRING$(30,61)
160 :
170 READ N   : REM Zahl der Zeiteinheiten
180 DIM K(N,N),M(N),BD(N),BS(N)
190 :
200 REM Einlesen
210 READ BK  : REM Bestellkosten
220 READ LK  : REM Lagerkosten je Stck u.Zeiteinheit
230 FOR I=1 TO N
240     READ BD(I)  : REM Bedarfsmengen
250 NEXT I
260 :
270 REM Startwerte
280 K(1,1)=BK:M(1)=1
290 :
300 REM Optimierung
310 FOR J=2 TO N
320     K(J,J)=K(M(J-1),J-1)+BK
330     FOR I=M(J-1) TO J-1
340         K(I,J)=K(I,J-1)+LK*BD(J)*(J-I)
350     NEXT I
```

```

360     M(J)=M(J-1)
370     FOR I=M(J-1) TO J-1
380         IF K(M(J),J) > K(I+1,J) THEN M(J)=I+1
390     NEXT I
400 NEXT J
410 :
420 REM Bestellmengen
430 N1=N
440 N2=M(N1):S=0
450 FOR I=N2 TO N1
460     S=S+BD(I)
470 NEXT I
480 BS(N2)=S
490 IF M(N1)=1 THEN 540
500 N1=N2-1:GOTO 440
510 :
520 REM Ausgabe
530 PRINT STRING$(39,45)
540 PRINT:PRINT"Periode Bedarf Bestellmenge Lagerbest"
550 PRINT STRING$(39,45)
560 LG=0
570 FOR I=1 TO N
580     IF LG=0 THEN LG=BS(I)
590     LG=LG-BD(I)
600     PRINT USING "   ###   #####   #####   #####"; I, BD(I), BS(I), LG
610 NEXT I

```

```
620 PRINT STRING$(39,45)
630 PRINT USING "Minimale Kosten =##### DM";K(M(N),N)
640 PRINT STRING$(39,45)
650 END
660 :
670 DATA 12
680 DATA 50
690 DATA .4
700 DATA 80,90,120,160,100,90,120,140,60,100,150,200
```

=====  
 Optimale Lagerhaltung  
 =====

Periode	Bedarf	Bestellmenge	Lagerbest
1	80	170	90
2	90	0	0
3	120	120	0
4	160	160	0
5	100	190	90
6	90	0	0
7	120	120	0
8	140	200	60
9	60	0	0
10	100	100	0
11	150	150	0
12	200	200	0

-----  
 Minimale Kosten = 546 DM  
 -----

## 31. WARTESCHLANGE BEI EINMANN-BEDIENUNG

Die Warteschlangen- bzw. Bedienungstheorie beschäftigt sich mit der Statistik der Warteschlangen bei Telefon-Vermittlungen, Verkehrsbetrieben, Geschäften usw.

Die Zahl der in einem bestimmten Zeitintervall erscheinenden Kunden wird als poissonverteilt, die mittlere Bedienungsdauer als exponentialverteilt angesehen.

Ist  $\lambda$  die mittlere Ankunftsrate von Kunden je Zeiteinheit,  $\mu$  die mittlere Bedienungsdauer je Zeiteinheit, so ist die Wahrscheinlichkeit, daß ein Kunde nicht warten muß

$$p = 1 - \lambda / \mu$$

Die mittlere Wartezeit  $W$  beträgt dann

$$W = \lambda / (\mu(\mu - \lambda))$$

Die Durchschnitts-Länge der Warteschlange ist gegeben durch

$$L = (\lambda / \mu) \wedge 2 / (1 - \lambda / \mu)$$

Diese Formeln gelten unter der Voraussetzung  $\lambda < \mu$ . Gilt dies nicht, so kommen stets mehr Kunden als abgefertigt werden können; die Warteschlange wird somit beliebig groß.

### **Zum folgenden Programm:**

Nach Eingabe der mittleren Kundenankunftsrate und der mittleren Bedienungsdauer berechnet das Programm die mittlere Wartezeit und Verweilzeit eines Kunden, die Wahrscheinlichkeit sofort bedient zu werden und die mittlere Länge der Warteschlange.

### **Beispiel:**

In einem Geschäft kommen durchschnittlich 5 Kunden pro Stunde. 6 Kunden können durchschnittlich je Stunde abgefertigt werden.

Ein Kunde wird dann mit der Wahrscheinlichkeit

$$1 - 5/6 = 16.67\% \text{ sofort bedient.}$$

Die mittlere Wartezeit beträgt

$$w = 5 / (6 * (6 - 5)) = 0.83 \text{ Std.}$$

Da er ein 1/6 Std. lang bedient wird, verweilt er somit

$$0.83 \text{ Std.} + 0.17 \text{ Std.} = 1 \text{ Std. im Laden.}$$

Es warten durchschnittlich

$$(5/6) \wedge 2 / (1 - 5/6) = 4.2 \text{ Kunden auf die Bedienung}$$

```

100 REM .....Warteschlange bei Einmannbedienung
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT"  Warteschlange bei Einmann-Bedienung"
150 PRINT STRING$(39,61):PRINT
160 :
170 REM .....Eingabe
180 INPUT " Mittl.Kundenankunftsrate je Zeiteinh. ";LAMBDA
190 INPUT " Mittl.Abfertigungsrate je Zeiteinh. ";MUE
200 IF LAMBDA<MUE THEN 260 ELSE 210
210 PRINT:PRINT CHR$(7);"Warteschlange wird beliebig gross,"
220 PRINT"da stets mehr Kunden kommen als"
230 PRINT"abgefertigt werden koennen":END
240 :
250 REM .....Berechnung
260 K=LAMBDA/(MUE-LAMBDA)
270 F=1-LAMBDA/MUE
280 L=LAMBDA^2/(MUE*(MUE-LAMBDA))
290 W=LAMBDA/(MUE*(MUE-LAMBDA))
300 B=1/MUE
310 A=LAMBDA/MUE
320 :
330 REM .....Ausgabe
340 PRINT:PRINT STRING$(39,45)
350 PRINT USING "Wahrscheinl.der Sofortbedienung ##.## %";F*100

```

```
360 PRINT USING " Mittlere Kundenzahl=    ##.##";K
370 PRINT USING " Mittlere Bedienungsdauer= ##.##";B
380 PRINT USING " Mittlere Wartezeit=      ##.##";W
390 PRINT USING " Mittlere Verweilzeit=    ##.##";B+W
400 PRINT USING " Mittl.Laenge d.Warteschlange= ###.#";L
410 PRINT USING " Auslastung der Bedienung= ###.# %";A*100
420 PRINT STRING$(39,45)
430 END
```

=====  
Warteschlange bei Einmann-Bedienung  
=====

Mittl.Kundenankunftsrate je Zeiteinh.? 5

Mittl.Abfertigungsrate je Zeiteinh.? 6

-----  
Wahrscheinl.der Sofortbedienung 16.67 %

Mittlere Kundenzahl= 5.00

Mittlere Bedienungsdauer= 0.17

Mittlere Wartezeit= 0.83

Mittlere Verweilzeit= 1.00

Mittl.Lnge d.Warteschlange= 4.2

Auslastung der Bedienung= 83.3 %  
-----

## 32. WARTESCHLANGE BEI MEHRFACHBEDIENUNG

Ist die Länge der Warteschlange sehr groß, so wird man die Zahl der Bedienungseinheiten erhöhen. Die folgende Theorie gilt unter der Bedingung, daß alle Kunden sich in eine Warteschlange einreihen und der jeweils erste zuerst bedient wird.

Ist  $d$  die Anzahl der Bedienungen, so können pro Zeiteinheit  $d \cdot \mu$  Kunden abgefertigt werden. Es muß  $\lambda < \mu$  gelten, da sonst die Warteschlange beliebig groß wird.

Die Wahrscheinlichkeit  $p$ , daß eine Kunde sofort bedient wird, ist

$$p = 1 - (\lambda/\mu) \cdot d \cdot s / ((d-1)! \cdot (d-\lambda/\mu))$$

dabei ist

$$s = 1 / ((\lambda/\mu) + (\lambda/\mu)^2 / 2! + (\lambda/\mu)^3 / 3! + \dots + (\lambda/\mu)^{d-1} / (d-1)! + (d-\lambda/\mu) \cdot d / ((d-1)! \cdot (d-\lambda/\mu))$$

und  $n!$  die Fakultätsfunktion, das Produkt alle Zahlen von 1 bis  $n$ .

Die mittlere Wartezeit  $W$  ist gegeben durch

$$W = (1-p) / (d \cdot \mu - \lambda)$$

Die mittlere Länge der Warteschlange beträgt

$$L = (1-p) \cdot (\lambda/\mu) / (d - \lambda/\mu)$$

Die Anzahl der sich im Laden aufhaltenden Kunden ist

$$K = \lambda \cdot (W + 1/\mu)$$

### Zum folgenden Programm:

Nach Eingabe der Anzahl der Bedienungseinheiten, der mittl. Anzahl von Kunden und der mittl. Bedienungsdauer berechnet das Programm die Wahrscheinlichkeit, daß ein Kunde sofort bedient wird, die Bedienungsdauer und die Wartezeit. Die Summe aus Bedienungs- und Wartezeit ergibt wieder die Verweilzeit des Kunden im Laden. Ferner wird die Anzahl der bedienten Kunden und die mittlere Länge der Warteschlange ermittelt.

### Beispiel:

In einem Geschäft mit 3 Verkäuferinnen erscheinen durchschnittlich 12 Kunden pro Std. Sechs Kunden werden pro Std. bedient. Das Programm liefert folgende Daten:

Mit 55.56% Wahrscheinlichkeit wird ein Kunde sofort bedient. Die Bedienungsdauer beträgt 0.17 Std., die mittl. Wartezeit 0.07 Std. Der Kunde verweilt somit 0.24 Std. im Laden. Es sind durchschnittlich 2.89 Kunden im Laden, davon stehen 0.89 in der Warteschlange.

```

100 REM .....Warteschlange bei Mehrfachbedienung
110 :
120 CLS
130 PRINT STRING$(39,61)
140 PRINT" Warteschlange bei Mehrfach-Bedienung"
150 PRINT STRING$(39,61):PRINT
160 :
170 REM .....Eingabe
180 INPUT " Wieviele Bedienungseinheiten ";D
190 INPUT " Mittl.Kundenankunftsrate je Zeiteinh.";LAMBDA
200 INPUT " Mittl.Abfertigungsrate je Zeiteinh.";MUE
210 IF LAMBDA<MUE*D THEN 270 ELSE 220
220 PRINT:PRINT CHR$(7);"Warteschlange wird beliebig gross,"
230 PRINT"da stets mehr Kunden kommen als"
240 PRINT"abgefertigt werden koennen":END
250 :
260 REM .....Berechnung
270 S=1:H=1:FAK=1
280 FOR I=1 TO D-1
290     H=H*LAMBDA/MUE:FAK=FAK*I
300     S=S+H/FAK
310 NEXT I
320 S=1/(S+H*LAMBDA/MUE/FAK/(D-LAMBDA/MUE))
330 P=1-(LAMBDA/MUE)^D*S/FAK/(D-LAMBDA/MUE)
340 W=(1-P)/(D*MUE-LAMBDA)
350 L=(1-P)*(LAMBDA/MUE)/(D-LAMBDA/MUE)

```

```
360 K=LAMBDA*(W+1/MUE)
370 B=1/MUE
380 :
390 REM .....Ausgabe
400 PRINT:PRINT STRING$(39,45)
410 PRINT USING "Wahrscheinl.der Sofortbedienung ##.## %";100*P
420 PRINT USING " Mittlere Bedienungsdauer= ##.##";B
430 PRINT USING " Mittlere Wartezeit=      ##.##";W
440 PRINT USING " Mittlere Verweilzeit=    ##.##";B+W
450 PRINT USING " Mittlere Kundenzahl =    ##.##";K
460 PRINT USING " Mittl.Laenge d.Warteschlange= ###.##";L
470 PRINT STRING$(39,45)
480 END
```

=====  
Warteschlange bei Mehrfach-Bedienung  
=====

Wieviele Bedienungseinheiten ? 3  
Mittl.Kundenankunftsrate je Zeiteinh.? 12  
Mittl.Abfertigungsrate je Zeiteinh.? 6

-----  
Wahrscheinl.der Sofortbedienung 55.56 %  
Mittlere Bedienungsdauer= 0.17  
Mittlere Wartezeit= 0.07  
Mittlere Verweilzeit= 0.24  
Mittlere Kundenzahl = 2.89  
Mittl.Laenge d.Warteschlange= 0.89  
-----

### 33. LINEARE OPTIMIERUNG

Viele wirtschaftliche Probleme wie Produktionsplanung, optimales Mischen und Transportprobleme können mit Hilfe der linearen Optimierung gelöst werden.

Unter der linearen Optimierung versteht man die Lösung eines Systems von linearen Ungleichungen so, daß eine vorgegebene lineare Funktion (Zielfunktion genannt) optimal wird.

Das grundlegende Lösungsverfahren der linearen Optimierung, Simplex-Algorithmus genannt, wurde 1947/1948 von dem Mathematiker G.B. Dantzig im Auftrag der amerikanischen Luftwaffe entwickelt. Im Rahmen dieses Buches ist es nicht möglich, auf die mathematischen Grundlagen des Simplex-Verfahrens einzugehen; es wird auf die zahlreiche Literatur verwiesen: [2], [12], [14], [16].

Eine Fabrik produziert 2 Werkstücke  $Y_1$  bzw.  $Y_2$ , die bei der Fertigung die Maschinen A, B, C durchlaufen müssen. Die Produktionskapazitäten werden durch folgende Tabelle gegeben:

		Bearbeitungszeit Stunden/Stück		Maschinenkapazität Stunden/Woche
		$Y_1$	$Y_2$	
Maschinentyp	A	2	1	200
	B	1	1	120
	C	1	3	240

Werden von den beiden Werkstücken  $x_1$  bzw.  $x_2$  Stück gefertigt, so benötigt die Maschine A dafür

$$2x_1 + x_2 \text{ Stunden.}$$

Da A höchstens 200 Stunden laufen kann, gilt

$$2x_1 + x_2 \leq 200.$$

Entsprechend gilt für Maschine B

$$x_1 + x_2 \leq 120$$

und für Maschine C

$$x_1 + 3x_2 \leq 240.$$

Betragen die Gewinne 2 DM bzw. 3 DM pro Stück, so soll die Gewinnfunktion (= Zielfunktion)

$$G = 2x_1 + 3x_2$$

maximal werden. Da negative Stückzahlen keinen Sinn haben, muß man noch – wie in der linearen Optimierung üblich – fordern:

$$x_1 \geq 0 \text{ und } x_2 \geq 0$$

Die optimale Lösung des Problems ist

$$x_1 = 60 \text{ und } x_2 = 60.$$

Das Unternehmen muß also wöchentlich 60 Stück vom Werkstück 1 bzw. 2 produzieren; der optimale Gewinn beträgt dann 300 DM.

ein solches System von linearen Ungleichungen muß jedoch nicht immer eine eindeutige Lösung haben. Es kann auch gar keine Lösung geben oder die Lösung ist nicht eindeutig oder sogar unbeschränkt.

### **Zum folgenden Programm**

Das folgende Programm löst beliebige Optimierungsprobleme, deren Nebenbedingungen lineare Ungleichungen sind.

Folgende DATA-Werte werden im Programm der Reihe nach eingelesen:

- MAX bzw. MIN, je nachdem ob die Zielfunktion maximal oder minimal werden soll
- die Zahl der Variablen
- die Zahl der Nebenbedingungen
- die Matrix der Nebenbedingungen
- die rechte Seite der Nebenbedingungen
- die Koeffizienten der Zielfunktion

Dabei ist vorausgesetzt, daß bei einem MAX-Problem die Ungleichungen in der Form » $\leq$ « vorliegen bzw. bei einem MIN-Problem in der Form » $\geq$ «.

### Beispiel aus [16]

Eine Firma verarbeitet einen Rohstoff zu drei Waren  $W_1$ ,  $W_2$  und  $W_3$ . Je Stück benötigt man davon 60 kg für  $W_1$ , 100 kg für  $W_2$  und 76 kg für  $W_3$ . An Arbeitszeit benötigt man 5 Std. für  $W_1$ , 10 Std. für  $W_2$  und 6 Std. für  $W_3$ . Im betrachteten Zeitraum stehen 1600 Arbeitsstunden und 20.000 kg Rohstoff zur Verfügung. Aus technischen Gründen muß von  $W_1$  mindestens doppelt so viel wie von  $W_2$  produziert werden. Der Gewinn je Stück beträgt 10 DM an  $W_1$ , 50 DM an  $W_2$  und 23 DM an  $W_3$ . Welche Stückzahlen müssen produziert werden, damit der Gewinn maximal wird?

Für den Rohstoff ergibt sich die Ungleichung

$$60x_1 + 100x_2 + 76x_3 \leq 20.000$$

für die Arbeitsstunden

$$5x_1 + 10x_2 + 6x_3 \leq 1600.$$

Außerdem muß gelten  $x_1 \geq 2x_2$  oder

$$-x_1 + 2x_2 + 0 \cdot x_3 \leq 0.$$

Die Zielfunktion

$$10x_1 + 50x_2 + 23x_3$$

soll maximal werden.

Nach Eingabe der Daten gemäß Programm ergibt sich die optimale Lösung

$$x_1 = 16, \quad x_2 = 8, \quad x_3 = 240 \text{ Stück.}$$

Der maximale Gewinn beträgt 6080 DM.

```

100 REM .....Lineare Optimierung
110 :
120 CLS
130 PRINT STRING$(35,61)
140 PRINT " *** Simplex-Verfahren ***"
150 PRINT STRING$(35,61)
155 :
160 REM Einlesen der DATA-Werte
170 READ P$: REM Minimum- oder Maximumsproblem
180 IF P#<>"MIN" AND P#<>"MAX" THEN PRINT"MIN oder MAX eingeben":END
190 READ N : REM Zahl der Variablen
200 READ M : REM Zahl der Nebenbedingungen
210 N1=N+M+1
220 IF P#="MIN" THEN M1=N+1:GOTO 240
230 M1=M+1
240 DIM BV(M1),A(M1,N1)
250 :
260 PRINT:PRINT" Matrix der Nebenbedingungen"
270 FOR I=1 TO M
280 FOR J=1 TO N
290 IF P#="MIN" THEN READ A(J,I):PRINT A(J,I);:GOTO 310
300 READ A(I,J):PRINT A(I,J);
310 NEXT J:PRINT
320 NEXT I:PRINT
330 :
340 PRINT" Rechte Seite"
350 FOR I=1 TO M

```

```
360 IF P#="MIN" THEN READ A(M1,I):PRINT A(M1,I);:GOTO 380
370 READ A(I,N1):PRINT A(I,N1);
380 NEXT I:PRINT:PRINT
390 :
400 PRINT " Zielfunktion"
410 FOR J=1 TO N
420 IF P#="MIN" THEN READ A(J,N1):PRINT A(J,N1);:GOTO 440
430 READ A(M1,J):PRINT A(M1,J);
440 NEXT J:PRINT:PRINT
450 IF P#="MIN" THEN HI=M:GOTO 470
460 HI=N
470 FOR J=1 TO HI
480 A(M1,J)=-A(M1,J)
490 NEXT J
500 A(M1,N1)=0
510 :
520 GOSUB 860:REM Aufruf Simplex-Prozedur
530 :
540 REM Ausgabe
550 PRINT STRING$(39,45)
560 PRINT "Optimale Loesung des ";
570 IF P#="MIN" THEN PRINT "Standard-Minimum-Problems":GOTO 590
580 PRINT "Standard-Maximum-Problems"
590 PRINT "nach";IT;"Schritten:"
600 PRINT STRING$(39,45)
610 FOR J=1 TO N
```

```
620 IF P#="MIN" THEN 690
630 V=0
640 FOR I=1 TO M
650 IF BV(I)=J THEN V=I:I=M
660 NEXT I
670 IF V=0 THEN X=0:GOTO 700
680 X=A(V,N1):GOTO 700
690 X=A(M1,M+J)
700 PRINT " x(";J;")=";X
710 NEXT J
720 PRINT STRING$(39,45)
730 PRINT "Optimaler Wert der Zielfunktion=";Z
740 PRINT STRING$(39,45)
750 END
760 :
770 DATA MAX
780 DATA 3
790 DATA 3
800 DATA 60,100,76
810 DATA 5,10,6
820 DATA -1,2,0
830 DATA 20000,1600,0
840 DATA 10,50,23
850 :
860 REM .....Prozedur Simplexverfahren
870 REM Basisvariable
```

```
880 IF P#="MIN" THEN 930
890 FOR I=1 TO M
900 BV(I)=N+I
910 NEXT I
920 :
930 REM Schlupfvariablen
940 IF P#="MIN" THEN H1=N:H2=M:GOTO 960
950 H1=M:H2=N
960 FOR I=1 TO M1
970 FOR J=1 TO H1
980 A(I,H2+J)=0
990 IF I=J THEN A(I,H2+J)=1
1000 NEXT J
1010 NEXT I
1020 :
1030 REM Pivotspalte
1040 IT=0
1050 IT=IT+1
1060 IF IT>(M*N*5) THEN 1490
1070 MI=A(M1,1):PS=1
1080 FOR J=2 TO N1-1
1090 IF A(M1,J)<MI THEN MI=A(M1,J):PS=J
1100 NEXT J
1110 IF MI>-.00001 THEN Z=A(M1,N1):RETURN
1120 :
1130 REM Pivotzeile
```

```

1140 MI=1E+20:PZ=0
1150 FOR I=1 TO M1-1
1160 IF A(I,PS)<=0 THEN 1190
1170 H=A(I,N1)/A(I,PS)
1180 IF H<MI THEN MI=H:PZ=I
1190 NEXT I
1200 IF PZ=0 THEN 1430
1210 :
1220 REM Basistausch
1230 IF P#="MAX" THEN BV(PZ)=PS
1240 HI=A(PZ,PS)
1250 FOR J=1 TO N1
1260 A(PZ,J)=A(PZ,J)/HI
1270 NEXT J
1280 IF PZ=1 THEN 1350
1290 FOR I=1 TO PZ-1
1300 HI=A(I,PS)
1310 FOR J=1 TO N1
1320 A(I,J)=A(I,J)-HI*A(PZ,J)
1330 NEXT J
1340 NEXT I
1350 FOR I=PZ+1 TO M1
1360 HI=A(I,PS)
1370 FOR J=1 TO N1
1380 A(I,J)=A(I,J)-HI*A(PZ,J)
1390 NEXT J

```

```
1400 NEXT I
1410 GOTO 1050
1420 :
1430 REM Unbeschraenkte Loesung
1440 IF P#="MIN" THEN 1460
1450 PRINT"Loesung ist unbeschraenkt":END
1460 PRINT"Es existiert keine Loesung"
1470 END
1480 :
1490 REM Entartete Loesung
1500 PRINT"Entartete Loesung"
1510 END:REM Prozedurende
```

```
=====
***   Simplex-Verfahren   ***
=====
```

```
Matrix der Nebenbedingungen
60  100  76
5   10   6
-1  2   0
```

```
Rechte Seite
20000  1600  0
```

```
Zielfunktion
10   50   23
```

```
-----
Optimale Loesung des Standard-Maximum-Problem:
nach 4 Schritten:
-----
```

```
x( 1 )= 16
x( 2 )= 8
x( 3 )= 240
```

```
-----
Optimaler Wert der Zielfunktion= 6080
-----
```

## 34. RENTABILITÄTSGRENZE

Während beim vorhergehenden Programm die Kosten- und Umsatzfunktion zur Ermittlung des optimalen Gewinns notwendig waren, kann bei linearen Produktionskosten die Rentabilitätsgrenze an Hand der Stückkosten berechnet werden.

### **Zum folgenden Programm**

Je nach Wahl der Bezugsbasis, wird 1 für die Produktionskapazität oder 2 für die Umsatzerwartung eingegeben.

Nach Eingabe der Fixkosten, der Stückkosten, des Stückpreises und der gewünschten Bezugsbasis, wird die Rentabilitätsgrenze der Produktion berechnet. Zusätzlich kann noch der Gewinn in Abhängigkeit vom Absatz tabellarisch ausgedrückt werden, falls die Produktionskapazität als Bezugsbasis gewählt wurde.

### **Beispiel**

Bei den Fixkosten 1700 DM, den Stückkosten von 3,60 DM und dem Stückpreis von 5 DM, wird die Rentabilitätsgrenze bei einer Produktion von 1214 Stück erreicht. Diese Stückzahl entspricht bei einer Produktionskapazität von 2000 Stück einem Anteil von 60,71 %. Der tabellarische Ausdruck des jeweiligen Gewinns kann dem Programmausdruck entnommen werden.

Literaturhinweis: [5]

```
100 REM .....RENTABILITAETSGRENZE
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT"  Rentabilitaetsgrenze"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 PRINT:PRINT" Welche Groesse soll als Bezugsbasis"
190 PRINT" benutzt werden?"
200 PRINT CHR$(150);STRING$(28,154);CHR$(156)
210 PRINT CHR$(149);SPACE$(28);CHR$(149)
220 PRINT CHR$(149);" Produktionskapazitaet ";CHR$(246);" 1  ";CHR$(149)
230 PRINT CHR$(149);SPACE$(28);CHR$(149)
240 PRINT CHR$(149);"          Umsatzerwartung ";CHR$(246);" 2  ";CHR$(149)
250 PRINT CHR$(149);SPACE$(28);CHR$(149)
260 PRINT CHR$(147);STRING$(28,154);CHR$(153):PRINT
270 INPUT W
280 IF W<>1 AND W<>2 THEN PRINT CHR$(7):GOTO 270
290 :
300 PRINT:INPUT" Festkosten in DM";F
310 PRINT:INPUT" Stueckkosten in DM";K
320 PRINT:INPUT" Stueckpreis in DM";P
330 :
340 ON W GOTO 360,500
350 :-
```

```

360 REM .....PRODUKTIONSKAPAZITAET ALS BASIS
370 PRINT:INPUT" Produktionskapazitaet in Stueck";C
380 S=F/((P-K)*C)
390 IF S>1 THEN PRINT CHR$(7);"FEHLER !":END
400 PRINT:PRINT STRING$(39,45)
410 PRINT USING " Rentabilitaetsgrenze = ##### Stueck";S*C
420 PRINT USING " = ##.#.% der Produktionskapazitaet";S*100
430 PRINT USING " Umsatz #####.## DM ";C*P
440 PRINT USING " Gewinn #####.## DM ";C*P-F*C*K
450 PRINT STRING$(40,45)
460 PRINT:INPUT" Tabelle erwuenscht (J/N)";A#
470 IF MID$(A#,1,1)="J" OR MID$(A#,1,1)="j" THEN GOSUB 610
480 END
490 :
500 REM .....UMSATZERWARTUNG ALS BASIS
510 PRINT:INPUT" Erwarteter Umsatz in Stueck";E
520 S=F/(P-K)
530 PRINT:PRINT STRING$(40,45)
540 PRINT USING " Rentabilitaetsgrenze = ##### Stueck";S
550 PRINT USING " Umsatz ##### Stueck";E*P
560 PRINT USING " Gewinn #####.## DM";E*P-F-E*K
570 PRINT STRING$(39,45)
580 END
590 :
600 REM .....TABELLE
610 PRINT:PRINT STRING$(40,45)

```

```
620 PRINT" Stueck    Umsatz      Kosten      Gewinn"
630 PRINT STRING$(39,45)
640 C1=INT(C/10)
650 FOR X=C1 TO C STEP C1
660   R=X*P
670   T=F+X*K
680   I=R-T
690   PRINT USING "   ####   #####.##   #####.##   #####.## ";X,R,T,I
700 NEXT X
710 PRINT STRING$(39,45)
720 RETURN
```

=====  
Rentabilitaetsgrenze  
=====

Welche Groesse soll als Bezugsbasis  
benutzt werden?

Produktionskapazitaet  $\nu$  1

Umsatzerwartung  $\nu$  2

? 1

Festkosten in DM? 1700

Stueckkosten in DM? 3.60

Stueckpreis in DM? 5

Produktionskapazitaet in Stueck? 2000

-----  
Rentabilitaetsgrenze = 1214 Stueck

= 60.7 % der Produktionskapazitaet

Umsatz 10000.00 DM

Gewinn 1100.00 DM  
-----

Tabelle erwuenscht (J/N)? j

Stueck	Umsatz	Kosten	Gewinn
200	1000.00	2420.00	-1420.00
400	2000.00	3140.00	-1140.00
600	3000.00	3860.00	-860.00
800	4000.00	4580.00	-580.00
1000	5000.00	5300.00	-300.00
1200	6000.00	6020.00	-20.00
1400	7000.00	6740.00	260.00
1600	8000.00	7460.00	540.00
1800	9000.00	8180.00	820.00
2000	10000.00	8900.00	1100.00



## 35. OPTIMALE ZUORDNUNG

Sollen z. B. 6 Vertreter in 6 Städte fahren oder 6 Bagger auf 6 verschiedene Baustellen geschafft werden, so sucht man eine Zuordnung

Vertreter <--> Stadt  
Bagger <--> Baustelle

so, daß die Summe der zurückgelegten Strecken oder Kosten ein Minimum wird.

Unterwirft man mehrere Arbeiter einem Test, wer an welcher Maschine welche Leistung erbringt, so wird man hier eine Zuordnung

Arbeiter <--> Maschine

so suchen, daß die Summe der erbrachten Leistungen maximal wird.

Sind die Entfernungen oder Kosten durch eine quadratische Matrix gegeben, so bedeutet die Zuordnung anschaulich, daß ein Element von jeder Zeile und Spalte der Matrix ausgesucht werden muß, so daß die Summe optimal ist. Die Menge aller Zuordnungen geht durch Permutationen in sich über. Eine naheliegende Lösung ist nun, alle möglichen Permutationen der zuzuordnenden Objekte zu erzeugen und diejenige auszuwählen, deren Kosten bzw. Bewertung optimal ist. Ein anderes Verfahren ist die Ungarische Methode (siehe [8]).

### **Zum folgenden Programm:**

Die Kostenmatrix und ihre Ordnung werden in Form von DATA-Werten eingelesen. Soll die Zuordnung maximal sein, so ist MAX, andernfalls MIN einzugeben.

**Beispiel:**

6 Bagger sollen von ihren Standorten an 6 verschiedene Baustellen gebracht werden. Die Entfernungen der Baustellen von den Standorten sind durch die Entfernungsmatrix gegeben:

Baustelle	1	2	3	4	5	6
Standort 1	60	35	28	53	29	26
Standort 2	81	43	37	23	36	45
Standort 3	42	42	33	47	43	51
Standort 4	29	70	42	53	48	37
Standort 5	81	69	40	66	69	60
Standort 6	10	21	32	31	24	77

Gesucht ist die minimale Summe der Entfernungen um jeden Bagger zu einer der Baustellen zu bringen.

Das Programm liefert die optimale Zuordnung

1 <--> 5  
2 <--> 4  
3 <--> 2  
4 <--> 6  
5 <--> 3  
6 <--> 1

d. h. die Bagger 1 bis 6 müssen zu den Baustellen 5, 4, 2, 6, 3 und 1 gebracht werden. Die minimale Entfernungssumme ist 181.

```

100 REM Optimale Zuordnung
110 :
120 CLS
130 PRINT STRING$(28,61)
140 PRINT"      Zuordnungsproblem"
150 PRINT STRING$(28,61)
160 :
170 READ N      : REM Anzahl der Zuordnungen
180 DIM A(N,N),Z(N),X(1,N),OPT(1)
190 :
200 REM Einlesen
210 PRINT:PRINT"Gegebene Bewertungsmatrix:"
220 FOR I=1 TO N
230     FOR J=1 TO N
240         READ A(I,J)
250         PRINT USING "### ";A(I,J);
260     NEXT J:PRINT
270 NEXT I:PRINT
280 READ OPT$
290 PRINT"** Bitte warten **"
300 :
310 REM Zuordnung
320 S=0
330 FOR I=1 TO N
340     Z(I)=I:X(0,I)=I:X(1,I)=I
350     S=S+A(I,I)

```

```
360 NEXT I
370 MX=S:MI=S
380 B=0
390 FOR I=1 TO N
400     B=B+A(I,Z(I))
410 NEXT I
420 IF B<=MX THEN 470
430 FOR I=1 TO N
440     X(1,I)=Z(I)
450 NEXT I
460 MX=B
470 IF B>=MI THEN 540
480 FOR I=1 TO N
490     X(0,I)=Z(I)
500 NEXT I
510 MI=B
520 :
530 REM Erzeugen einer neuen Permutation
540 FOR I=N-1 TO 1 STEP -1
550     FOR J=I+1 TO N
560         IF Z(I)>Z(J) THEN 590
570             H=Z(I):Z(I)=Z(J):Z(J)=H
580             J=N:GOTO 380
590     NEXT J
600     GOSUB 780 : REM Sortieren
610 NEXT I
```

```

620 :
630 REM Ausgabe
640 PRINT STRING$(30,45)
650 PRINT TAB(4)" Optimale Zuordnung"
660 PRINT STRING$(30,45)
670 IF OPT#="Max" OR OPT#="MAX" THEN P=1 ELSE P=0
680 OPT(1)=MX:OPT(0)=MI
690 FOR I=1 TO N
700 PRINT USING "   ### <-->###";I,X(P,I)
710 NEXT I
720 PRINT STRING$(30,45)
730 PRINT OPT#;". Bewertungssumme =";OPT(P)
740 PRINT STRING$(30,45)
750 END
760 :
770 REM Unterprogramm Sortieren
780 FOR J=I TO N-1
790   M=J
800   FOR K=J+1 TO N
810     IF Z(M)>Z(K) THEN M=K
820   NEXT K
830   IF M<>J THEN H=Z(J):Z(J)=Z(M):Z(M)=H
840 NEXT J
850 RETURN
860 :
870 DATA 6

```

880 DATA 60,35,28,53,29,26  
890 DATA 81,43,37,23,36,45  
900 DATA 42,42,33,47,43,51  
910 DATA 29,70,42,53,48,37  
920 DATA 81,69,40,66,69,60  
930 DATA 10,21,32,31,24,77  
940 DATA Min

=====  
                  Zuordnungsproblem  
=====

Gegebene Bewertungsmatrix:

60	35	28	53	29	26
81	43	37	23	36	45
42	42	33	47	43	51
29	70	42	53	48	37
81	69	40	66	69	60
10	21	32	31	24	77

\*\* Bitte warten \*\*

-----  
                  Optimale Zuordnung  
-----

1	<-->	5
2	<-->	4
3	<-->	2
4	<-->	6
5	<-->	3
6	<-->	1

-----  
Min. Bewertungssumme = 181  
-----

### 36. LINEARE REGRESSION

Unter der linearen Regression versteht man die Anpassung einer linearen Funktion, meist einer Geraden, an eine Schar von vorgegebenen Punkten so, daß die Summe der senkrechten Abstände der Punkte von der Geraden minimal ist. Dieses Verfahren stammt von dem schon erwähnten deutschen Mathematiker C.F. Gauß und heißt Methode der kleinsten Quadrate.

Sind  $(x_i, y_i)$  die Koordinaten der  $n$  Punkte, an die die Gerade

$$y = ax + b$$

angepaßt werden soll, so gilt nach der Methode der kleinsten Quadrate


$$a = \frac{n \sum x_i y_i - (\sum x_i) (\sum y_i)}{n \sum x_i^2 - (\sum x_i)^2}$$

und

$$b = \frac{\sum y_i - a \sum x_i}{n}$$

dabei bedeuten die Summenzeichen, daß die hinter dem  $\sum$  stehenden Terme addiert werden müssen.

Bundesausgaben in Mrd. DM



### Zum folgenden Programm

Nach Eingabe der Anzahl der Punkte und ihrer Koordinaten, berechnet das Programm nach der angegebenen Methode die Regressionsgerade. Der Funktionswert dieser Geraden kann an beliebigen Stellen ausgewertet werden; die Regressionsgerade kann auch zur Prognose benützt werden.

### Beispiel

Die Ausgaben des Bundes betragen nach [3]

Jahr	Mrd. DM
1970	88,0
1971	98,5
1972	111,1
1973	122,6
1974	134,0
1975	156,9
1976	162,5
1977	172,0
1978	189,5
1979	203,4
1980	215,7
1981	233,0

Zählt man die Jahre ab 1970 (= 0), so erhält man die Regressionsgerade

$$y = 13,11x + 85,15.$$

Sie zeigt anschaulich, daß die Bundesausgaben vom Stand 85,15 Mrd. im Jahr 1970 jährlich um durchschnittlich 13,11 Mrd. DM angewachsen sind. Durch Einsetzen von  $x=12$  in die Funktionsgleichung erhält man für 1982 die Prognose

242,4 Mrd. DM.

Wie gut die durch die Regression ermittelten Werte mit den tatsächlichen Werten übereinstimmen, kann dem Programmausdruck entnommen werden.

```
100 REM .....LINEARE REGRESSION
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT"  Lineare Regression"
150 PRINT STRING$(30,61)
160 :
170 REM .....Einlesen
180 S=0: T=0: U=0
190 V=0: W=0
200 READ N
210 DIM X(N),Y(N)
220 FOR I=1 TO N
230     READ X(I),Y(I)
240     S=S+X(I)
250     T=T+Y(I)
260     U=U+X(I)^2
270     V=V+Y(I)^2
280     W=W+X(I)*Y(I)
290 NEXT I
300 :
310 REM .....PARAMETER
320 A=(W-S*T/N)/(U-S*S/N)
330 B=(T-A*S)/N
340 :
350 REM .....AUSGABE
```

```

360 PRINT:PRINT" Regressionsgerade:";
370 PRINT USING " y= ###.## *x+###.## ";A,B
380 :
390 PRINT:PRINT STRING$(30,45)
400 PRINT"      x          y beob.      y berech."
410 PRINT STRING$(30,45)
420 FOR I=1 TO N
430     X=X(I):Y=Y(I)
440     PRINT USING "###.##      #####.##      #####.##";X,Y,A*X+B
450 NEXT I
460 PRINT STRING$(30,45)
470 :
480 REM .....Extrapolation
490 PRINT:INPUT" Weitere y-Werte gesucht (J/N)";A$
500 IF LEFT$(A$,1)="J" OR LEFT$(A$,1)="j" THEN 510 ELSE END
510 INPUT" x-Wert";X
520 PRINT STRING$(30,45)
530 PRINT USING "x = #####.##  y = #####.##";X,A*X+B
540 PRINT STRING$(30,45)
550 GOTO 490
560 :
570 REM .....DATEN
580 DATA 12
590 DATA 0,88.0,1,98.5,2,111.1,3,122.6,4,134.0,5,156.9,6,162.5
600 DATA 7,172.0,8,189.5,9,203.4,10,215.7,11,233.0

```

```
=====
  Lineare Regression
=====
```

Regressionsgerade:  $y = 13.11 * x + 85.15$

x	y beob.	y berech.
0.0	88.00	85.15
1.0	98.50	98.26
2.0	111.10	111.38
3.0	122.60	124.49
4.0	134.00	137.60
5.0	156.90	150.71
6.0	162.50	163.82
7.0	172.00	176.93
8.0	189.50	190.05
9.0	203.40	203.16
10.0	215.70	216.27
11.0	233.00	229.38

Weitere y-Werte gesucht (J/N)? j  
 x-Wert? 12

x = 12.0 y = 242.49

Weitere y-Werte gesucht (J/N)?



## 37. GLEITENDE MITTELWERTE

Um die bei Zeitreihen zufälligen und irregulären Schwankungen auszugleichen, verwendet man die Methode der gleitenden Mittelwerte. Dieses Glätten geschieht derart, daß jeder Monatswert als Mittelwert seiner 12 Nachbarwerte bestimmt wird. Da aber bei einer geraden Zahl von Werten kein Wert genau in der Mitte liegt, zieht man noch einen 13. Monatswert heran und berücksichtigt diesen und den 1. jeweils zur Hälfte. Die Formel für das  $i$ -te Monatsmittel ist somit

$$M(i) = (x(i-6)/2 + x(i-5) + \dots + \dots + x(i+5) + x(i+6)/2) / 12$$

dabei sind  $x(i)$  die beobachteten Monatswerte.

Damit diese Formel anwendbar ist, müssen mindestens 13 Monatswerte eingegeben werden. Für die ersten und letzten 6 Monate kann kein gleitender Durchschnitt ermittelt werden.

### **Zum folgenden Programm:**

Nach Eingabe von mindestens 13 Monatswerten in Form von DATA-Werten bestimmt das Programm die gleitenden Durchschnitte und druckt sie tabellarisch aus.

**Beispiel:**

Die folgende Tabelle zeigt den Umsatz eines Geschäfts in Tsd DM während 4 Jahren:

Monat	Jahr 1	Jahr 2	Jahr 3	Jahr 4
Jan	7.4	8.8	10.9	14.0
Feb	7.7	9.1	11.4	15.0
Mrz	7.3	9.3	11.8	14.9
Apr	7.8	9.2	12.0	15.5
Mai	8.4	9.0	12.2	15.3
Jun	8.5	9.3	11.6	14.5
Jul	8.9	9.8	11.0	14.4
Aug	8.7	9.6	11.4	13.0
Sep	9.1	10.1	12.0	12.0
Okt	9.2	10.3	12.6	11.5
Nov	8.9	10.5	13.7	11.6
Dez	8.6	10.3	14.4	12.0

Die gleitenden Mittelwerte für die Monate 7 bis 42 können dem Programmausdruck entnommen werden.

```
100 REM .....GLEITENDE MITTELWERTE
110 :
120 CLS
130 PRINT STRING$(30,61)
140 PRINT "      Gleitende Mittelwerte"
150 PRINT STRING$(30,61)
160 :
170 REM .....EINGABE
180 READ N:REM Anzahl der Monate
190 DIM X(N),G(N)
200 IF N>12 THEN 230 ELSE 210
210 BEEP:PRINT"Mittelwertbildung ist bei weniger als"
220 PRINT"als 12 Monaten nicht moeglich !":END
230 FOR I=1 TO N
240   READ X(I)
250 NEXT I
260 :
270 REM .....BERECHNUNG
280 S=(X(1)+X(13))/2
290 FOR I=2 TO 12
300   S=S+X(I)
310 NEXT I
320 G(7)=S/12
330 FOR I=8 TO N-6
340   S=S-X(I-7)/2-X(I-6)/2+X(I+5)/2+X(I+6)/2
350   G(I)=S/12
```

```

360 NEXT I
370 :
380 REM .....AUSGABE
390 PRINT STRING$(39,45)
400 PRINT"  Monat      Beobachtet      Geglaettet"
410 PRINT STRING$(39,45)
420 FOR I=1 TO N
430     PRINT USING " ###      #####.##";I,X(I);
440     IF I<7 OR I>N-6 THEN PRINT TAB(30)" --":GOTO 460
450     PRINT TAB(25) USING "#####.##";G(I)
460 NEXT I
470 PRINT STRING$(39,45)
480 END
490 :
500 DATA 48
510 DATA 7.4,7.7,7.3,7.8,8.4,8.5,8.9,8.7,9.1,9.2,8.9,8.6
520 DATA 8.8,9.1,9.3,9.2,9,9.3,9.8,9.6,10.1,10.3,10.5,10.3
530 DATA 10.9,11.4,11.8,12,12.2,11.6,11,11.4,12,12.6,13.7,14.4
540 DATA 14,15,14.9,15.5,15.3,14.5,14.4,13,12,11.5,11.6,12

```

=====  
 Gleitende Mittelwerte  
 =====

Monat	Beobachtet	Geglaettet
1	7.40	--
2	7.70	--
3	7.30	--
4	7.80	--
5	8.40	--
6	8.50	--
7	8.90	8.43
8	8.70	8.55
9	9.10	8.69
10	9.20	8.83
11	8.90	8.92
12	8.60	8.98
13	8.80	9.05
14	9.10	9.12
15	9.30	9.20
16	9.20	9.29
17	9.00	9.40
18	9.30	9.54
19	9.80	9.70
20	9.60	9.88
21	10.10	10.08
22	10.30	10.30
23	10.50	10.55
24	10.30	10.78
25	10.90	10.93
26	11.40	11.05
27	11.80	11.20
28	12.00	11.38
29	12.20	11.61
30	11.60	11.91
31	11.00	12.21
32	11.40	12.49
33	12.00	12.77
34	12.60	13.05
35	13.70	13.32
36	14.40	13.57
37	14.00	13.83
38	15.00	14.04
39	14.90	14.11
40	15.50	14.06
41	15.30	13.93
42	14.50	13.74
43	14.40	--
44	13.00	--
45	12.00	--
46	11.50	--
47	11.60	--
48	12.00	--



## 38. MONATS-INDIZES

Man unterscheidet im wesentlichen 4 Faktoren, die auf Zeitreihen einwirken:

- a. langfristige Trends, z.B. Inflation
- b. zyklische Einflüsse, z.B. Konjunkturschwankungen
- c. saisonale Einflüsse, z.B. das Weihnachtsgeschäft
- d. irreguläre und zufällige Fluktuationen.

Aufgabe der Zeitreihenanalyse ist es, Einflüsse dieser Faktoren zu erkennen und zu bereinigen. Durch Glätten von Daten können z.B. zufällige Schwankungen eliminiert werden.

Im folgenden Programm erfolgt das Glätten von Zeitreihen durch gleitende Mittelwertbildung. Eine andere Methode, das exponentielle Glätten, wird im folgenden Programm dargestellt.

Um Monatsdaten mehrerer Jahre zu glätten, verwendet man meist gleitende 12-Monats-Durchschnitte; d.h. jeder Wert wird so geändert, daß er den Mittelwert seiner 12 Nachbarwerte darstellt. Mittelt man nun noch die geglätteten Werte gleicher Monate, z.B. alle Januarwerte, so wird der Trend eliminiert. Bezieht man diese Monatsmittel auf ihren Mittelwert (= 100%), so erhält man die Indexwerte aller Monate.

### Zum folgenden Programm


Nach Eingabe der Anzahl der Jahre, der Monatswerte in Form von DATA-Anweisungen, berechnet das Programm die gleitenden Monatsdurchschnitte und die Monatsindex-Werte.

### Beispiel

Die folgende Tabelle zeigt den monatlichen Umsatz (in Tsd. DM) eines Geschäfts für den Zeitraum von vier Jahren:

		Jan	Feb	Mrz	Apr	Mai	Jun	Jul	Aug	Sep	Okt	Nov	Dez
Jahr	1	7,4	7,7	7,3	7,8	8,4	8,5	8,9	8,7	9,1	9,2	8,9	8,6
	2	8,8	9,1	9,3	9,2	9,0	9,3	9,8	9,6	10,1	10,3	10,5	10,3
	3	10,9	11,4	11,8	12,0	12,2	11,6	11,0	11,4	12,0	12,6	13,7	14,4
	4	14,0	15,0	14,9	15,5	15,3	14,5	14,4	13,0	12,0	11,5	11,6	12,0
einfacher Mittelwert		10,3	10,8	10,8	11,1	11,2	11,0	11,0	10,7	10,8	10,9	11,2	11,3
Indexwerte des einfachen Mittels		94,3	98,9	98,9	101,6	102,5	100,7	100,7	97,9	98,9	99,8	102,5	103,4
Indexwerte des gleitenden Mittels		98,5	102,3	103,1	104,0	102,6	99,3	98,0	95,9	98,7	99,4	99,8	98,3

In der graphischen Darstellung sind zum Vergleich auch die ungeglätteten Monats-Indexwerte, in der Graphik einfache Mittelwerte genannt, aufgenommen:



———— Indexwerte der gleitenden Mittelwerte  
- - - - - Indexwerte der einfachen Mittelwerte

```

100 REM .....MONATS-SAISONINDIZES
110 :
120 CLS
130 PRINT STRING$(26,61)
140 PRINT"      Saison-Indizes"
150 PRINT STRING$(26,61)
160 :
170 READ N:REM Anzahl der Jahre
180 T=12*N
190 DIM X(T+11),S(T+5),Y(T),Z(12),M$(12)
200 :
210 REM Einlesen der Monatsnamen
220 FOR I=1 TO 12
230   READ M$(I)
240 NEXT I
250 :
260 REM Einlesen der Data-Werte
270 V=0
280 FOR I=1 TO N
290   FOR J=1 TO 12
300     V=V+1
310     READ X(V)
320   NEXT J
330 NEXT I
340 :
350 REM .....BERECHNUNG

```

```
360 B=5: M=T-12
370 FOR I=1 TO N
380   FOR J=1 TO 12
390     B=B+1
400     C=B-6
410     FOR K=1 TO 12
420       C=C+1
430       S(B)=S(B)+X(C)
440     NEXT K
450   NEXT J
460 NEXT I
470 FOR J=6 TO T-7
480   Y(J+1)=(X(J+1)/((S(J)+S(J+1))/24))*100
490 NEXT J
500 FOR I=1 TO 12
510   Q=I+6
520   IF Q>12 THEN Q=Q-12
530   R=Q
540   FOR J=1 TO N
550     Z(Q)=Z(Q)+Y(R)/(N-1)
560     R=R+12
570   NEXT J
580   G=G+Z(Q)
590 NEXT I
600 S=1200/G
610 :
```

```

620 REM .....AUSGABE
630 PRINT:PRINT STRING$(30,45)
640 PRINT "  Monat      Saison-Index in %"
650 PRINT STRING$(30,45)
660 FOR I=1 TO 12
670   PRINT USING "          #####.## ";M$(I),Z(I)*5
680 NEXT I
690 PRINT STRING$(30,45)
700 PRINT TAB(6)"Mittel = 100 %"
710 PRINT STRING$(30,45)
720 END
730 :
740 DATA 4
750 DATA Jan, Feb, Mrz, Apr, Mai, Jun, Jul, Aug, Sep, Okt, Nov, Dez
760 DATA 7.4,7.7,7.3,7.8,8.4,8.5,8.9,8.7,9.1,9.2,8.9,8.6
770 DATA 8.8,9.1,9.3,9.2,9,9.3,9.8,9.6,10.1,10.3,10.5,10.3
780 DATA 10.9,11.4,11.8,12,12.2,11.6,11,11.4,12,12.6,13.7,14.4
790 DATA 14,15,14.9,15.5,15.3,14.5,14.4,13,12,11.5,11.6,12

```

=====  
Saison-Indizes  
=====

Monat	Saison-Index in %
Jan	98.54
Feb	102.34
Mrz	103.09
Apr	103.99
Mai	102.65
Jun	99.25
Jul	98.02
Aug	95.88
Sep	98.75
Okt	99.36
Nov	99.84
Dez	98.29

-----  
Mittel = 100 %  
-----



## 39. EXPONENTIELLES GLÄTTEN VON DATEN

Eine weitere Methode, zufällige Schwankungen zu eliminieren, ist das exponentielle Glätten. Sind  $x_i$  die Werte der Zeitreihe, so erhält man die geglätteten Werte  $y_i$  über die Vorschrift

$$\begin{cases} y_2 = x_1 \\ y_{i+1} = \alpha x_i + (1 - \alpha)y_i \quad i=2,3, \dots, n \end{cases}$$

Die Konstante  $\alpha$  bestimmt dabei den Einfluß der Nachbarwerte ( $0 < \alpha < 1$ ), je mehr sich  $\alpha$  dem Wert 1 nähert, desto mehr Nachbarwerte werden in die gewichtete Mittelwertbildung einbezogen.

Die Konstante  $\alpha$  wird nun so bestimmt, daß die Summe der quadratischen Abweichungen zwischen den Werten  $x_i$  und  $y_i$  möglichst klein wird. Dieses Vorgehen entspricht wieder der Methode der kleinsten Quadrate.

### Zum folgenden Programm

Die Zeitreihe und die Anzahl ihrer Werte werden in Form von DATA-Werten eingelesen. Für jeden Wert von  $\alpha$  für

0,1; 0,2; 0,3; . . . . . ; 0,9

wird die Summe der quadratischen Abweichungen berechnet und diejenigen geglätteten Werte ausgedruckt, die das kleinste Abweichungsmaß haben.

### Beispiel

Der Anteil des Mineralöls am Primärenergie-Verbrauch betrug in der Bundesrepublik (zitiert nach [3]):

Jahr	Anteil in %
1970	53,1
1971	54,7
1972	55,4
1973	55,2
1974	51,5
1975	52,1
1976	52,9
1977	52,1
1978	52,3
1979	50,7
1980	47,6
1981	44,8

Einlesen dieser Werte ins Programm liefert die im Programmausdruck gegebenen Werte.

Anteil des Mineralöls am Primärenergieverbrauch in %



Literaturhinweis: [16]

```
100 REM .....EXPONENTIELLES GLAETTEN VON DATEN
110 :
120 CLS
130 PRINT STRING$(35,61)
140 PRINT"   Exponent.Glaetten von Daten"
150 PRINT STRING$(35,61):PRINT
160 :
170 REM .....EINLESEN
180 READ N
190 DIM X(N),Y(N),Z(N)
200 FOR I=1 TO N
210 READ X(I)
220 NEXT I
230 :
240 REM .....BERECHNUNG
250 F1=1E+20
260 FOR A =.1 TO .9 STEP .1
270 F=0
280 Y(1)=X(1)
290 FOR I=2 TO N
300 Y(I)=A*X(I)+(1-A)*Y(I-1)
310 F=F+(Y(I)-X(I))^2
320 NEXT I
330 IF F<F1 THEN GOSUB 470
340 NEXT A
350 :
```

```

360 REM .....AUSGABE
370 PRINT USING " Beste Anpassung fuer Alpha= #.#";A1
380 PRINT:PRINT STRING$(35,45)
390 PRINT" Periode   Geg.Daten   Geglaettet"
400 PRINT STRING$(35,45)
410 FOR I=1 TO N
420 PRINT USING "      ##          #####.#          #####.# ";I,X(I),Z(I)
430 NEXT I
440 PRINT STRING$(35,45)
450 END
460 :
470 REM .....ZWISCHENSPEICHERN
480 F1=F
490 A1=A
500 FOR J=1 TO N
510 Z(J)=Y(J)
520 NEXT J
530 RETURN
540 :
550 DATA 12
560 DATA 53.1,54.7,55.4,55.2,51.5,52.1,52.9,52.1,52.3,50.7,47.6,44.8

```

=====  
 Exponent.Glaetten von Daten  
 =====

Beste Anpassung fuer Alpha= 0.8

Periode	Geg.Daten	Geglaettet
1	53.1	53.1
2	54.7	54.4
3	55.4	55.2
4	55.2	55.2
5	51.5	52.2
6	52.1	52.1
7	52.9	52.7
8	52.1	52.2
9	52.3	52.3
10	50.7	51.0
11	47.6	48.3
12	44.8	45.5



## 40. PROGNOSE DURCH EXPONENTIELLES GLÄTTEN

Wie die lineare Regression, kann auch das exponentielle Glätten (Programm 39) zur Prognose herangezogen werden. Für den fehlenden, zu glättenden Wert, verwendet man zur Prognose einen Mittelwert der gegebenen Zeitreihe. Die Anzahl der Zeitreihenwerte, die gemittelt werden sollen, ist im Programm frei wählbar.

### Zum folgenden Programm

Die Zeitreihe, die Anzahl ihrer Werte, die Anzahl der Werte, über die gemittelt werden soll, werden in Form von DATA-Werten eingelesen.

Zusätzlich zu den  $\alpha$ -Werten von Programm 39

0,1; 0,2; 0,3; . . . . . ; 0,9

werden noch die Werte 0,01 und 0,05 betrachtet. Um abschätzen zu können, wie stark der Prognosewert in Abhängigkeit von  $\alpha$  variiert, werden alle Prognosewerte ausgedruckt.

Wie bei Programm 39 wird auch derjenige Wert von  $\alpha$  angegeben, bei dem die quadratische Abweichungssumme von geglätteten und ungeglätteten Zeitreihenwerten am kleinsten ist.

### Beispiel

Der Index der Aktienkurse (Jahresdurchschnittswerte) in der BRD seit 1975 betrug (zitiert nach [3]):

Jahr	Index
1975	94
1976	102
1977	101
1978	109
1979	106
1980	100
1981	102

Die angegebenen Kurse sind bezogen auf den Stand vom 29.12.1972 (= 100%). Eingabe dieser Werte ins Programm liefert für 1982 die Prognose 101,9 für  $\alpha = 0,2$ .

Gibt man den Prognosewert für 1982 mit den anderen Daten ins Programm ein, so erhält man eine Prognose für 1983. Die so erhaltenen Werte werden für die folgenden Jahre immer unsicherer.

```
100 REM PROGNOSE DURCH EXPONENTIELLES GLAETTEN VON DATEN
110 :
120 CLS
130 PRINT STRING$(35,61)
140 PRINT"  Prognose durch Expon.Glaetten  "
150 PRINT STRING$(35,61)
160 :
170 REM .....DATENEINLESEN
180 READ N: REM ANZAHL DER DATEN
190 READ P: REM ANZAHL DER ZU MITTELNDEN ZEITRAEUME
200 IF P<1 OR P>=N THEN PRINT"DATENFEHLER": END
210 DIM Y(N),X(12),S(12)
220 FOR I=1 TO N
230 READ Y(I)
240 NEXT I
250 :
260 REM .....INIALISIEREN
270 S=0
280 FOR I=1 TO 10
290 X(I)=0
300 S(I)=0
310 NEXT I
320 FOR I=1 TO P
330 S=S+Y(I)
340 NEXT I
350 T=S/P
```

```
360 A=.01: M=2
370 GOSUB 760
380 A=.05: M=3
390 GOSUB 760
400 FOR J=1 TO 9
410     A=J/10
420     M=J+3
430     GOSUB 760
440 NEXT J
450 :
460 REM .....BESTIMMEN DER FEHLERSUMME
470 R=S(2): L=2
480 FOR I=3 TO 12
490     IF S(I)>R THEN 520
500     R=S(I)
510     L=I
520 NEXT I
530 IF L=3 THEN 560
540 IF L>3 THEN 570
550 A=.01: GOTO 590
560 A=.05: GOTO 590
570 A=.1*(L-3)
580 :
590 REM .....AUSGABE
600 PRINT:PRINT" Prognose fuer Zeitraum Nr.";N+1
610 PRINT USING " Bester Wert fuer Alpha #.## nach";A
```

```

620 PRINT" der Methode der kleinsten Quadrate"
630 PRINT STRING$(35,45)
640 PRINT"      Alpha      Wert"
650 PRINT STRING$(35,45)
660 PRINT USING "      .01      #####.## ";X(2)
670 PRINT USING "      .05      #####.## ";X(3)
680 FOR I=1 TO 9
690     A=I/10
700     K=3+I
710     PRINT USING "      #.##      #####.## ";A,X(K)
720 NEXT I
730 PRINT STRING$(35,45)
740 END
750 :
760 REM ....UNTERPROGRAMM EXPONENTIELLES GLAETTEN
770 Q=P+1
780 E=0: F=T
790 FOR I=Q TO N
800     E=E+(F-Y(I))^2
810     F=A*Y(I)+(1-A)*F
820     IF I<>N THEN 850
830     X(M)=F
840     S(M)=E
850 NEXT I
860 RETURN
870 :
880 REM .....DATEN
890 DATA 7
900 DATA 5

```

=====  
Prognose durch Expon.Glaetten  
=====

Prognose fuer Zeitraum Nr. 8  
Bester Wert fuer Alpha 0.20 nach  
der Methode der kleinsten Quadrate

Alpha	Wert
.01	102.37
.05	102.27
0.10	102.14
0.20	101.94
0.30	101.78
0.40	101.66
0.50	101.60
0.60	101.58
0.70	101.62
0.80	101.70
0.90	101.82



STATISTISCHE LEBENSERWARTUNG <sup>1)</sup>

Vollendetes Altersjahr <sup>2)</sup>	Männliche Personen	Weibliche Personen
	Lebenserwartung in Jahren	
0	67,87	74,36
1	68,61	74,80
2	67,70	73,88
5	64,86	71,01
10	60,03	66,14
15	55,16	61,22
20	50,53	56,39
25	45,93	51,54
30	41,24	46,70
35	36,58	41,90
40	31,99	37,15
45	27,54	32,52
50	23,27	28,01
55	19,25	23,67
60	15,52	19,46
65	12,20	15,48
70	9,43	11,88
75	7,19	8,79
80	5,40	6,30
85	3,97	4,46
90	2,87	3,20

1) 1972/74 (abgekürzte Berechnung)

2) Es beziehen sich: das Alter 0 auf den Zeitpunkt der Geburt, die anderen Altersangaben auf den Zeitpunkt, an dem jemand genau x Jahre alt geworden ist

Quelle: Statistisches Jahrbuch 1976, Seite 73



## FORMELSAMMLUNG

$p$  = Zinssatz in %

$i$  = Tilgungsrate in %

$q = 1 + \frac{p}{100}$  Zinsfaktor

$t$  = Zeit in Tagen

$K$  = Anfangskapital, Kredithöhe

$K_n$  = Endkapital

$B$  = Barwert

$E$  = Endwert

$A$  = jährliche Annuität

$m$  = Zahl der Zinsperioden pro Jahr

Zinsformel  $Z = \frac{Kpt}{360 \cdot 100}$  Zinsertrag

Kapitalendwert  $K_n = Kq^n$

Zinsfaktor, Rendite  $q = \sqrt[n]{\frac{K_n}{K}}$

Konformer Zinssatz  $\bar{p} = ({}^m\sqrt{q} - 1) \cdot 100\%$

Endbetrag regelmäßiger Zahlungen

$E = \frac{R(q^n - 1)}{q - 1}$  nachschüssig, jährliche Verzinsung

$E = \frac{Rq(q^n - 1)}{q - 1}$  vorschüssig, jährliche Verzinsung

$E = R\left(\frac{m}{q-1} + \frac{m-1}{2}\right) (q^n - 1)$  nachschüssig, unterjährig  
m-malige Verzinsung

$E = R\left(\frac{m}{q-1} + \frac{m+1}{2}\right) (q^n - 1)$  vorschüssig, unterjährig  
m-malige Verzinsung

## Barwert regelmäßiger Zahlungen

$$B = \frac{R(q^n - 1)}{q^n(q - 1)} \quad \text{nachschüssig, ganzjährige Verzinsung}$$

$$B = \frac{R(q^n - 1)}{q^n - 1(q - 1)} \quad \text{vorschüssig, ganzjährige Verzinsung}$$

$$B = \frac{R\left(\frac{m}{q-1} + \frac{m-1}{2}\right) (q^n - 1)}{q^n} \quad \text{nachschüssig, unterjährige m-malige Verzinsung}$$

$$B = \frac{R\left(\frac{m}{q-1} + \frac{m+1}{2}\right) (q^n - 1)}{q^n - 1} \quad \text{vorschüssig, unterjährige m-malige Verzinsung}$$

## Barwert einer ewigen Rente

$$B = \frac{R}{q-1} \quad \text{nachschüssig}$$

$$B = \frac{Rq}{q-1} \quad \text{vorschüssig}$$

## Umwandlung eines Kapitals in Rente

$$R = \frac{Kq^n(q-1)}{q^n-1} \quad \text{nachschüssig, ganzjährige Verzinsung}$$

$$R = \frac{Kq^{n-1}(q-1)}{q^n-1} \quad \text{vorschüssig, ganzjährige Verzinsung}$$

$$R = \frac{Kq^n}{\left(\frac{m}{q-1} + \frac{m-1}{2}\right)(q^n-1)} \quad \begin{array}{l} \text{nachschüssig, unterjährige} \\ \text{m-malige Verzinsung} \end{array}$$

$$R = \frac{Kq^{n-1}}{\left(\frac{m}{q-1} + \frac{m+1}{2}\right)(q^n-1)} \quad \begin{array}{l} \text{vorschüssig, unterjährige} \\ \text{m-malige Verzinsung} \end{array}$$

## Annuitätentilgung eines Darlehens

$$A = \frac{Kq^n(q-1)}{q^n-1}$$

$$i = \frac{p}{q^n-1}$$

$$n = \frac{\log\left(1 + \frac{p}{i}\right)}{\log q}$$

## Ratentilgung eines Darlehens

$$A_k = \frac{K}{100 \cdot n} [100 + p(n-k+1)] \quad \begin{array}{l} \text{Annuität am Ende} \\ \text{des } k\text{-ten Jahres} \end{array}$$



### LITERATURVERZEICHNIS

- [1] Ayres F.: Finanzmathematik, Düsseldorf, New York 1979
- [2] Bialy H., Olbrich M.: Optimierung, Leipzig 1975
- [3] Bundeswirtschaftsministerium: Leistung in Zahlen '81, Bonn 1982
- [4] Churchman C. W., Ackoff R. L., Arnoff E. L.: Operations Research, München, Wien 1971<sup>5</sup>
- [5] Hartmann G., Hertel S.: Grundlagen des betrieblichen Rechnungswesens Band 1—3, Rinteln 1980<sup>4</sup>
- [6] Herrmann D.: Mathematikprogramme in BASIC, Köln 1982
- [7] Herrmann D.: Wahrscheinlichkeitsrechnung/Statistik — 30 Programme, Wiesbaden 1983
- [8] Herrmann D.: Angewandte Matrizenrechnung, Wiesbaden 1985
- [9] Krüger K.: Finanzmathematik, Paderborn 1977
- [10] Löffelholz J.: Repetitorium der Betriebswirtschaftslehre, Wiesbaden 1980<sup>6</sup>
- [11] Laux H.: Entscheidungskriterien bei Unsicherheit in: Wirtschaftswissenschaftliches Studium 4/1975, München
- [12] Müller-Merbach H.: Mathematik für Wirtschafts-Wissenschaftler I, München 1974
- [13] Schröder H.: Die Effektivverzinsung, Wiesbaden 1978
- [14] Schwarze J.: Mathematik für Wirtschaftswissenschaftler Band 3, Herne, Berlin 1981<sup>5</sup>
- [15] Soom E.: Monte-Carlo-Methoden und Simulationstechnik, Bern, Stuttgart 1968
- [16] Vogt H.: Aufgaben und Beispiele zur Wirtschaftsmathematik, Würzburg, Wien 1976



D. Herrmann G. Schnellhardt  
**Schneider CPC**  
**Mathematik**



Programm-  
Beispiele für den  
Anwender

1

iwv

Dieses Buch enthält 39 mathematische Programme aus den Bereichen: Mehrregister-Arithmetik – Zahlentheorie – Kombinatorik – Algebra – Geometrie – numerische Mathematik. Neu ist die Langzahl-Arithmetik. Sie gestattet die Grundrechenarten für Zahlen bis 255 Stellen.

In Vorb. 2. Q. '85. 192 Seiten.  
Kart. DM 44,- / Fr. 44,-  
ISBN 3-88322-152-X

G. Schnellhardt S. Port  
**Schneider CPC**  
**dBase II**



Einführung  
Programmierung  
Anwendung

3

iwv

Das Buch führt den Leser mit vielen Beispielen aus der Praxis in das Datenbanksystem dBase II ein. Es sind keine Vorkenntnisse von dBase erforderlich. Das Werk ist ähnlich einer (PU)Programmierten Unterweisung aufgebaut und ist daher sehr gut verständlich.

1985. Ca. 300 Seiten.  
Kart. DM 48,- / Fr. 48,-  
ISBN 3-88322-154-6

J. Hegner  
**Schneider CPC**  
**Grafik**



Einführung  
Beispiele  
Anwendungen


4

iwv

Dieses Buch behandelt alle grafischen Fähigkeiten des Schneider CPC und führt in die Grafikprogrammierung ein. Von der elementaren Programmierung bis hin zur komplexen mehrfarbigen Abbildung im hochauflösten Grafikmodus werden zahlreiche Beispiele erläutert und entsprechend illustriert.

In Vorb. 2. Q. '85. 288 Seiten.  
Kart. DM 48,- / Fr. 48,-  
ISBN 3-88322-147-3

L. Dödt  
**Schneider CPC**  
**SuperCalc**



Einführung  
Beispiele  
Anwendungen

5

iwv

Dieses Buch wendet sich an SuperCalc-Anwender. Anfänger führt es schnell zur sicheren Handhabung, Fortgeschrittene finden ausgereifte Anwendungen. Beispiele mit ausführlicher Beschreibung machen das Buch auch für VisiCalc-Anwender interessant. Die Unterschiede zwischen VisiCalc und SuperCalc sind in einem Kapitel dargestellt.

In Vorb. 2. Q. '85. Ca. 200 Seiten.  
Kart. DM 48,- / Fr. 48,-  
ISBN 3-88322-155-4

Günther Daubach

**BASIC 80**  
**MICROSOFT**  
**BASIC 80**  
**MICROSOFT**  
**BASIC 80**  
**MICROSOFT**  
**BASIC 80**  
**MICROSOFT**  
**BASIC 80**

für Personalcomputer  
mit CP/M Betriebssystem  
aufgezeigt am Beispiel  
ITT 3030

iwv

Teil 1 gibt eine Einführung in die Grundlagen; Sprach-Interpreter und Compiler werden erläutert. Teil 2 enthält eine umfassende Beschreibung aller Anweisungen, Befehle und Funktionen, abgerundet mit zahlreichen Beispielen. Teil 3 beschreibt die Anwendung des Compilers und seine Unterschiede zum Interpreter.

1983. 306 Seiten. Spiralh. DM 56,- / Fr. 56,-  
ISBN 3-88322-024-8

Bernd Pol **Vom**  
**Umgang**  
**mit CP/M**

Eine allgemein-verständliche Einführung




CP/M für die Praxis 1

iwv


Das Buch ist in drei Teile gegliedert: Teil 1 führt in die Eigenschaften von Computern und CP/M im besonderen ein. Aufbauend auf diesen Kenntnissen werden im 2. Teil die zentralen CP/M-Hilfsprogramme vorgestellt. Der 3. Teil geht – nach einer Einführung in die Funktionsweise des 8080-Prozessors – auf die CP/M-Systembesonderheiten ein.

1982. 386 S. Mit zahlr. prakt. Beispielen.  
Geb. DM 48,- / Fr. 48,-  
ISBN 3-88322-004-3



Dieses Buch beschreibt den Leistungsumfang von FORTRAN-80 für Personalcomputer, wie es im Release 3.44 von MicroSoft festgelegt wurde. Alle Sprachelemente sind in detaillierter Form abgehandelt und in Beispielen dargestellt. Das Buch kann in dieser Form als Programmierhandbuch und auch als Nachschlagewerk genutzt werden.

1984. 344 Seiten.  
Spiralhb. DM 56,-/Fr. 56.-  
ISBN 3-88322-079-5



Dieses Buch ermöglicht die schnelle Einarbeitung in die Programmiersprache CBASIC. Alle Eigenschaften werden mit Beispielen erklärt. Syntaxdiagramme zeigen grafisch die Einsatzmöglichkeiten. Die Funktionen des Compilers, des Linkers und des Bibliotheksprogrammes LIB werden erläutert.

1983. 190 Seiten.  
Spiralhb. DM 56,-/Fr. 56.-  
ISBN 3-88322-034-5



Dieses Buch behandelt die Problematik der Schnittstellen V 24 und dem IEC-Bus. Die Übertragung zu peripheren Geräten (Drucker, Plotter, Terminal u. a.) ist gut verständlich aufgezeigt. Hardwareaufbau und Beispielanschlüsse machen das Buch zu einem guten Nachschlagewerk.

In Vorb. I.Q.'85. 200 Seiten.  
Geb. Ca. DM 48,-/ca. Fr. 48.-  
ISBN 3-88322-094-9



Wer hat nicht bereits verzweifelt versucht, das »Computerchinesisch« zu verstehen? Hier hilft das Wörterbuch der Computerei mit seinen über tausend Begriffen. Außerdem sind die wichtigsten Begriffe erklärt. Ein handliches Nachschlagewerk für jeden, der sich mit Computerei beschäftigt.

1983. 2., erw. Aufl. 144 Seiten.  
Kart. DM 32,-/Fr. 32.-  
ISBN 3-88322-026-4



Die Programmiersprache »C« ist besonders zur Erstellung schneller, maschinennaher Programme geeignet, weist jedoch gegenüber der Assemblerprogrammierung wesentliche Vorteile auf, wie z. B. lokale und globale Variable, Prozeduren, Funktionen usw. Diese Einführung setzt beim Leser keine umfassenden Kenntnisse voraus.

1984. 256 Seiten.  
Geb. DM 56,-/Fr. 56.-  
ISBN 3-88322-041-8



Dieses Buch führt Sie zur sicheren Handhabung der Tabellenkalkulation. Zahlreiche Anwendungen mit ausführlicher Beschreibung schließen sich an. Unter anderem: Werbeplanung, Reisekosten, Kapazitätsauslastung, Baufinanzierung, Reaktionszeiten. – Auch für die englische Version nutzbar.

In Vorb. Juni 1984. Ca. 250 Seiten. Spiralhb.  
Ca. DM 56,-/ca. Fr. 56.-  
ISBN 3-88322-074-4



D. Herrmann, G. Schnellhardt

## **Schneider CPC Band 2: Wirtschaft**

Für den Anwender des Schneider CPC bietet dieses Buch eine Reihe von Programmen aus den Bereichen Finanzmathematik, Unternehmensforschung und Betriebswirtschaft.

Wirtschaftliche Fragestellungen treffen jeden von uns, entweder als Steuerzahler, Sparer oder Kreditnehmer. So werden hier allen Praktikern, Studenten und Computerfans zahlreiche nützliche und anwendungsbezogene Programme aus vielfältigen Bereichen wie:

- Zins- und Renditen-Berechnungen
- Renten- und Tilgungsberechnungen
- Optimierungs- und Entscheidungstheorie
- Investitionsrechnung
- Abschreibung

vorgestellt.

Alle Programme werden durch zahlreiche Beispiele erläutert und durch vollständige Listings ergänzt.

ISBN 3-883**22-153**-8



**D. McFARLAND**  
**SCIENCE CENTER**  
**WISCONSIN**  
**2**

# AMSTRAD

# CPC



**MÉMOIRE ÉCRITE**  
**MEMORY ENGRAVED**  
**MEMORIA ESCRITA**



<https://acpc.me/>

[FRA] Ce document a été préservé numériquement à des fins éducatives et d'études, et non commerciales.

[ENG] This document has been digitally preserved for educational and study purposes, not for commercial purposes.

[ESP] Este documento se ha conservado digitalmente con fines educativos y de estudio, no con fines comerciales.